
      welcome

    


    Thanks for buying the MEAP of Elm in Action! I hope you find this a wonderfully practical journey into a delightful language.


    As an early adopter of Elm, I felt like a wide-eyed sprinter barreling down an exciting path that had not yet been paved. Now that I’ve had time to catch my breath, I’m eager to help pave that road for future travelers. My goal for this book is for you to enjoy learning Elm even more than I did...and I loved it so much I’m writing a book about it!


    More than anything, I hope this book gives you the confidence to use Elm professionally. As much as I enjoyed using Elm on a side project, it’s brought me even more joy at work. When I think back on what motivated my team to try it, it was the practical benefits: reliability, a quicker development loop, and a lower maintenance burden.


    The most authentic way I know to convey the benefits of building an application in Elm is...well, to build an application in Elm. That’s exactly what we’ll be doing over the course of this book: developing an application from start to finish, learning the language along the way. By the end you’ll have built, refactored, and tested an Elm code base. If using it at work sounds appealing, you’ll be able to convey its benefits based on firsthand experience!


    Throughout the MEAP process I intend both to release new chapters as I finish them, and to revise past chapters based on feedback from readers like you. I’d love it if you could leave some comments in the Author Online Forum. All feedback is helpful! Mentioning which parts felt difficult tells me where to focus my revision time, and mentioning your favorite parts tells me what not to change.


    Thanks again, and enjoy!


     


    —Richard Feldman

    1    Welcome to Elm


    This chapter covers


    ·   How and why to introduce Elm to a project


    ·   Using elm-repl


    ·   Building expressions


    ·   Writing and importing functions


    ·   Working with collections


    Back in 2014 I set out to rewrite a side project, and ended up with a new favorite programming language. Not only was the rewritten code faster, more reliable, and easier to refactor, writing it was the most fun I’d had in over a decade writing code. Ever since that project, I’ve been hooked on Elm.


    The rewrite in question was a writing application I’d built even longer ago, in 2011. Having tried out several writing apps over the course of writing a novel, and being satisfied with none, I decided to scratch my own itch and build my dream writing app. I called it Dreamwriter.


    For those keeping score: yes, I was indeed writing code in order to write prose better.


    Things went well at first. I built the basic Web app, started using it, and iterated on the design. Months later I’d written over fifty thousand words in Dreamwriter. If I’d been satisfied with that early design, the story might have ended there. However, users always want a better experience...and when the user and the developer are the same person, further iteration is inevitable.


    The more I revised Dreamwriter, the more difficult it became to maintain. I’d spend hours trying to reproduce bugs that knocked me out of my writing groove. At some point the copy and paste functions stopped working, and I found myself resorting to the browser’s developer tools whenever I needed to move paragraphs around.


    Right around when I’d decided to scrap my unreliable code base and do a full rewrite, a blog post crossed my radar. After reading it I knew three things:


    1. The Elm programming language compiled to JavaScript, just like Babel or CoffeeScript. (I already had a compile step in my build script, so this was familiar territory.)


    2. Elm used the same rendering approach as React.js—which I had recently grown to love—except Elm had rendering benchmarks that outperformed React’s!


    3. Elm’s compiler would catch a lot of the errors I’d been seeing before they could harm me in production. I did not yet know just how many it would catch.


    I’d never built anything with a functional programming language like Elm before, but I decided to take the plunge. I didn’t really know what I was doing, but the compiler’s error messages kept picking me up whenever I stumbled. Eventually I got the revised version up and running, and began to refactor.


    The refactoring experience blew me away. I revised the Elm-powered Dreamwriter gleefully, even recklessly—and no matter how dramatic my changes, the compiler always had my back. It would point out whatever corner cases I’d missed, and I’d go through and fix them. As soon as the code compiled, lo and behold, everything worked again. I felt invincible.


    I related my Elm experience to my coworkers at NoRedInk, and they were curious but understandably cautious. How could we find out if the team liked it without taking a big risk? A full rewrite may have been fine for Dreamwriter, but it would have been crazy to attempt that for our company’s entire front-end.


    So we introduced Elm gently, by rewriting just one portion of one production feature in Elm. It went well, so we did a bit more. And then more.


    Today our front-end is as Elm-powered as we can make it, and our team has never been happier. Our test suites are smaller, yet our product is more reliable. Our feature set has grown more complex, yet refactoring remains delightful. We swap stories with other companies using Elm about how long our production code has run without throwing a single runtime exception.


    In this book we’ll explore all of these benefits.


    After learning some basics, we’ll build an Elm Web application the way teams typically do: ship a basic version that works, but which has missing features and some technical debt. As we advance through the chapters, we’ll expand and refactor this application, adding features and paying off technical debt as we learn more about Elm. We’ll debug our business logic. We’ll even interoperate with JavaScript, using its vast library ecosystem to avoid reinventing the wheel. By the end of the book we will have transformed our application into a more featureful product, with a more maintainable code base, than the one we initially shipped.


    With any luck, we’ll have a lot of fun doing it.


    Welcome to Elm!


    1.1   How Elm Fits In


    Elm can be used either as a replacement for in-browser JavaScript code, or as a complement to it. You write some .elm files, run them through Elm’s compiler, and end up with plain old .js files that the browser runs as normal. If you have separate stylesheets that you use alongside JavaScript, they’ll work the same way alongside Elm.

    Figure 1.1 Elm files are compiled to plain old JavaScript files


    The appropriate Elm-to-JavaScript ratio can vary by project. Some projects may want primarily JavaScript and only a touch of Elm for business logic or rendering. Others may want a great deal of Elm but just a pinch of JavaScript to leverage its larger ecosystem. No single answer applies to every project.


    What distinguishes Elm from JavaScript is maintainability.


    Handwritten JavaScript code is notoriously prone to runtime crashes like “undefined is not a function.” In contrast, Elm code has a reputation for never throwing runtime exceptions in practice. This is because Elm is built around a small set of simple primitives like expressions, immutable values, and managed effects. That design lets the compiler identify lurking problems just by scanning your source code. It reports these problems with such clarity that it has earned a reputation for user-friendliness even among programming legends.


    That should be an inspiration for every error message.


    
      —John Carmack, after seeing one of Elm’s compiler errors

    


    Having this level of compiler assistance makes Elm code dramatically easier to refactor and debug, especially as code bases grow larger. There is an up-front cost to learning and adopting Elm, but you reap more and more maintainablity benefits the longer the project remains in active development.


    Tip Most teams that use Elm in production say they used a “planting the seed” approach. Instead of waiting for a big project where they could build everything in Elm from the ground up, they rewrote a small part of their existing JavaScript code base in Elm. This was low-risk and could be rolled back if things did not go as planned, but having that small seed planted in production meant they could grow their Elm code at a comfortable pace from then on.


    Although Elm is in many ways a simpler language than JavaScript, it’s also much younger. This means Elm has fewer off-the-shelf solutions available for any given problem. Elm code can interoperate with JavaScript code to piggyback the larger JavaScript library ecosystem, but Elm’s design differs enough from JavaScript’s that incorporating JavaScript libraries takes effort.


    Balancing these tradeoffs depends on the specifics of a given project. Let’s say you’re on a team where people are comfortable with JavaScript but are new to Elm. Here are some projects I’d expect would benefit from learning and using Elm:


    ·   Feature-rich Web applications whose code bases are large or will grow large


    ·   Individual features that will be revised and maintained over an extended period of time


    ·   Projects where most functionality comes from in-house code, not off-the-shelf libraries


    In contrast, I’d stick to a more familiar language and toolset for projects like these:


    ·   Proof-of-concept prototypes that will not be maintained long-term


    ·   Time-crunched projects where learning a language is unrealistic given the deadline


    ·   Projects that will consist primarily of gluing together off-the-shelf components


    We’ll explore these tradeoffs in more detail throughout the course of the book.


    1.2   Expressions


    To get our feet wet with Elm, let’s tap into one of the most universal traits across the animal kingdom: the innate desire to play.


    Researchers have developed many theories as to why we play, including to learn, to practice, to experiment, and of course for the pure fun of it. These researchers could get some high-quality data by observing a member of the homo sapiens programmerus species in its natural environment for play—the Read-Eval-Print Loop, or REPL.


    You'll be using Elm's REPL to play as you take your first steps as an Elm programmer.


    1.2.1   Using elm-repl


    The Elm Platform includes a nice REPL called elm-repl, so if you have not installed the Elm Platform yet, head over to Appendix A to get hooked up.


    Once you’re ready, enter elm-repl at the terminal. You should see this prompt:
  ---- elm-repl 0.18.0 -------------------------------------------------------  :help for help, :exit to exit, more at https://github.com/elm-lang/elm-repl ----------------------------------------------------------------------------> 

    Alexander Graham Bell invented the telephone over a century ago. There was no customary greeting back then, so Bell suggested one: lift the receiver and bellow out a rousing “Ahoy!” Thomas Edison later proposed the alternative “Hello,” which stuck, and today programmers everywhere append “World” as the customary way to greet a new programming language.


    Let’s spice things up a bit, shall we? Enter this at the prompt.
> "Ahoy, World!"

    You should see this response from elm-repl:
"Ahoy, World!" : String

    Congratulations, you are now an Elm programmer!


    NOTE  To focus on the basics, for the rest of this chapter we'll omit the type annotations that elm-repl prints. For example, the previous code snippet would have omitted the : String portion of "Ahoy, World!" : String. We'll get into these annotations in Chapter 3.


    If you're the curious sort, by all means feel free to play as we continue. Enter things that occur to you, and see what happens! Whenever you encounter an error you don't understand yet, picture yourself as a tiger cub building intuition for physics through experimentation: adorable for now, but powerful in time.


    1.2.2   Building Expressions


    Let's rebuild our "Ahoy, World!" greeting from two parts, and then play around from there. Try entering these into elm-repl.


    Listing 1.1  Combining Strings
> "Ahoy, World!" "Ahoy, World!"   > "Ahoy, " ++ "World!" "Ahoy, World!"   > "Pi is " ++ toString pi ++ " (give or take)"  ❶ "Pi is 3.141592653589793 (give or take)"

    ❶   toString is a standalone function, not a method


    In Elm, we use the ++ operator to combine strings, instead of the + operator JavaScript uses. At this point you may be wondering: Does Elm even have a + operator? What about the other arithmetic operators?


    Let’s find out by experimenting in elm-repl!


    Listing 1.2 Arithmetic Expressions
> 1234 + 103 1337   > 12345 - (5191 * -15) ❶  90210   > 2 ^ 11 2048   > 49 / 10 4.9   > 49 // 10             ❷  4   > -5 % 2               ❸ 1

    ❶   Nest expressions via parentheses


    ❷   Integer division (decimals get truncated)


    ❸   Remainder after integer division


    Sure enough, Elm has both a ++ and a + operator. They are used for different things:


    ·   The ++ operator is for appending. Using it on a number is an error.


    ·   The + operator is for addition. It can only be used on numbers.


    You will see this preference for being explicit often in Elm. If two operations are sufficiently different—in this case, adding and appending—Elm implements them separately, so each implementation can do one thing well.


    Strings and Characters


    Elm also distinguishes between strings and the individual UTF-8 characters that comprise them. Double quotes in Elm represent string literals, just like in JavaScript, but single quotes in Elm represent character literals.


    Table 1.1 Strings and Characters


    
      
        	
          
            Elm Literal

          

        

        	
          
            Result

          

        
      


      
        	
          "a"

        

        	
          a string with a length of 1

        
      


      
        	
          'a'

        

        	
          a single character

        
      


      
        	
          "abc"

        

        	
          a string with a length of 3

        
      


      
        	
          'abc'

        

        	
          error: character literals must contain exactly 1 character

        
      


      
        	
          ""

        

        	
          an empty string

        
      


      
        	
          ''

        

        	
          error: character literals must contain exactly 1 character

        
      

    


    Comments


    Comments also work a bit differently in Elm:


    ·   Use -- instead of // for inline comments


    ·   Use {- -} instead of /* */ for block comments


    Let’s see these in action!


    Listing 1.3 Characters, Comments, and Constants
> 'a'  -- This is a comment. It will be ignored.                ❶  'a'   > "a"  {- This is a block comment. It will also be ignored. -}  ❷  "a"   > milesPerHour = 88                                             ❸  88   > milesPerHour 88 

    ❶   JavaScript comment: //


    ❷   JavaScript comment: /* ... */


    ❸   JavaScript: const milesPerHour = 88;


    Naming values with Constants


    In the last two lines of code above, we did something new: we assigned the constant milesPerHour to the value 88.


    Definition A constant assigns a name to a value. Once assigned, this name cannot be later reassigned to a different value in the same scope.


    There are a few things to keep in mind when naming constants.


    ·   The name must begin with a lowercase letter. After that it can be a mix of letters, numbers, and underscores.


    ·   By convention, all letters should be in one uninterrupted sequence. For example, map4 is a reasonable name, but map4ever is not, because the sequence of letters is interrupted by the 4.


    ·   Because of the previous two rules, you should never use snake_case or SCREAMING_SNAKE_CASE to name constants. Use camelCase instead.


    ·   If you absolutely must know whether the compiler will accept some_raD__TH1NG___ as a valid constant name, remember: what happens in elm-repl stays in elm-repl.


    Assigning constants to expressions


    Not only can you assign constants to literal values, you can also assign them to expressions.


    DEFINITION An expression is anything that evaluates to a single value.


    Here are some expressions we’ve seen so far.


    
      
        	
          
            Expression

          

        

        	
          
            Evaluates to

          

        
      


      
        	
          "Ahoy, " ++ "World!"

        

        	
          "Ahoy, World!"

        
      


      
        	
          2 ^ 11

        

        	
          2048

        
      


      
        	
          pi

        

        	
          3.141592653589793

        
      


      
        	
          42

        

        	
          42

        
      

    


    Note Since an expression is anything that evaluates to a value, literal values like "Ahoy, World!" and 42 are expressions too—just expressions that have already been fully evaluated.


    Expressions are the basic building block of Elm applications. This is different from JavaScript, which offers many features as statements instead of expressions.


    Consider these two lines of JavaScript code.
label = (num > 0) ? "positive" : "negative"  // ternary expression label = if (num > 0) { "positive" } else { "negative" }  // if-statement

    The first line is ternary expression. Being an expression, it evaluates to a value, and JavaScript happily assigns that value to label.


    The second line is an if-statement, and since statements do not evaluate to values, trying to assign it to label yields a syntax error.


    This distinction does not exist in Elm, as Elm programs express logic using expressions only. As such, Elm has if-expressions instead of if-statements. As we will see in Chapter 2, every Elm application is essentially one big expression built up from lots of smaller ones!


    1.2.3   Booleans and Conditionals


    There aren’t terribly many booleans out there—just the two, really—and working with them in Elm is similar to working with them in JavaScript. There are a few differences, though.


    ·   You write True and False instead of true and false


    ·   You write /= instead of !==


    ·   To negate values, you use Elm’s not function instead of JavaScript’s ! prefix


    Let’s try them out!


    Listing 1.4 Boolean Expressions
> pi == pi         ❶  True               ❷    > pi /= pi         ❸  False              ❹    > not (pi == pi)   ❺  False   > pi <= 0 || pi >= 10 False   > 3 < pi && pi < 4 ❻  True 

    ❶   JavaScript: pi === pi


    ❷   JavaScript: true


    ❸   JavaScript: pi !== pi


    ❹   JavaScript: false


    ❺   JavaScript: !(pi === pi)


    ❻   3 < pi < 4 would be an error


    Now let’s say it’s a lovely afternoon at the North Pole, and we’re in Santa’s workshop writing a bit of UI logic to display how many elves are currently on vacation. The quick-and-dirty approach would be to add the string " elves" after the number of vacationing elves, but then when the count is 1 we’d display "1 elves", and we’re better than that.


    Let’s polish our user experience with the if-expression shown in Figure 1.2.

    Figure 1.2 Comparing an Elm if-expression to a JavaScript ternary


    Like JavaScript ternaries, Elm if-expressions require three ingredients:


    1. A condition


    2. A branch to evaluate if the condition passes


    3. A branch to evaluate otherwise


    Each of these ingredients must be expressions, and the if-expression itself evaluates to the result of whichever branch got evaluated. You’ll get an error if any of these three ingredients are missing, so make sure to specify an else branch every time!


    NOte There is no such thing as “truthiness” in Elm. Conditions can be either True or False, and that’s it. Life is simpler this way.


    Now let’s say we modified our pluralization conditional to include a third case:


    ·   If we have one Elf, evaluate to "elf"


    ·   Otherwise, if we have a positive number of elves, evaluate to "elves"


    ·   Otherwise, we must have a negative number of elves, so evaluate to "anti-elves"


    In JavaScript you may have used else if to continue branching conditionals like this. It’s common to use else if for the same purpose in Elm, but it’s worth noting that else if in Elm is nothing more than a stylish way to combine the concepts we learned a moment ago.


    Check it out!


    Listing 1.6 Using else if
if elfCount == 1 then                                    ❶    "elf"                                                  ❶  else                                                     ❶    (if elfCount >= 0 then "elves" else "anti-elves")      ❶      if elfCount == 1 then                                    ❷    "elf"                                                  ❷  else (if elfCount >= 0 then                              ❷     "elves"                                                ❷  else                                                     ❷    "anti-elves")                                          ❷      if elfCount == 1 then                                    ❸    "elf"                                                  ❸  else if elfCount >= 0 then                               ❸    "elves"                                                ❸  else                                                     ❸   "anti-elves"                                           ❸ 

    ❶   Use an if-expression inside else


    ❷   Rearrange some whitespace


    ❸   Drop the parentheses


    This works because the else branch of an if-expression must be an expression, and it just so happens that if-expressions themselves are expressions. As shown in Figure 1.3, all it takes is putting an if-expression after another one’s else, and voilà! Additional branching achieved.

    Figure 1.3 The else if technique: use an if-expression as the else branch of another if-expression


    Nesting expressions for fun and profit is a recurring theme in Elm, and we’ll see plenty more recipes like else if thoughout the book.


    Chapter 3 will add a powerful new conditional to our expression toolbox, one which has no analogue in JavaScript: the case-expression.


    1.3   Functions


    Earlier we wrote this expression:
elfLabel = if vacationingElves == 1 then "elf" else "elves"

    Suppose it turns out a general-purpose singular/plural labeler would be really useful, and we want to reuse similar logic across the code base at Santa’s Workshop. Search results might want to display "1 result" and "2 results" as appropriate, for example.


    We can write a function to make this pluralization logic easily reusable.


    Definition Elm functions represent reusable logic. They are not objects. They have no fields, no prototypes, and no ability to store state. All they do is accept values as arguments, and then return a value.


    If you thought expressions would be a recurring theme in Elm, wait ‘til you see functions!


    1.3.1   Defining Functions


    Let’s define our first function: isOdd. It will take a number and then:


    ·   Return True if the number is odd


    ·   Return False if the number is even


    We can define isOdd in elm-repl and try it out right away.


    Listing 1.7 Defining a function
> isOdd num = num % 2 == 1   ❶  <function>   > isOdd 5                    ❷  True   > (isOdd 5)                  ❸  True   > isOdd (5 + 1)              ❹ False

    ❶   JavaScript: function isOdd(num) { return num % 2 === 1; }


    ❷   JavaScript: isOdd(5)


    ❸   JavaScript: (isOdd(5))


    ❹   JavaScript: isOdd(5 + 1)


    As you can see, in Elm we put the function parameter name before the = sign. We also don’t surround the function body with { }. And did you notice the return keyword is nowhere to be seen? That’s because Elm doesn’t have one! In Elm, a function body is a single expression, and since an expression evaluates to a single value, Elm uses that value as the function’s return value. This means all Elm functions return values!


    For our isOdd function, the expression num % 2 == 1 serves as the function’s body, and provides its return value.


    
      Refactoring out an early return

    


    In JavaScript, return is often used to exit a function early. This is harmless when used responsibly, but can lead to unpleasant surprises when used in the middle of large functions. Elm does not support these unpleasant surprises, because it has no return keyword.


    Let’s refactor the early return out of this function:


     
function capitalize(str) {     if (!str) {         return str;    ❶      }       return str[0].toUpperCase() + str.slice(1); }

    ❶   Early return


     


    Without making any other changes, we can refactor this early return into a ternary:


     
function capitalize(str) {     return !str ? str : str[0].toUpperCase() + str.slice(1); }

     


    Poof! There it goes. Since JavaScript’s ternaries are semantically analogous to Elm’s if-expressions, this code is now much more straightforward to rewrite in Elm. More convoluted JavaScript functions may require more steps than this, but it is always possible to untangle them into plain old conditionals.


    Removing an early return is one of many quick refactors you can do to ease the transition from legacy JavaScript to Elm, and we’ll look at more of them throughout the book. When doing these, do not worry if the intermediate JavaScript code looks ugly! It’s intended to be a stepping stone to nicer Elm code, not something to be maintained long-term.


    
       

    


    Let’s use what we just learned to generalize our previous elf-labeling expression into a reusable pluralize function. Our function this time will have a longer definition than last time, so let’s use multiple lines to give it some breathing room. In elm-repl, you can enter multiple lines by adding \ to the end of the first line and indenting the next line.


    Note Indent with spaces only! Tab characters are syntax errors in Elm.


    Listing 1.8 Using multiple REPL lines
> pluralize singular plural count = \ |     if count == 1 then singular else plural   ❶  <function>   > pluralize "elf" "elves" 3                     ❷  "elves"   > pluralize "elf" "elves" (round 0.9)           ❸ "elf"

    ❶   Don’t forget to indent!


    ❷   No commas between arguments!


    ❸   (round 0.9) returns 1


    When passing multiple arguments to an Elm function, separate the arguments with whitespace and not commas. That last line of code is an example of passing the result of one function call, namely round 0.9, as an argument to another function. (Think about what would happen if we did not put parentheses around (round 0.9)...how many arguments would we then be passing to pluralize?)


    1.3.2   Importing Functions


    So far we’ve only used basic operators and functions we wrote ourselves. Now let’s expand our repertoire of functions by using one from an external module.


    DEFINITION A module is a named collection of Elm functions and other values.


    The String module is one of the core modules that ships with Elm. Additional modules can be obtained from Elm’s official package repository, copy-pasting code from elsewhere, or through a back-alley rendezvous with a shadowy figure known as Dr. Deciduous. Chapter 4 will cover how to do the former, but neither the author nor Manning Publications endorses obtaining Elm modules through a shadowy back-alley rendezvous.


    Let’s import the String module and try out two of its functions, toLower and toUpper.


    Listing 1.9 Importing functions
> String.toLower "Why don't you make TEN louder?" "why don't you make ten louder?"   > String.toUpper "These go to eleven.""THESE GO TO ELEVEN."

    Observant readers may note a striking resemblance between Elm’s String.toUpper function and the toUpperCase() method one finds on JavaScript strings. This is the first example of a pattern we will encounter many times!


    JavaScript has several ways of organizing string-related functionality: fields on a string, methods on a string, or methods on the String global istelf.


    In contrast, Elm strings have neither fields nor methods. The String module houses the standard set of string-related features, and exposes them in the form of plain old functions like toLower and toUpper.


    Table 1.2 String Functionality Comparison


    
      
        	
          
            JavaScript

          

        

        	
          
            Elm

          

        
      


      
        	
          "storm".length

        

        	
          String.length "storm"

        
      


      
        	
          "dredge".toUpperCase()

        

        	
          String.toUpper "dredge"

        
      


      
        	
          String.fromCharCode(something)

        

        	
          String.fromChar something

        
      

    


    Not only is this organizational pattern consistent within the String module, it’s consistent across Elm. Want a standard date-related feature? Look no further than the functions in the Date module. Regular Expression functions? Hit up the Regex module.


    Methods are never the answer in Elm; over here it’s all vanilla functions, all the time.


    tip Complete documentation for String, Date, Regex, and other tasty modules can be found in the core section of the package.elm-lang.org website.


    We’ll learn more about modules in the coming chapters, including how to write our own!


    Using String.filter to filter out characters


    Another useful function in the String module is filter. It lets us filter out unwanted characters from a string, such as non-numeric digits from a phone number.


    To do this, we must give filter a function which specifies which characters to keep. The function will take a single character as an argument and return True if we should keep that character or False if we should chuck it. Figure 1.4 illustrates using String.filter to remove dashes from a US telephone number.

    Figure 1.4 Using String.filter to remove dashes from a US phone number


    As in JavaScript, Elm functions are first-class values that can be passed around just like any other value. This lets us provide filter with the function it expects by defining that function and then passing it in as a plain old argument.


    Listing 1.10 Filtering with a named function
> isKeepable character = character /= '-'  ❶  <function>   > isKeepable 'z' True   > isKeepable '-' False   > String.filter isKeepable "800-555-1234"  ❷ "8005551234"

    ❶   A function describing which characters to keep


    ❷   Passing our function to String.filter


    This normalizes telephone numbers splendidly. Alexander Graham Bell would be proud!


    String.filter is one of the higher-order functions (that is, functions which accept other functions as arguments) that Elm uses to implement customizable logic like this.


    1.3.3   Creating scope with let-expressions


    Let’s say we find ourselves removing dashes from phone numbers so often, we want to make a reusable function for it. We can do that with our trusty isKeepable function:
withoutDashes str = String.filter isKeepable str

    This works, but in a larger Elm program, it might be annoying having isKeepable in the global scope like this. After all, its implementation is only useful to withoutDashes. Can we avoid globally reserving such a nicely self-documenting name?


    Absolutely! We can scope isKeepable to the implementation of withoutDashes using a let-expression.


    Definition A let-expression adds locally scoped constants to an expression.


    Figure 1.5 shows how we can implement withoutDashes using a single let-expression.


    [image: ]



    Figure 1.5 Anatomy of the wild let-expression


    The above code does very nearly the same thing as entering the following in elm-repl:
> dash = '-' > isKeepable character = character /= dash> withoutDashes str = String.filter isKeepable str

    In both versions, the implementation of withoutDashes boils down to String.filter isKeepable str. The only difference between the two is the scope of dash and isKeepable.


    ·   In the elm-repl version above, dash and isKeepable are in the global scope.


    ·   In Figure 1.5, dash and isKeepable are scoped locally to the let-expression.


    Whenever you see a let-expression, you can mentally replace it with the part after its in keyword—in this case, String.filter isKeepable str. All the values between let and in are intermediate constants that are no longer in scope once the expression after in gets evaluated.


    Note The indentation in Figure 1.5 is no accident! In a multiline let-expression, the let and in keywords must be at the same indentation level, and all other lines in the let-expression must be indented further than they are.


    Anywhere you’d write a normal expression, you can swap in a let-expression instead. Because of this, you don’t need to learn anything new to define locally-scoped constants inside function bodies, branches of if-expressions, or anyplace else.


    Wherever you want some local scope, reach for a refreshing let-expression!


    1.3.4   Anonymous Functions


    Anonymous functions in Elm work the same way they do in JavaScript: like named functions but without the name.


    Listing 1.11 Named and Anonymous Functions
function area(w, h) { return w * h; }  ❶    function(w, h) { return w * h; }       ❷    area w h = w * h  ❸   \w h -> w * h     ❹ 

    ❶   JavaScript named function


    ❷   JavaScript anonymous function


    ❸   Elm named function


    ❹   Elm anonymous function


    Elm’s anonymous functions differ from its named functions in three ways.


    1. They have no names


    2. They begin with a \


    3. Their parameters are followed by -> instead of =


    Once defined, anonymous functions and named functions work the same way; you can always use the one in place of the other. For example, the following do exactly the same thing:
isKeepable char = char /= '-' isKeepable = \char -> char /= '-'

    Let’s use an anonymous function to call String.filter in one line instead of two, then see if we can improve the business logic! For example, we can try using Char.isDigit to cast a wider net, filtering out any non-digit characters instead of just dashes.


    Listing 1.12 Filtering with anonymous functions
> String.filter (\char -> char /= '-') "800-555-1234" "8005551234"   > String.filter (\char -> char /= '-') "(800) 555-1234" "(800) 5551234"  ❶    > import Char > String.filter (\char -> Char.isDigit char) "(800) 555-1234" "8005551234"     ❷    > String.filter Char.isDigit "(800) 555-1234"  ❸ "8005551234"

    ❶   Our simple filter fell short here


    ❷   Much better!


    ❸   Refactor of previous approach


    Anonymous functions are often used with higher-order functions like String.filter.


    1.3.5   Operators


    So far we’ve seen functions such as  String.filter, as well as operators such as ++, -, and ==. How do operators and functions relate?


    As it turns out, Elm’s operators are functions! There are a few things that distinguish operators from normal functions:


    ·   Operators must always accept exactly two arguments—no more, no fewer.


    ·   Normal functions have names that begin with a letter. You typically call them by writing the name of the function followed by its arguments. This is prefix-style calling.


    ·   Operators have names that contain neither letters nor numbers. You typically call them by writing the first argument, followed by the operator, followed by the second argument. This is infix-style calling.


    Wrapping an operator in parentheses treats it as a normal function—prefix-style calling and all!

    Figure 1.6 Calling the (-) operator in both infix-style and prefix-style


    Let’s play with some operators in elm-repl:


    Listing 1.13  Operators are functions
> (/) <function>   > divideBy = (/) <function>   > 7 / 2     ❶  3.5   > (/) 7 2   ❷  3.5   > divideBy 7 23.5

    ❶   infix-style calling


    ❷   prefix-style calling


    Operator Precedence


    Try entering an expression involving both arithmetic operators and (==) into elm-repl:
> 3 + 4 == 8 - 1 True : Bool

    Now consider how we’d rewrite this expression in prefix-style:
> (==) ((+) 3 4) ((-) 8 1) True : Bool

    Notice anything about the order in which these operators appear? Reading the infix-style expression from left to right, you see + first, then ==, and finally -. In the prefix-style expression, the order is different: first you see ==, then +, and finally -. Why is this?


    They get reordered because (==), (+), and (-) have different precedence values.

    Figure 1.7 (==) gets evaluated after (+) and (-) because it has lower precedence


    Definition In any expression containing multiple operators, the operators with higher precedence get evaluated before those with lower precedence. This only applies to infix-style calls, as all prefix-style calls implicitly have the same precedence.


    There isn’t much formal documentation on operators’ relative precedence values, but operators that appear in many programming languages (such as the (==), (+), and (-) operators) tend to work similarly in Elm to how they do everywhere else.


    We’ll dig into operator precedence in greater depth in later chapters.


    Normal function calls have top precedence


    Here are two ways to write the same thing:
> negate 1 + negate 5 -6   > (negate 1) + (negate 5)-6

    These two are equivalent because normal function calls have higher precedence than any operator. This means any time you want to pass the results of two normal function calls to an operator, you won’t need to add any parentheses! You’ll still get the result you wanted.


    Operator Associativity


    Besides precedence, the other factor that determines evaluation order for operators called in infix-style is whether the operators are left-associative, right-associative, or non-associative. Every operator is one of these.


    An easy way to think about operator associativity is in terms of where the implied parentheses go. Infix expressions involving left-associative operators, such as arithmetic operators, have implied parentheses that cluster on the left:


    Table 1.3 Implied parentheses for the (-) operator


    
      
        	
          
            Parentheses Shown

          

        

        	
          
            Expression

          

        

        	
          
            Result

          

        
      


      
        	
          none

        

        	
            10 -  6  - 3

        

        	
          1

        
      


      
        	
          assuming left-associative

        

        	
          ((10 -  6) - 3)

        

        	
          1

        
      


      
        	
          assuming right-associative

        

        	
           (10 - (6 - 3))

        

        	
          7

        
      

    


    If (-) were right-associative, 10 - 6 - 3 would have parentheses clustering on the right, meaning it would evaluate to (10 - (6 - 3)) and the undesirable result of 10 - 6 - 3 == 7. Good thing arithmetic operators are left-associative!


    Non-associative operators cannot be chained together. For example, foo == bar == baz does not result in clustered parentheses, it results in an error!


    In Chapter 7, we’ll see how to define our own operators, including specifying their associativity and precedence values.


    1.4   Collections


    Elm’s most basic collections are lists, records, and tuples. Each has varying degrees of similarity to JavaScript’s arrays and objects, but one way in which they differ from JavaScript collections is in that Elm collections are always immutable.


    DEFINITION An immutable value cannot be modified in any way once created.


    This is in contrast to JavaScript, where some values (like strings and numbers) are immutable, but collections (like arrays and objects) can be mutated.


    1.4.1   Lists


    An Elm list has many similarities to a JavaScript array.


    ·   You can create one with a square bracket literal, e.g. [ "one fish", "two fish" ]


    ·   You can ask for its first element


    ·   You can ask for its length


    ·   You can iterate over its elements in various ways


    An Elm list does have some differences, though.


    ·   It is immutable


    ·   It has no fields or methods. You work with it using functions from the List module.


    ·   Because it is a linked list, you can ask for its first element, but not for other individual elements. (If you need to ask for elements at various different positions, you can first convert from an Elm List to an Elm Array. We’ll discuss Elm Arrays in Chapter 3.)


    ·   You can combine it with another list using the ++ operator. In JavaScript this is done with the concat method rather than an operator.


    ·   All elements in an Elm list must have a consistent type. For example, it can be a “list of numbers” or a “list of strings,” but not a “list where strings and numbers intermingle.” (Making a list containing both strings and numbers involves first creating wrapper elements for them, using a feature called union types that we’ll cover in Chapter 3.)


    Although Elm supports both (immutable) lists and (also immutable) arrays, lists are used far more often because they have better performance characteristics in typical Elm use cases. We’ll get more into the performance differences between linked lists and arrays in Chapter 10.


    Here are some examples of how Elm lists and JavaScript arrays differ.


    Table 1.4 Contrasting JavaScript Arrays and Elm Lists


    
      
        	
          
            JavaScript Array

          

        

        	
          
            Elm List

          

        
      


      
        	
          [ 1, 2, 3 ].length

        

        	
          List.length [ 1, 2, 3 ]

        
      


      
        	
          [ "one fish", "two fish" ][0]

        

        	
          List.head [ "one fish", "two fish" ]

        
      


      
        	
          [ "one fish", "two fish" ][1]

        

        	
          No arbitrary position-based element access

        
      


      
        	
          [ 1, 2 ].concat([ 3, 4 ])

        

        	
          [ 1, 2 ] ++ [ 3, 4 ]

        
      


      
        	
          [ 1, 2 ].push(3)

        

        	
          Cannot be modified; use e.g. append instead

        
      


      
        	
          [ 1, "Paper", 3 ]

        

        	
          All elements in a list must have a consistent type

        
      

    


    Let’s focus on that last one. Why must all elements in an Elm list have a consistent type?


    To understand how this requirement benefits us, let’s consider the List.filter function, which works like the String.filter function we used earlier.


    Recall that String.filter takes a function which returns True when the given character should be kept, and False when it should be dropped. List.filter differs only in that the function you provide doesn’t necessarily receive characters—instead it receives elements from the list, whatever they may be.


    Let’s see that in action. Quick! To elm-repl!


    Listing 1.14 Filtering lists
> List.filter (\char -> char /= '-') [ 'Z', '-', 'Z' ]   ❶  ['Z','Z']   > List.filter (\str -> str /= "-") [ "ZZ", "-", "Top" ]  ❷  ["ZZ","Top"]   > import Char > List.filter Char.isDigit [ '7', '-', '9' ]             ❸  ['7','9']   > List.filter (\num -> num % 2 == 1) [ 1, 2, 3, 4, 5 ]   ❹ [1,3,5]

    ❶   Same function we passed to String.filter earlier


    ❷   Strings instead of characters


    ❸   Works just like with String.filter


    ❹   Keep only the odd numbers


    Here’s how we would rewrite that last line of code in JavaScript:
[ 1, 2, 3, 4, 5 ].filter(function(num) { return num % 2 === 1; })

    This looks straightforward enough, but JavaScript arrays permit inconsistent element types. Without looking it up, can you guess what happens if we change it to the following?
[ 1, "2", "cat", 4, "5", "" ].filter(function(num) { return num % 2 === 1; })

    Will it crash? Will it happily return numbers? What about strings? It’s a bit of a head-scratcher.


    Because Elm requires consistent element types, this is a no-brainer: in Elm it would be an error. Even better, it would be an error at build time—meaning you can rest easy knowing whatever surprises would result from executing this code will not inflict pain on your users. Requiring consistent element types means all lists in Elm guarantee this level of predictability.


    By the way, the above filter() call successfully returns [ 1, "5" ]. (Like, duh, right?)


    1.4.2   Records


    We’ve now seen how JavaScript’s mutable arrays resemble Elm’s immutable lists. In a similar vein, JavaScript’s mutable objects resemble Elm’s immutable records.


    Definition A record is a collection of named fields, each with an associated value.


    Whereas array and list literals between the two languages are syntactically identical, where JavaScript object literals use : to separate fields and values, Elm record literals use = instead.


    Let’s get a taste for some of their other differences.


    
      
        	
          
            JavaScript Object

          

        

        	
          
            Elm Record

          

        
      


      
        	
          { name: "Li", cats: 2 }

        

        	
          { name = "Li", cats = 2 }

        
      


      
        	
          ({ name: "Li", cats: 2 }).toString()

        

        	
          toString { name = "Li", cats = 2 }

        
      


      
        	
          ({ name: "Li", cats: 2 }).cats

        

        	
          ({ name = "Li", cats = 2 }).cats

        
      


      
        	
          ({ name: "Li", cats: 2 })["cats"]

        

        	
          Fields can only be accessed directly, using a dot

        
      


      
        	
          ({ name: "Li", cats: 2 }).cats = 3

        

        	
          Cannot be modified. (New cat? New record!)

        
      


      
        	
          { NAME: "Li", CATS: 2 }

        

        	
          Fields have the same naming rules as constants

        
      


      
        	
          ({ name: "Li", cats: 2 }).__proto__

        

        	
          No prepackaged fields, only the ones you define

        
      


      
        	
          Object.keys({ name: "Li", cats: 5 })

        

        	
          No listing of field names is available on demand

        
      


      
        	
          Object.prototype

        

        	
          Records have no concept of inheritance

        
      

    


    Wow—compared to objects, records sure don’t do much! It’s like all they do is sit around holding onto the data we gave them. (Yep.) Personally I’ve found Elm’s records a welcome reprieve from the intricacies of JavaScript’s this keyword.


    Record updates


    Record updates let us concisely obtain a new record by copying the old one and changing only the specified values. (As we will see in Chapter 10, behind the scenes Elm does not actually copy the entire record—that would be slow!—but rather only the parts that will be different.)


    Let’s use this technique to represent someone obtaining an extra cat, going from { name = "Li", cats = 2 } to { name = "Li", cats = 3 } by way of a record update.


    Listing 1.15 Record updates
> catLover = { name = "Li", cats = 2 } { name = "Li", cats = 2 }   > catLover { name = "Li", cats = 2 }   > withThirdCat = { catLover | cats = 3 }         ❶  { name = "Li", cats = 3 }   > withThirdCat { name = "Li", cats = 3 }   > catLover                  ❷  { name = "Li", cats = 2 }   ❷    > { catLover | cats = 88, name = "LORD OF CATS" } ❸ { name = "LORD OF CATS", cats = 88 }

    ❶   Record update syntax


    ❷   Original record unmodified!


    ❸   Update multiple fields (order doesn’t matter)


    Record updates let us represent this incremental evolution without mutating our records or recreating them from scratch. In Chapter 2 we’ll represent our application state with a record, and use record updates to make changes based on user interaction.


    1.4.3   Tuples


    Lists let us represent collections of varying size, whose elements share a consistent type. Records let us represent collections of fixed fields, but whose corresponding values may have varied types.


    Tuples introduce no new capabilities to this mix, as there is nothing a tuple can do that a record couldn’t. Compared to records, though, what tuples bring to the party is conciseness.


    DEFINITION A tuple is a record-like value whose fields are accessed by position rather than by name.


    In other words, tuples are for when you want a record, but don’t want to bother naming its fields. They are often used for things like key-value pairs where writing out { key = "foo", value = "bar" } would add verbosity but not much clarity.


    Let’s try some out!


    Listing 1.17 Using Tuples
> ( "Tech", 9 ) ("Tech",9)   > Tuple.first ( "Tech", 9 )   ❶  "Tech"   > Tuple.second ( "Tech", 9 )  ❷ 9

    ❶   Return first element (only works on 2-element tuples)


    ❷   Return second element (only works on 2-element tuples)


    You can only use the Tuple.first and Tuple.second functions on tuples that contain two elements. If they have more than two, you can use tuple destructuring to extract their values.


    Definition Tuple destructuring extracts the values inside a tuple and assigns them to constants in the current scope.


    Let’s use tuple destructuring to implement a function that takes a tuple of three elements.


    Listing 1.18 Tuple Destructuring
> multiply3d ( x, y, z ) = x * y * z  ❶  <function>   > multiply3d ( 6, 7, 2 ) 84   > multiply2d someTuple = let ( x, y ) = someTuple in x * y ❷ <function>

    ❶   Destructuring a tuple into three constants: x, y, and z


    ❷   Destructuring a tuple inside a let-expression


    As demonstrated in Listing 1.18, once you have named the values inside the tuple, you can use them just like you would any other constant.


    Tip Mind the difference between a tuple and a parenthetical function call! ( foo, bar ) is a tuple, whereas ( foo bar ) is a call to the foo function passing bar as an argument. A simple mnemonic to remember the difference is “comma means tuple.”


    Table 1.5 Comparing Lists, Records, and Tuples


    
      
        	
          
            List

          

        

        	
          
            Record

          

        

        	
          
            Tuple

          

        
      


      
        	
          Variable Length

        

        	
          Fixed Length

        

        	
          Fixed Length

        
      


      
        	
          Can Iterate Over

        

        	
          Cannot Iterate Over

        

        	
          Cannot Iterate Over

        
      


      
        	
          No Names

        

        	
          Named Fields

        

        	
          No Names

        
      


      
        	
          Immutable

        

        	
          Immutable

        

        	
          Immutable

        
      

    


    Since any tuples can be represented (more verbosely) using a record instead, it’s often better to refactor long tuples—say, with more than three elements—into records. Choose tuples or records based on whichever would yield more readable code; their performance characteristics are equivalent.


    1.5   Summary


    We’re off to a fantastic start! First we discussed some of the toughest problems Web programmers face: crashing is too easy in JavaScript, and maintenance is too error-prone. Then we learned how Elm addresses these problems, with a design that prioritizes maintainability and a helpful compiler that catches would-be runtime exceptions before they can cause user pain. From there we dove in and wrote our first Elm code in elm-repl.


    Here is a brief review of things we covered along the way.


    ·   The ++ operator combines strings and lists, whereas the + operator is for addition only.


    ·   Double quotes refer to strings. Single quotes refer to individual UTF-8 characters.


    ·   let-expressions introduce scoped constants to an expression.


    ·   There is no concept of “truthiness” in Elm, just True and False.


    ·   if foo /= bar then "different" else "same" is an if-expression. Like JavaScript ternaries, if-expressions require an else branch and always evaluate to a value.


    ·   Lists like [ 3, 1, 4 ] are immutable. Their elements must share a consistent type.


    ·   List.filter (\num -> num < 0) numbersList returns a list containing all the negative numbers in the original numbersList


    ·   catLover = { name = "Li", cats = 2 } assigns a record to the constant catLover. Once assigned, constants cannot be reassigned.


    ·   { catLover | cats = 3 } returns a new record that is the same as the catLover record, except the cats value is now 3.


    ·   ( foo, bar ) destructures a tuple such as ( 2, 3 ). In this example, foo would be 2 and bar would be 3.


    Table 1.6 summarizes some of the differences between JavaScript and Elm.


    Table 1.6 Differences between JavaScript and Elm


    
      
        	
          
            JavaScript

          

        

        	
          
            Elm

          

        
      


      
        	
          // This is an inline comment

        

        	
          -- This is an inline comment

        
      


      
        	
          /* This is a block comment */

        

        	
          {- This is a block comment -}

        
      


      
        	
          true && false

        

        	
          True && False

        
      


      
        	
          "Ahoy, " + "World!"

        

        	
          "Ahoy, " ++ "World!"

        
      


      
        	
          "A spade" === "A spade"

        

        	
          "A spade" == "A spade"

        
      


      
        	
          "Calvin" !== "Hobbes"

        

        	
          "Calvin" /= "Hobbes"

        
      


      
        	
          Math.pow(2, 11)

        

        	
          2 ^ 11

        
      


      
        	
          Math.trunc(-49 / 10)

        

        	
          -49 // 10

        
      


      
        	
          n % 2 === 1 ? "odd" : "even"

        

        	
          if n % 2 == 1 then "odd" else "even"

        
      


      
        	
          nums.filter(function(n) { ... })

        

        	
          List.filter (\n -> n % 2 == 1) nums

        
      


      
        	
          function pluralize(s, p, c) { ... }

        

        	
          pluralize singular plural count = ...

        
      

    


    We also learned about several differences between normal functions and operators:


    Table 1.7 Differences between normal functions and operators


    
      
        	
          
            Function

          

        

        	
          
            How to identify one

          

        

        	
          
            Calling style

          

        

        	
          
            Examples

          

        
      


      
        	
          Normal

        

        	
          Name begins with a letter

        

        	
          prefix-style

        

        	
          negate, not, toString

        
      


      
        	
          Operator

        

        	
          Name has no letters or numbers

        

        	
          infix-style

        

        	
          (++), (*), (==)

        
      

    


    In Chapter 2 we’ll expand on what we’ve learned here to create a working Elm application.


    Let’s go build something!

  


  
    2    Your First Elm Application


    This chapter covers


    ·   Declaratively rendering a page


    ·   Managing state with Model-View-Update


    ·   Handling user interaction


    Elm applications are built to last. They have a reputation for being scalable, easy to refactor, and difficult to crash unexpectedly. Since JavaScript applications have...erm...a different reputation, it stands to reason that Elm must be doing things differently. And so it is!


    Where each line of code in a JavaScript application can potentially result in a change or effect—like “update that text!” or “send this to the server!”—the code in an Elm application builds up a description of what the program should do in response to various inputs. Elm’s compiler translates this description into the appropriate JavaScript commands for the browser to run at the appropriate times, and the end user may have no idea Elm was involved at all.


    In this chapter we’ll build our first Elm application: Photo Groove, a simple photo browsing Web app where users select thumbnails to view larger versions. We’ll create a user interface using declarative rendering and manage state using the Elm Architecture. By the end, we will have a fully functioning application—and a code base we can build on for the rest of the book!


    2.1   Rendering a Page


    Since the very early days of the Web, browsers have been translating HTML markup into a Document Object Model (or DOM for short) which represents the structure of the current page. The DOM consists of DOM nodes, and it’s only by changing these nodes that Web applications can modify the current page on the fly.


    In this chapter we’ll work with the two most common types of DOM nodes: elements and text nodes.


    ·   Elements have a tagName (such as "div" or "img"), and may have child DOM nodes.


    ·   Text nodes have a textContent property instead of a tagName, and are childless.


    As Figure 2.1 shows, elements and text nodes can freely intermingle inside the DOM.


    [image: ]



    Figure 2.1 Intermingling element DOM nodes and text nodes


    Here we’ve pulled back the curtain on the markup <p>How <i>do</i> you do?</p> to see that despite its two element tags—namely <p> and <i>—we are actually working with five DOM nodes here! The other three are not elements, but rather text nodes.


    2.1.1   Describing a page using the Html Module


    When describing how a page looks in Elm, we do not write markup. Instead, we call functions to create representations of individual DOM nodes. The most flexible of these functions is called node, and as Figure 2.2 shows, its arguments line up neatly with the analogous markup.

    Figure 2.2 Representing a button using HTML markup (top) and Elm’s node function (bottom)


    There are four function calls in this line of Elm code. Can you spot them?


    1. A call to the node function, passing three arguments: the string "button", a list of attributes, and a list of child nodes


    2. A call to the class function, passing "funky"


    3. A call to the id function passing "submitter"


    4. A call to the text function passing "Submit"


    These are plain old Elm functions. Each returns a representation of some portion of the DOM: a button element, a text node, and some class and id attributes. You can call these functions anytime you like, and pass their return values to other functions as normal.


    In Elm we usually refer to a “a virtual DOM node” as “Html” for short. This name comes from the Html module, which provides a plethora of intuitively-named functions that let you avoid calling node directly. For example, the Html module’s img function is shorthand for calling node passing "img" as the first argument. The following two lines are equivalent:
node "img" [ src "logo.png" ] []       img  [ src "logo.png" ] []

    It’s best practice to use helper functions like img as much as possible, and to fall back on node only in cases where no helper function is available. (For example, you may notice that there is no helper for the deprecated <blink> element. I’m not saying you should call node "blink" [] [ text "<BLINK> LIVES AGAIN" ], but I’m not not saying it either.)


    Rendering a page


    Let’s use what we’ve learned to render our first page with Elm!


    For the rest of this chapter we’ll be building an application called Photo Groove. Eventually we’ll add features like searching and viewing larger versions, but first we need to render a basic page that says “Photo Groove” across the top, with some thumbnail images below.


    Since our output is visual this time, elm-repl won’t get us very far. Instead, let’s bring up the Try Elm website at http://elm-lang.org/try and enter the following into the left-hand pane.


    Listing 2.2 Building a view in Try Elm
import Html exposing (div, h1, img, text)    ❶  import Html.Attributes exposing (..)         ❶      view model =     div [ class "content" ]         [ h1 [] [ text "Photo Groove" ]      ❷          , div [ id "thumbnails" ]            ❸              [ img [ src "http://elm-in-action.com/1.jpeg" ] []             , img [ src "http://elm-in-action.com/2.jpeg" ] []             , img [ src "http://elm-in-action.com/3.jpeg" ] []  ❹              ]         ]     main =  ❺     view "no model yet"

    ❶   We’ll discuss “exposing” later


    ❷   h1 element with an empty attributes list


    ❸   Put commas at the start of the line


    ❹   img element with an empty children list


    ❺   We’ll discuss “main” later


    Press Compile. The words “Photo Groove” should appear in the right-hand pane, followed by three images. (Depending on the size of your browser, the images may not be side-by-side.)

    Figure 2.3 The Try Elm website


    Congratulations! You’ve rendered your first user interface in Elm.


    Figure 2.4 shows the DOM structure of the interface we just rendered.

    Figure 2.4 The DOM structure of your first Elm user interface


    Notice how the helper functions that create elements—in this case, div, h1, and img—take exactly two arguments in all cases:


    1. A list of attributes. If an element has no attributes, we pass [] like so:

    h1 [] [ text "Photo Groove" ]


    2. A list of child DOM nodes. If an element has no children, we pass [] like so:

    img [ src "1.jpeg" ] []


    If an element has neither attributes nor children? In that case we pass [] [] like so:
br [] []

    This pattern is consistent throughout the Html module, and it’s worth following if you ever decide to make a custom element helper of your own using the node function.


    Importing unqualified values with ‘exposing’


    We’ve now used several functions from the Html module, but we wrote them in a different style from the way we did in Chapter 1. Back then we wrote functions like String.filter in a qualified style—that is, we included the String module’s name right there in the function call. With the Html module’s functions, we used an unqualified style—we wrote div instead of Html.div, we wrote h1 instead of Html.h1, and so forth.


    We could do this because we used exposing when we imported the Html module:
import Html exposing (div, h1, img, text)

    This line of code both imports the Html module so we can use its contents, and also brings Html.div, Html.h1, Html.img, and Html.text into the global scope. That lets us refer to them as div, h1, img, and text without the prefix of Html.


    We could have achieved a similar result by assigning them to constants:
import Html   div = Html.div h1 = Html.h1 img = Html.imgtext = Html.text

    However, since this pile of code can be replaced by a single line—import Html exposing (div, h1, img, text)—it’s normal to use exposing for this purpose instead.


    Exposing everything with (..)


    When we imported the Html module, we listed exactly which values we wanted to expose: div, h1, img, and text. For the Html.Attributes module, we wrote this instead.
import Html.Attributes exposing (..)

    Using exposing (..) means “expose everything,” which lets us use every value in the Html module in an unqualified style. Let’s change our first import to use exposing (..) instead:
import Html exposing (..)

    Now we won’t need to extend the list of div, h1, img, and text whenever we want to use a new element type. Everything the Html module has to offer is now in our global scope!


    
      When to use qualified over unqualified imports

    


    In Chapter 1 we wrote out String.toUpper and List.filter, instead of toUpper and filter. Here we’re doing the opposite, writing img and div instead of Html.img and Html.div.


    This begs the question: when is it a good idea to use the qualified style (with the module name prefixed) over the unqualified style? The unqualified style is more concise, so why not use exposing (..) every time?


    There are two primary downsides to unqualified imports. One is that unqualified names can become ambiguous. Try this in elm-repl:
> import String exposing (..) > import List exposing (..) > filter

    You’ll get an error saying that filter is ambiguous. After importing and exposing both String.filter and List.filter, it’s no longer clear which of the two you meant! (In cases like this you can still use the qualified style to resolve the ambiguity, so if you now put String.filter or List.filter into elm-repl, they will still work as normal.)


    Unqualified imports are also less self-documenting. Suppose you come across code that says partition foo bar, and you’ve never seen partition before. Naturally you wonder: “How can I find out what partition does? Is it defined in this file?” You search through the file and can’t find it, so it must come from an import. You scroll up to the imports and discover a long list of exposing (..) declarations. Argh! partition could be in any of those!


    This could take awhile...


    Suppose instead you see the code List.partition foo bar. You want to know what List.partition does, so you bring up the documentation for the List module on http://package.elm-lang.org. You learn about List.partition, then get on with your day!


    Scenarios like this are why it’s best practice to write things in a qualified way by default.


    Still, sometimes there’s a good reason to prefer the unqualified style—like how unqualified Html functions are designed to resemble HTML markup. In these cases, it’s best to limit yourself to one exposing (..) (or perhaps one “family” of them, such as Html and Html.Attributes) per file. This way if you encounter an unfamiliar function of mysterious origin, you’ll have the fewest modules to hunt through to find its documentation!


    
       

    


    2.1.2   Building a Project


    Now that we’ve gotten something on the screen, let’s add some styles! We could style our elements inline using the Html.Attributes.style attribute, but instead we’ll organize things the traditional way: with our CSS declarations in a separate file.


    The only way to get multiple files involved in the same page is to give a browser some HTML markup, so our first step in the process of styling our application will be to create an index.html file. Since Try Elm doesn’t let you bring your own HTML file to the party, this means we’ll need to start building our project locally.


    1. Make a new directory called PhotoGroove and open it in a terminal.


    2. Create a file called PhotoGroove.elm inside this directory.


    3. Copy the Try Elm code from Listing 2.2 into PhotoGroove.elm.


    4. Add the line module PhotoGroove exposing (..) to the top of PhotoGroove.elm.


    5. Change import Html exposing (div, h1, img, text) to use exposing (..).


    6. When the dust settles, we should have a PhotoGroove.elm file that looks like this:


    Listing 2.3 PhotoGroove.elm
module PhotoGroove exposing (..)   ❶    import Html exposing (..)          ❷  import Html.Attributes exposing (..)   view model =     div [ class "content" ]         [ h1 [] [ text "Photo Groove" ]         , div [ id "thumbnails" ]             [ img [ src "http://elm-in-action.com/1.jpeg" ] []             , img [ src "http://elm-in-action.com/2.jpeg" ] []             , img [ src "http://elm-in-action.com/3.jpeg" ] []             ]         ]   main =   ❸     view "no model yet"

    ❶   “This file describes a module called PhotoGroove”


    ❷   Remember: this uses “exposing (..)” now!


    ❸   This will be the application’s entry point


    
      Why Commas in Front?

    


    When writing a multi-line literal in JavaScript, the usual convention is to put commas at the end of each line. Consider the following code:
rules = [     rule("Do not talk about Sandwich Club."),     rule("Do NOT talk about Sandwich Club.")     rule("No eating in the common area.") ]

     


    Did you spot the mistake? There’s a comma missing after the second call to rule, meaning this is not syntactically valid JavaScript. Running this code will result in a SyntaxError.


    Now consider the equivalent Elm code, with the same missing comma:
rules = [     rule "Do not talk about Sandwich Club.",     rule "Do NOT talk about Sandwich Club."     rule "No eating in the common area." ]  

    The mistake is just as easy to overlook, but harder to fix because this is syntactically valid Elm code—just not the code you intended to write!


    The missing comma means the above code is essentially equivalent to the following:
rules = [     (rule "Do not..."),     (rule "Do NOT..." rule "No eating...") ]  

    Instead of calling rule three times, each time with one argument, here the second call to rule is receiving three arguments—and there is no third call! This means instead of the syntax error JavaScript gave you, you’ll get a seemingly nonsensical error about arguments.


    Now try to make this mistake when writing in a commas-first style:
rules =     [ rule "Do not talk about Sandwich Club."       rule "Do NOT talk about Sandwich Club."     , rule "No eating in the common area."     ]

     


    This style makes it blindingly obvious that a comma is missing. Now we don’t even need to compile our code to identify the problem!


    It may feel different at first, but the commas-first style gives you one less potential error to worry about once you get used to it.


    
       

    


    Declaring the PhotoGroove module


    By writing module PhotoGroove exposing (..) at the top of our PhotoGroove.elm file, we defined a new module of our own. This means other files in our project can now import this PhotoGroove module just like they would the String or Html modules, for example like so:
import PhotoGroove exposing (main)

    Because we wrote exposing (..) after module PhotoGroove, we are exposing all of our top-level values—namely main and view—for other modules to import.


    If we wanted to expose main but leave view invisible to other modules, we could have written exposing (main) instead. In such a case, another module that imported PhotoGroove would get an error if it tried to access PhotoGroove.view. Only exposed values can be accessed by other modules!


    Installing Package Dependencies


    At this point we have a PhotoGroove.elm file, but it’s missing a dependency. All this fancy rendering logic we’ve been writing comes from the Html module, but unlike the modules we’ve been using so far—like String and List—the Html module does not come preinstalled. To access it, we need to install the package that contains the Html module.


    definition An Elm package is an installable collection of modules.


    The elm-package utility downloads and installs packages when you give it the name of the package you want. Package names consist of a username followed by a / and then the package name; in this case, the package we seek is named elm-lang/html.


    Let’s use elm-package to get some Html going. Run this command in the terminal:
elm-package install elm-lang/html

    You should see something like this:
Some new packages are needed. Here is the upgrade plan.

    (it lists the packages here)
Do you approve of this plan? (Y/n)

    Answer y, and you should shortly see the text Packages configured successfully! You should now have a file called elm-package.json in your current directory, which elm-package will have created for you automatically. We’ll dig into elm-package.json in future chapters.


    Compiling to JavaScript and running the application


    Now we’re ready to compile our Elm code into JavaScript. Run this in the terminal:
elm-make PhotoGroove.elm --output elm.js

    This will compile our PhotoGroove.elm file into a JavaScript file we can give to a browser. (That generated JavaScript file will be called elm.js, because we passed --output elm.js to elm-make.) Now all we need is a HTML file that will load up our compiled elm.js file!


    Make a file called index.html and put the following inside it.


    Listing 2.4 index.html
<!doctype html> <html>     <head>         <style>             body { background-color: rgb(44, 44, 44); color: white; }             img { border: 1px solid white; margin: 5px; }             .large { width: 500px; float: right; }             .selected { margin: 0; border: 6px solid #60b5cc; }             .content { margin: 40px auto; width: 960px; }             #thumbnails { width: 440px; float: left }             h1 { font-family: Verdana; color: #60b5cc; }         </style>     </head>       <body>         <div id="elm-area"></div>       ❶            <script src="elm.js"></script>  ❷          <script>             Elm.PhotoGroove.embed(document.getElementById("elm-area"));  ❸          </script>     </body></html>

    ❶   Our Elm application will render into this div


    ❷   PhotoGroove.elm will get compiled into elm.js


    ❸   The Elm object comes from elm.js


    The markup we put in this file covers things like:


    ·   The standard <!doctype>, <html>, and <body> tags


    ·   Importing the elm.js file we generated with elm-make


    ·   Whatever <head> inclusions we need—styles, metadata, <title>, and so on


    The line Elm.PhotoGroove.embed(document.getElementById("elm-area")); starts our Elm code running in the <div id="elm-area"></div> element we set up earlier in index.html. Let’s break down what happens when our Elm code starts running.


    The Elm Runtime and ‘main’


    When Elm compiles your code into JavaScript, it includes an extra bit of JavaScript known as the Elm Runtime. The Elm Runtime is behind-the-scenes logic that silently handles things like:


    ·   Adding and removing event listeners for any events your code depends on


    ·   Efficiently scheduling tasks like HTTP requests and DOM updates


    ·   Storing and managing application state


    When we called Elm.PhotoGroove.embed, we told the Elm Runtime to open the module called PhotoGroove and use its exposed main value as the entry point for the application. If we did not have a module called PhotoGroove, or if it did not expose a main value, we’d get an error.


    This means when the browser runs our compiled code, view "no model yet" will be the first line of code executed, because that’s what we assigned to main. If we renamed the PhotoGroove module to CubeDraft, we’d have to call Elm.CubeDraft.embed instead, but otherwise everything would still work. If the CubeDraft module did not expose a value called main, however, the application would not start. There’s no renaming main!


    If you open index.html, you should see our application displaying as it does in Figure 2.5.

    Figure 2.5 Rendering the application


    Fantastic! Next we’ll make it interactive.


    2.2   Handling User Input with the Elm Architecture


    So far we haven’t had much data flowing through our application. Okay, really we haven’t had any—we just generated some Html and rendered it. That will soon change, as we’re about to start handling user input! This brings us to a common question: how will we keep data flow manageable as our code scales?


    JavaScript offers a staggering selection of data flow architectures to choose from, but Elm has just one. It’s called the Elm Architecture, and the Elm Runtime is optimized for applications that follow it. We’ll learn about the Elm Architecture as we add interactivity to Photo Groove.


    Figure 2.6 shows a preview of the architecture we’ll be building toward in this chapter. Don’t worry if this does not make sense yet! We will get there, one step at a time.

    Figure 2.6 The Elm Runtime uses the Elm Architecture to manage data flow


    Let’s begin where data flow naturally begins: with application state as a whole.


    2.2.1   Representing Application State with a Model


    Back in the Wild West days of the Web, it was common to store application state primarily in the DOM itself. Is that menu expanded or collapsed? Check whether one of its DOM nodes has class="expanded" or class="collapsed". Need to know what value a user has selected in a dropdown? Query it out of the DOM at the last possible instant.


    This approach turned out not to scale very well, especially as applications grew more complex and unit testing became increasingly important. Today it’s common practice to store application state completely outside the DOM, and to propagate changes from that independent state over to the DOM as necessary. This is what we do in the Elm Architecture.


    Declaring a Model


    We’re going to store our application state separately from the DOM, and will refer to that state as our model.


    definition A model stores the current state of an application. Any value necessary to render a user interface should be stored in its model.


    Remember how earlier we wrote this code?
main =     view "no model yet"

    Let’s replace this code with the contents of Listing 2.5.


    Listing 2.5 Adding a Model
initialModel =     [ { url = "1.jpeg" }  ❶      , { url = "2.jpeg" }  ❶      , { url = "3.jpeg" }  ❶      ]     main =    view initialModel     ❷ 

    ❶   We’ll add more fields beyond url later


    ❷   Pass our new initialModel record to view


    Excellent! Now we have an initial model to work with. So far it contains a list of photos, each of which is represented by a record containing a url string.


    Writing a view function


    Next we’ll render a thumbnail for each photo in our list. A typical Elm application does this through a view function, which describes how the DOM should look based on its arguments.


    At the top of a typical Elm application is a single view function, which accepts our current model as an argument and then returns some Html. The Elm Runtime takes the Html returned by this view function and renders it.


    By pure coincidence, we’ve already written just such a view function—it’s the function we had the foresight to name view. Unfortunately, our current view implementation ignores the model argument it receives, which means changing our model won’t result in a visible change to the end user. Let’s fix that! view should base its return value on its model argument.


    It’ll be easier to do this if we first write a separate viewThumbnail function, which renders a single thumbnail as Html. We can set the stage for that design by replacing our view implementation with the following:


    Listing 2.6 Splitting out viewThumbnail
urlPrefix =                                     ❶      "http://elm-in-action.com/"                 ❶      view model =     div [ class "content" ]         [ h1 [] [ text "Photo Groove" ]         , div [ id "thumbnails" ] []         ]     viewThumbnail thumbnail =    img [ src (urlPrefix ++ thumbnail.url) ] []  ❷ 

    ❶   We’ll prepend this to strings like “1.jpeg”


    ❷   Prepend urlPrefix to get a complete URL like “http://elm-in-action.com/1.jpeg”


    Figure 2.7 illustrates how our current model and view connect to the Elm Runtime.

    Figure 2.7 Model and view connecting with the Elm Runtime


    Next, we’ll iterate over our list of photo records and call viewThumbnail on each one, in order to translate it from a dusty old record to a vibrant and inspiring img.


    Fortunately, the List.map function does exactly this!


    List.map


    List.map is another higher-order function similar to the List.filter function we used in Chapter 1. You pass List.map a translation function and a list, and it runs that translation function on each value in the list. Once that’s done, List.map returns a new list containing the translated values.


    Take a look at Figure 2.8 to see List.map do its thing for viewThumbnail.

    Figure 2.8 Using List.map to transform photo records into img nodes


    Since div is just a plain Elm function that accepts two lists as arguments—first a list of attributes, followed by a list of child nodes—we can swap out our entire hardcoded list of child img nodes with a single call to List.map! Let’s go ahead and do that now.
view model =     div [ class "content" ]         [ h1 [] [ text "Photo Groove" ]         , div [ id "thumbnails" ] (List.map viewThumbnail model)         ]   viewThumbnail thumbnail =    img [ src (urlPrefix ++ thumbnail.url) ] []

    If you run elm-make PhotoGroove.elm --output elm.js again to recompile this code, you should see the same result as before. The difference is that now we have a more flexible internal representation, setting us up to add interactivity in a way that was impossible before we connected model and view.


    Expanding the Model


    Now let’s add a feature: when the user clicks on a thumbnail, it will become selected—indicated by a blue border surrounding it—and we’ll display a larger version of it beside the thumbnails.


    To do this, we first need to store which thumbnail is selected. That means we’ll want to convert our model from a list to a record, so we can store both the list of photos and the current selectedUrl value at the same time.


    Listing 2.7 Converting the model to a record
initialModel =     { photos =         [ { url = "1.jpeg" }         , { url = "2.jpeg" }         , { url = "3.jpeg" }         ]     , selectedUrl = "1.jpeg"  ❶     }

    ❶   Select the first photo by default


    Next let’s update viewThumbnail to display the blue border for the selected thumbnail.


    That’s easier said than done! Being a lowly helper function, viewThumbnail has no way to access the model—so it can’t know the current value of selectedUrl. But without knowing which thumbnail is selected, how can it possibly know whether to return a selected or unselected img?


    It can’t! We’ll have to pass that information along from view to viewThumbnail.


    Let’s rectify this situation by passing selectedUrl into viewThumbnail as an additional argument. Armed with that knowledge, it can situationally return an img with the "selected" class—which our CSS has already styled to display with a blue border—if the url of the given thumbnail matches selectedUrl.
viewThumbnail selectedUrl thumbnail =     if selectedUrl == thumbnail.url then         img             [ src (urlPrefix ++ thumbnail.url)             , class "selected"             ]             []     else         img             [ src (urlPrefix ++ thumbnail.url) ]            []

    Comparing our then and else cases, we see quite a bit of code duplication. The only thing different about them is whether class "selected" is present. Can we trim down this code?


    Absolutely! We can use the Html.classList function. It builds a class attribute using a list of tuples, with each tuple containing first the desired class name, and second a boolean for whether to include the class included in the final class string.


    Let’s refactor our above code to the following, which accomplishes the same thing:
viewThumbnail selectedUrl thumbnail =     img         [ src (urlPrefix ++ thumbnail.url)         , classList [ ( "selected", selectedUrl == thumbnail.url ) ]         ]        []

    Now all that remains is to pass in selectedUrl, which we can do with an anonymous function. While we’re at it, let’s also add another img to display a larger version of the selected photo.


    Listing 2.8 Rendering Selected Thumbnail via anonymous function
view model =     div [ class "content" ]         [ h1 [] [ text "Photo Groove" ]         , div [ id "thumbnails" ]             (List.map (\photo -> viewThumbnail model.selectedUrl photo)                 model.photos             )         , img    ❶              [ class "large"             , src (urlPrefix ++ "large/" ++ model.selectedUrl)             ]             []        ]

    ❶   Display a larger version of the selected photo


    If you recompile with the same elm-make command as before, the result should now look like Figure 2.9.

    Figure 2.9 Rendering the selected thumbnail alongside a larger version.


    Looking good!


    Replacing Anonymous Functions with Partial Application


    Although the way we’ve written this works, it’s not quite idiomatic Elm code. The idiomatic style would be to remove the anonymous function like so:
Before: List.map (\photo -> viewThumbnail model.selectedUrl photo) model.photos After:  List.map           (viewThumbnail model.selectedUrl)       model.photos

    Whoa! Does the revised version still work? Do these two lines somehow do the same thing?


    It totally does, and they totally do! This is because calling viewThumbnail without passing all of its arguments is an example of partially applying a function.


    Definition Partially applying a function means providing one or more of its arguments, and getting back a new function which accepts the remaining arguments and finishes the job.


    When we called viewThumbnail model.selectedUrl photo, we provided viewThumbnail with both of the arguments it needed to return some Html. If we call it without that second photo argument, what we get back is not Html, but rather a function—specifically a function that accepts the missing photo argument and then returns some Html.


    Let’s think about how this would look in JavaScript, where functions aren’t set up to support partial application by default. If we’d written viewThumbnail in JavaScript, and wanted it to support partial application, it would have to look like this:
function viewThumbnail(selectedUrl) {     return function(thumbnail) {         if (selectedUrl === thumbnail.url) {             // Render a selected thumbnail here         } else {             // Render a non-selected thumbnail here         }     };}

    Functions that can be partially applied, such as the one in this JavaScript code, are known as curried functions.


    Definition A curried function is a function that can be partially applied.


    All Elm functions are curried. That’s why when we call (viewThumbnail model.selectedUrl) we end up partially applying viewThumbnail, not getting an undefined argument or an error.


    In contrast, JavaScript functions are not curried by default. They are instead tupled, which is to say they expect a complete “tuple” of arguments. (In this case, “tuple” refers to “a fixed-length sequence of elements,” not specifically one of Elm’s Tuples.)


    Elm and JavaScript both support either curried or tupled functions. The difference is which they choose as the default:


    ·   In JavaScript, functions are tupled by default. If you’d like them to support partial application, you can first curry them by hand—like we did in our JavaScript viewThumbnail implementation above.


    ·   In Elm, functions are curried by default. If you’d like to partially apply them...go right ahead! They’re already set up for it. If you’d like a tupled function, write a curried function that accepts a single Tuple as its argument, then destructure that tuple.


    Table 2.1 shows how to define and use both curried and tupled functions in either language.


    Table 2.1 Curried functions and Tupled functions in Elm and JavaScript


    
      
        	
           

        

        	
          
            Elm

          

        

        	
          
            JavaScript

          

        
      


      
        	
          Curried Function

        

        	splitA separator str =     String.split separator str  

        	function splitA(sep) {   return function(str) {     return str.split(sep);   } }
      


      
        	
          Tupled Function

        

        	splitB ( separator, str ) =     String.split separator str

        	function splitB(sep, str) {   return str.split(sep); }
      


      
        	
          Total Application

        

        	splitB ( "-", "867-5309" )

        	      splitB("-", "867-5309")
      


      
        	
          Total Application

        

        	splitA   "-"  "867-5309"

        	      splitA("-")("867-5309")
      


      
        	
          Partial Application

        

        	splitA   "-"

        	      splitA("-")
      

    


    We can use our newfound powers of partial application to make view more concise! We now know we can replace our anonymous function with a partial application of viewThumbnail.
Before: List.map (\photo -> viewThumbnail model.selectedUrl photo) model.photos After:  List.map           (viewThumbnail model.selectedUrl)       model.photos

    tip In Elm, an anonymous function like (\foo -> bar baz foo) can always be rewritten as (bar baz) by itself. Keep an eye out for this pattern; it comes up surprisingly often.


    Here’s how our updated view function should look.


    Listing 2.9 Rendering Selected Thumbnail via partial application
view model =     div [ class "content" ]         [ h1 [] [ text "Photo Groove" ]         , div [ id "thumbnails" ]             (List.map (viewThumbnail model.selectedUrl) model.photos)  ❶          , img             [ class "large"             , src (urlPrefix ++ "large/" ++ model.selectedUrl)             ]             []        ]

    ❶   Partially apply viewThumbnail with model.selectedUrl


    Since all Elm functions are curried, it’s common to give a helper function more information by adding an argument to the front of its arguments list.


    For example, when viewThumbnail needed access to selectedUrl, we made this change:
Before: List.map  viewThumbnail                    model.photos After:  List.map (viewThumbnail model.selectedUrl) model.photos

    Because we added the new selectedUrl argument to the front, we could pass it in using partial application instead of an anonymous function. This is a common technique in Elm code!


    tip Since operators are functions, you can partially apply them too! List.map ((*) 2) [ 1, 2, 3 ] evaluates to [ 2, 4, 6 ].


    Incidentally, currying is named after acclaimed logician Haskell Brooks Curry. The Haskell programming language is also named after his first name, and whether the Brooks Brothers clothing company is named after his middle name is left as an exercise to the reader.


    2.2.2   Handling Events with Messages and Updates


    Now that we can properly render which thumbnail is selected, we need to change the appropriate part of the model whenever the user clicks a different thumbnail.


    If we were writing JavaScript, we might implement this logic by attaching an event listener to each thumbnail like so:
thumbnail.addEventListener("click", function() { model.selectedUrl = url; });

    Elm wires up event handlers a bit differently. Similarly to how we wrote a view function that used Virtual DOM nodes to describe our desired page structure, we’re now going to write an update function that uses messages to describe our desired model.


    definition A message is a value used to pass information from one part of the system to another.


    When the user clicks a thumbnail, a message will be sent to an update function as follows:

    Figure 2.10 Handling the event when a user clicks a thumbnail


    The format of our message is entirely up to us. We could represent it as a string, or a list, or a number, or anything else we please. Here’s a message implemented as a record:
{ operation = "SELECT_PHOTO", data = "2.jpeg" }

    This record is a message which conveys the following information:


    “We should update our model to set 2.jpeg as the selectedUrl.”


    The update function receives this message and does the following:


    1. Looks at the message it received.


    2. Looks at our current model.


    3. Uses these two values to determine a new model, then returns it.


    We can implement our “select photo” logic by adding this update function right above main:
update msg model =     if msg.operation == "SELECT_PHOTO" then         { model | selectedUrl = msg.data }     else        model

    Notice how if we receive an unrecognized message, we return the original model unchanged. This is important! Whatever else happens, the update function must always return a new model, even if it happens to be the same as the old model.


    Adding onClick to viewThumbnail


    We can request that a SELECT_PHOTO message be sent to update whenever the user clicks a thumbnail, by adding an onClick attribute to viewThumbnail:
viewThumbnail selectedUrl thumbnail =     img         [ src (urlPrefix ++ thumbnail.url)         , classList [ ( "selected", selectedUrl == thumbnail.url ) ]         , onClick { operation = "SELECT_PHOTO", data = thumbnail.url }         ]        []

    The Elm Runtime takes care of managing event listeners behind the scenes, so this one-line addition is the only change we need to make to our view. We’re ready to see this in action!


    The Model-View-Update Loop


    To wire our Elm application together, we’re going to change main = view model to the following, which incorporates update according to how we’ve set things up so far.
main =     Html.beginnerProgram         { model = initialModel         , view = view         , update = update        }

    The Html.beginnerProgram function takes a record with three fields:


    ·   model - A value that can be anything you please.


    ·   view - A function that takes a model and returns a Html node.


    ·   update - A function that takes a message and a model, and returns a new model.


    It uses these arguments to return a description of a program, which the Elm Runtime sets in motion when the application starts up. Before we got beginnerProgram involved, main could only render static views. beginnerProgram lets us specify how we want to react to user input!


    Figure 2.11 demonstrates how data flows through our revised application.

    Figure 2.11 Data flowing from the start of the program through the Model-View-Update loop


    Notice that view builds fresh Html values after every update. That might sound like a lot of performance overhead, but in practice, it’s almost always a performance benefit!


    This is because Elm doesn’t actually recreate the entire DOM structure of the page every time. Instead, it compares the Html it got this time to the Html it got last time and updates only the parts of the page that are different between the two requested representations.


    This approach to “Virtual DOM” rendering, popularized by the JavaScript library React, has several benefits over manually altering individual parts of the DOM:


    ·   Updates are automatically batched to avoid expensive repaints and layout reflows


    ·   It becomes far less likely that application state will get out of sync with the page


    ·   Replaying application state changes effectively replays user interface changes


    Since onClick lives in the Html.Events module, we’ll need to import it:
import Html.Events exposing (onClick)

    And with that final touch...it’s alive! You’ve now written an interactive Elm application!


    The complete PhotoGroove.elm file should look like this:


    Listing 2.10 PhotoGroove.elm with complete Model-View-Update in place
module PhotoGroove exposing (..)   import Html exposing (..) import Html.Attributes exposing (..) import Html.Events exposing (onClick)     urlPrefix =     "http://elm-in-action.com/"     view model =     div [ class "content" ]         [ h1 [] [ text "Photo Groove" ]         , div [ id "thumbnails" ]             (List.map (viewThumbnail model.selectedUrl) model.photos)         , img             [ class "large"             , src (urlPrefix ++ "large/" ++ model.selectedUrl)             ]             []         ]     viewThumbnail selectedUrl thumbnail =     img         [ src (urlPrefix ++ thumbnail.url)         , classList [ ( "selected", selectedUrl == thumbnail.url ) ]         , onClick { operation = "SELECT_PHOTO", data = thumbnail.url }         ]         []     initialModel =     { photos =         [ { url = "1.jpeg" }         , { url = "2.jpeg" }         , { url = "3.jpeg" }         ]     , selectedUrl = "1.jpeg"     }     update msg model =     if msg.operation == "SELECT_PHOTO" then         { model | selectedUrl = msg.data }     else         model     main =     Html.beginnerProgram         { model = initialModel         , view = view         , update = update        }

    Let’s compile it once more with elm-make PhotoGroove.elm --output elm.js. If you open index.html, you should now be able to click a thumbnail to select it. Huzzah!

    Figure 2.12 Our final Photo Groove application


    At this point we’ve also worked our way through the complete Elm Architecture diagram from the beginning of the chapter. Figure 2.13 shows where things ended up.

    Figure 2.13 Our final Elm Architecture setup


    Congratulations on a job well done!


    2.3   Summary


    In this chapter we learned about three ways to handle interactions, each of which differs from the JavaScript way of handling the same. Table 2.2 summarizes these differences.


    Table 2.2 Handling interactions in JavaScript compared to Elm


    
      
        	
          
            Interaction

          

        

        	
          
            JavaScript approach

          

        

        	
          
            Elm approach

          

        
      


      
        	
          Changing the DOM

        

        	
          Directly alter DOM nodes

        

        	
          Return some Html from a view function

        
      


      
        	
          Reacting to user input

        

        	
          Attach a listener to an element

        

        	
          Specify a message to send to update

        
      


      
        	
          Changing application state

        

        	
          Alter an object in place

        

        	
          Return a new model in update

        
      

    


    We covered many other concepts in the course of building our first Elm application, including:


    ·   A model represents our application state


    ·   A view function takes a model and returns a list of Html nodes


    ·   User events such as clicks get translated into message values


    ·   Messages get run through the update function to produce a new model


    ·   After an update, the new model is sent to the view function to determine the new DOM


    ·   Html.beginnerProgram wires together model, view, and update


    ·   List.map is a higher-order function that translates one list into another


    ·   All Elm functions are curried, which means they can be partially applied


    In the next chapter we’ll get into ways to improve upon the application we’ve made so far, both adding features and refactoring to make it easier to maintain.


    Onward!

  


  
    3    Compiler as Assistant


    This chapter covers


    ·   Documenting guarantees with type annotations


    ·   Implementing multi-way conditionals with case-expressions


    ·   Storing flexible data with union types


    ·   Using Array and Maybe for positional element access


    ·   Generating random numbers using commands


    In Chapter 2 we built our first Elm application. It doesn’t do much yet, but it has potential! So far it displays thumbnails of three photos, and lets users click one to view a larger version.


    We showed it to our manager, who was thrilled with what we’d made. “Wow, this is looking incredible. The part where you click the thumbnail and it shows the bigger version? Just brilliant. I’m going to get some more team members working with you on this.”


    Nice! Quite a vote of confidence. Sure, we have no tests or documentation to help get these new teammates up to speed, but there’s no time like the present to clean up our code!


    Our manager also had a couple of feature requests. “Let’s give users the ability to choose between viewing small, medium, or large thumbnails. Also, for this next version I want to kick the fun factor into overdrive. Let’s add a button that says Surprise Me! and when you click it, it selects one of the photos...at random.”


    We resist the urge to tell our manager “whoa there, that might be too much fun” and instead review the tasks we’ve just received:


    1. Improve code quality to help new team members get up to speed


    2. Let users choose between small, medium, and large thumbnails


    3. Add a Surprise Me! button that randomly selects a photo


    Improving code quality while adding new features is often a tall order. Even more so because these particular features involve aspects of Elm we have not yet encountered.


    Fortunately, we have an assistant to help us out: Elm’s compiler. In this chapter we’ll learn how it can help us improve documentation, refactor without introducing regressions, and accomplish both while introducing new features. Let’s get to it!


    3.1   Documenting Guarantees with Type Annotations


    One of the quickest ways we can make our code nicer for incoming teammates is to add some comments that document what our code does. Although comments are simple and flexible, they’re also notoriously unreliable. Sometimes they’re written inaccurately. Other times they start out accurate but become inaccurate as the code base changes out from under them.


    “Code never lies. Comments sometimes do.”


    
      —Ron Jeffries             

    


    Elm gives us access to a genuinely trustworthy form of documentation: type annotations.


    3.1.1   Adding Optional Type Annotations


    In Chapter 2 we assigned the constant urlPrefix to "http://elm-in-action.com". Let’s edit our PhotoGroove.elm file to add a type annotation on top of that urlPrefix assignment:

    Figure 3.1 Adding a type annotation to urlPrefix 


    This annotation is saying “urlPrefix is a String,” and it’s not kidding around with that. Elm’s compiler will check our entire code base to verify this claim. If finds even one incident where urlPrefix is not used as a String, it will give us an error at compile time. This means if our code compiles, we can be certain urlPrefix is a String absolutely everywhere it’s used!


    Tip Searching for "urlPrefix :" in your editor is now a quick way to jump to urlPrefix’s definition.


    This guarantee is even more useful for annotated functions. When teammates are new to the codebase, being able to reliably tell at a glance what arguments a function takes, and what it returns, can be an incredible time-saver.


    Annotating Functions


    We can annotate functions by writing -> between their arguments and return values.
isEmpty : String -> Bool isEmpty str = str == ""

    This annotation says “isEmpty takes a String and returns a Bool.” Bool refers to one of those True or False boolean values that we can use as the condition of an if-expression.


    Annotating Records


    Record annotations use : instead of = but otherwise look about the same as record values.
  selectPhoto : { operation : String, data : String } selectPhoto = { operation = "SELECT_PHOTO", data = "1.jpeg" } 

    This annotation says “selectPhoto is a record with an operation field and a data field, and each of those fields is a String.” We can add a similar annotation to our initialModel:
  initialModel =     { photos = [ { url = "1.jpeg" }, { url = "2.jpeg" }, { url = "3.jpeg" } ]     , selectedUrl = "1.jpeg"     } 

    Ah, but this record has a list in it! How do we annotate lists?


    Annotating Lists


    In Chapter 1 we learned that all elements in an Elm list must have a consistent type. This is reflected in their type annotations. For example, Elm represents a “list of strings” with the type annotation List String, and a “list of booleans” with the type annotation List Bool.


    Let’s play around with some different lists to see their type annotations in elm-repl:


    Listing 3.1 Annotating Lists
> [ "funk", "soul" ] ["funk","soul"] : List String                              ❶    > [ [ "don’t", "forget" ], [ "about", "Dre" ] ] [["don’t","forget"],["about","Dre"]] : List (List String)  ❷    > [ { url = "1.jpeg" }, { url = "2.jpeg" } ]               ❸ [{ url = "1.jpeg" },{ url = "2.jpeg" }] : List { url : String }

    ❶   “a list of strings”


    ❷   “a list of lists of strings”


    ❸   “a list of records”


    Notice anything about the structure of that last example? It looks just like the list of photos in our model! This means we can use it to write an annotation for our model. Let’s do that:
initialModel : { photos : List { url : String }, selectedUrl : String } initialModel =    ...

    Model annotations are among the most helpful forms of documentation for new teammates, because they concisely (and reliably!) describe the structure of our entire application state.


    3.1.2   Annotating Functions with Type Variables


    Functions, records, and lists served us well in Chapter 2, but implementing the Surprise Me! button will involve a collection we haven’t used before.


    lists and arrays


    We mentioned in Chapter 1 that Elm supports arrays as well as lists. Both are sequential collections of varying length, whose elements share a consistent type. However, there are a few differences between arrays and lists:


    ·   Lists can be created with square bracket literals, whereas arrays have no literal syntax in Elm. We always create arrays by calling functions.


    ·   Lists have better performance in typical Elm use cases. This makes them the standard choice for representing a sequence of values, in contrast to JavaScript where arrays are standard. In Chapter 10 we’ll see why lists are a better standard choice in Elm.


    ·   Arrays are better for arbitrary positional access. That means it’s quick and easy to say “give me the 3rd element in this array.” Lists have no first-class way to do this.


    Randomly choosing a photo involves arbitrary positional access. As such, we’ve found ourselves in exactly the sort of situation where an array will serve us better than a list!


    Array.fromList


    The most common way to obtain an array is by calling Array.fromList on an existing list. Table 3.1 shows the results of a few different calls to Array.fromList.


    Table 3.1 Translating a list into an array using Array.fromList


    
      
        	
          
            Expression              

          

        

        	
          
            Result

          

        
      


      
        	
          Array.fromList [ 2, 4, 6 ]

        

        	
          Array containing 3 elements: 2, 4, and 6

        
      


      
        	
          Array.fromList [ "foo" ]

        

        	
          Array containing 1 element: "foo"

        
      


      
        	
          Array.fromList []

        

        	
          Array containing 0 elements

        
      

    


    Type Variables


    How would we write a type annotation for Array.fromList? It’s bit different from the functions we’ve annotated so far, as it returns a different type depending on what you pass it.


    Table 3.2 shows some examples of the different types of values it can return.


    Table 3.2: Array.fromList’s return type depends on its input


    
      
        	
          
            Given one of these

          

        

        	
          
            Array.fromList returns one of these

          

        
      


      
        	
          List Float

        

        	
          Array Float

        
      


      
        	
          List String

        

        	
          Array String

        
      


      
        	
          List { url : String }

        

        	
          Array { url : String }

        
      

    


    Notice the pattern? Whatever type of list we pass in, that’s what type of array we get out.


    We can capture this pattern in a type annotate like so:
fromList : List elementType -> Array elementType

    In this annotation, elementType is a type variable.


    Definition A type variable represents more than one possible type. Type variables have lowercase names, making them easy to tell them apart from concrete types like String, which are always capitalized.


    When you call Array.fromList passing some list, the type variable elementType gets replaced by that list’s element type. Table 3.3 shows this process in action.


    Table 3.3 Replacing Array.fromList’s type variable


    
      
        	
          
            When we pass this in...

          

        

        	
          
            ...elementType becomes...

          

        

        	
          
            ...and we get this back

          

        
      


      
        	
          List Float

        

        	
          Float

        

        	
          Array Float

        
      


      
        	
          List String

        

        	
          String

        

        	
          Array String

        
      


      
        	
          List { url : String }

        

        	
          { url : String }

        

        	
          Array { url : String }

        
      

    


    choosing names for type variables


    When annotating fromList, we can choose a different name for its type variable besides elementType. Since a type variable essentially serves as a placeholder for a concrete type like String or List Int, it can have just about any name, so long as it’s lowercase and we use it consistently. For example, we could annotate Array.fromList in any of the following ways:


    ·   List elementType -> Array elementType


    ·   List foo         -> Array foo


    ·   List a           -> Array a


    In contrast, none of the following would work as annotations for Array.fromList:


    ·   List elementType -> Array blah


    ·   List foo         -> Array bar


    ·   List a           -> Array b


    These three annotations are saying “Array.fromList takes a list containing elements of one type and returns an array that potentially contains elements of another type.”


    That just ain’t so! Array.fromList can only return an array whose elements have the same type as the list it receives. Its type annotation must reflect that, by having the same type variable for both the List it accepts and the Array it returns.


    note Elm has three type variable names that have special meanings—number, appendable, and comparable—which we’ll dive into later. For now, avoid choosing any of those as your type variable names!


    By the way, you’ll often encounter type variables in documentation with single-letter names like a, b, c, and so on. (For example, the official documentation for Array.fromList annotates it as List a -> Array a.) You are by no means obliged to do the same! Feel free to use the most self-descriptive names you can think of for your own type variables.


    Creating an array of photos


    Since we want to access our photos by position, we’ll be using Array.fromList on our model.photos list to translate it into an array.


    Let’s add this code below our initialModel definition in PhotoGroove.elm:
photoArray : Array { url : String } photoArray =    Array.fromList initialModel.photos

    If we tried to compile this right now, we’d get an error because we have not imported the Array module! Let’s add it to our imports at the top of the file:
import Array exposing (Array)

    Nice. Let’s verify that our code still compiles by running elm-make on it before proceeding.


    
      Exposing Imported Types

    


    Notice how we wrote exposing (Array) there? If we’d written import Array without the exposing (Array) part, we’d need to refer to the Array type in a qualified style, like so:
  photoArray : Array.Array { url : String }  

    Array.Array refers to “the Array type from the Array module” just like how Array.fromList refers to “the fromList function from the Array module.” Kinda verbose, right?


    By writing exposing (Array) after import Array, we get to skip the “Array.” prefix and use the more concise annotation:
  photoArray : Array { url : String }  

    Much better!


    
       

    


    3.1.3   Reusing Annotations with Type Aliases


    Now that we’ve created photoArray, we can see that our annotations for initialModel and photoArray have a bit of code duplication going on: they both include { url : String }.
initialModel : { photos : List { url : String }, selectedUrl : String } photoArray : Array { url : String }

    We can replace this duplication with shared code using a type alias.


    Definition A type alias assigns a name to a type. Anywhere you would refer to that type, you can substitute this name instead.


    Let’s create a type alias called Photo, and then use that in place of { url : String }.


    Listing 3.2 Creating a type alias for Photo
type alias Photo =      ❶      { url : String }    ❶      initialModel : { photos : List Photo, selectedUrl : String } initialModel =    ...   photoArray : Array Photo photoArray =    Array.fromList initialModel.photos

    ❶   “whenever I say Photo, I mean { url : String }”


    Not only does this make our initialModel annotation more concise, it also makes it easier to maintain! Now if we add a new field to our Photo record, we can change the type alias in one place instead of having to hunt down several individual annotations.


    annotating initialmodel using a type alias


    Since initialModel, view, and update all involve our model record, let’s add a type alias for Model as well, and then revise initialModel’s type annotation to use it:
type alias Model =     { photos : List Photo     , selectedUrl : String     }    initialModel : Model

    That takes care of initialModel. What about view?


    Html’s type variable


    We know view takes a Model and returns Html, but we can’t just write “view returns Html” and call it a day. This is because, just like List and Array, the Html type has a type variable!


    Html’s type variable reflects the type of message it sends to update in response to events from handlers like onClick.


    Table 3.4 Comparing type variables for List and Html


    
      
        	
          
            Value

          

        

        	
          
            Type

          

        

        	
          
            Description

          

        
      


      
        	
          [ "foo" ]

        

        	
          List String

        

        	
          List of String elements

        
      


      
        	
          [ 3.14 ]

        

        	
          List Float

        

        	
          List of Float elements

        
      


      
        	
          div [ onClick "foo" ] []

        

        	
          Html String

        

        	
          Html producing String messages

        
      


      
        	
          div [ onClick 3.14 ] []

        

        	
          Html Float

        

        	
          Html producing Float messages

        
      


      
        	
          div [ onClick { x = 3.3 } ] []

        

        	
          Html { x : Float }

        

        	
          Html producing { x : Float } messages

        
      

    


    Since our onClick handler produces messages in the form of records that have an operation string and a data string, our view function’s return type is:
Html { operation : String, data : String }

    That’s pretty verbose! Even though it won’t remove any code duplication yet, it’s perfectly fine to introduce a type alias to make view’s annotation more concise. Let’s add one above view:
type alias Msg =     { operation : String, data : String }    view : Model -> Html Msg

    Excellent! Next we’ll look into annotating view’s helper function, viewThumbnail.


    3.1.4   Annotating Longer Functions


    We’ve now seen a few type annotations for functions that take a single argument, but none for functions that take multiple arguments. Our viewThumbnail function will be the first.


    To learn how to annotate a multi-argument function, let’s play around with one that has a simpler return type: the humble padLeft function. padLeft makes sure strings meet a certain minimum length. You give it a minimum length, a “filler character,” and a string. If the string is not at least the given length, padLeft adds filler characters to its left until the string reaches that length.


    We can try it out in elm-repl, since padLeft is included in Elm’s core String module:
> String.padLeft 9 '.' "not!" ".....not!" : String   > String.padLeft 2 '.' "not!""not!" : String

    We can see that String.padLeft takes three arguments—an Int, a Char, and a String—and then returns another String. How can we annotate a function like that?


    Believe it or not, we’ve already seen the answer! It’s just one of those answers that likes to hide in plain sight.


    Annotating a partially applied function


    Let’s think back to Chapter 2, where we learned that all functions in Elm are curried. That means they all support partial application—the practice of calling a function without passing all the arguments it requires. If we call padLeft passing only one of its three arguments, the result will be a function that takes the remaining arguments and “finishes the job.”


    Here’s a partial application of padLeft. What would the annotation for this be?
padNine = String.padLeft 9

    padLeft 9 is a partial application, so we know padNine must be a function. That’s a start!


    We also know that padLeft takes a Char as its next argument after the 9. That gives us enough information to rough out some pseudocode:
padNine : Char -> (a function of some sort) padNine = String.padLeft 9

    Now let’s suppose we gave this padNine function the Char it wants. Passing a '.' character to padNine would be the the same as passing 9 and then '.' to the original padLeft, like so:
padNineDots = String.padLeft 9 '.'

    How would we annotate padNineDots? Since it gives padLeft two of three arguments, that means only one more argument is needed to finish the job. That last argument is a String, so padNineDots must take a String and return a String. We know how to write that one!
padNineDots : String -> String padNineDots = String.padLeft 9 '.'

    We now know that:


    ·   padNine takes a Char and returns a function (of some type)


    ·   If you pass padNine a Char, it returns a String -> String function


    Putting those two together, our mystery function must be a String -> String!


    That tells us we can annotate padNine like so:
padNine : Char -> (String -> String) padNine = String.padLeft 9

    Now let’s unwind our original partial application. If we cease partially applying the 9, we can follow the pattern we saw here to arrive at a valid annotation for the original padLeft itself:
padLeft : Int -> (Char -> (String -> String))

    This is a perfectly accurate annotation, but it’s a bit heavy on the parentheses, yeah? Fortunately, we can be more concise. Elm’s syntax lets us omit the parentheses here, so the following two annotations are equivalent:
padNine : Int -> (Char -> (String -> String)) padNine : Int ->  Char ->  String -> String

    Either style works just as well as the other, but the latter is considered best practice. You can see elm-repl use this annotation style in Listing 3.3, as we partially apply String.padLeft step by step, until it has no more arguments to accept and finally returns a String.


    Listing 3.3 Multi-Argument function annotations
> String.padLeft <function:padLeft> : Int -> Char -> String -> String   > String.padLeft 9 <function> : Char -> String -> String   > String.padLeft 9 '.' <function> : String -> String   > String.padLeft 9 '.' "not!"".....not!" : String

    Notice how each time we partially applied padLeft, our annotation got shorter:


    
      
        	
          
            Function

          

        

        	
          
            Type Annotation

          

        
      


      
        	
          String.padLeft

        

        	
          Int -> Char -> String -> String

        
      


      
        	
          String.padLeft 9

        

        	
          Char -> String -> String

        
      


      
        	
          String.padLeft 9 '.'

        

        	
          String -> String

        
      


      
        	
          String.padLeft 9 '.' "not!"

        

        	
          String

        
      

    


    If you think about it, this means technically every Elm function takes only one argument.


    After all, any function that appears to take multiple arguments is ultimately calling out to single-argument functions behind the scenes. The fact that Elm lets you omit the parentheses for these nested calls is just a syntactic convenience.


    TIP You can now tell your friends that in Chapter 2 of this book, you wrote an entire working Elm application where every function took only a single argument. This sounds like some really hardcore programming unless your friends are familiar with currying!


    Annotating viewThumbnail


    Armed with this knowledge, we can follow this pattern to annotate viewThumbnail like so:
viewThumbnail : String -> Photo -> Html Msg viewThumbnail selectedUrl thumbnail =

    Splendid! Our code is getting easier and easier for teammates to pick up at a glance.


    3.2   Case-expressions and Union Types


    Now that our documentation situation is looking better, let’s shift gears to work on those two new features: the Surprise Me! button and the thumbnail size chooser.


    We’ll introduce these iteratively, first adding the visual elements with only a token level of interactivity, and then circling back to make them work properly afterwards.


    3.2.1   Using case-expressions


    Let’s start by adding the Surprise Me! button to our view, right above the thumbnails div.
, button     [ onClick { operation = "SURPRISE_ME", data = "" } ]     [ text "Surprise Me!" ], div [ id "thumbnails" ]

    Next let’s revise our update function to add a quick else if branch for our new SURPRISE_ME operation, and a shiny new type annotation while we’re at it.
update : Msg -> Model -> Model update msg model =     if msg.operation == "SELECT_PHOTO" then         { model | selectedUrl = msg.data }     else if msg.operation == "SURPRISE_ME" then         { model | selectedUrl = "2.jpeg" }     else        model

    This implementation always selects the second photo. That’s not much of a surprise, granted, but we’ll get there. For now, let’s smooth out that code duplication we just introduced!


    refactoring an if-expression into a case-expression


    We now have two conditionals that do nothing more than compare msg.operation to a string. Let’s express this more cleanly by rewriting these if-expressions as a case-expression:

    Figure 3.2 Refactoring an if-expression into a case-expression


    Whereas an if-expression is a two-way conditional, a case-expression is a multi-way conditional. It lets us compare something to a wider range of values than just True or False.


    Just like with a JavaScript switch-statement, we begin a case-expression by providing a value which will be run through a series of comparisons. Here we wrote case msg.operation of because we want to run msg.operation through these comparisons.


    Following the case are a series of branches such as the following:
"SELECT_PHOTO" ->     { model | selectedUrl = msg.data }

    This says that if msg.operation is equal to "SELECT_PHOTO", then the branch after the -> will be evaluated, and the entire case-expression will evaluate to that branch’s result.


    If msg.operation is not equal to "SELECT_PHOTO", then the next branch (in this case the "SURPRISE_ME" -> branch) will be checked in the same way, and so on.


    Just like if-expressions, case-expressions must always evaluate to a single value, meaning exactly one branch must always be chosen for evaluation. When writing if-expressions we have else to ensure we end up with a value no matter what, whereas in case-expressions we can use the default branch of _ -> for the same purpose.


    Note Elm has no equivalent of JavaScript’s break statement. This is no problem, since case-expression branches don’t “fall through” like branches in JavaScript’s switch-statements do.


    Our code is really shaping up! Let’s copy the case-expression from Figure 3.2 into PhotoGroove.elm, then run elm-make PhotoGroove.elm --output elm.js to recompile it.


    Trying out the ‘surprise me!’ button


    Before we open up index.html and view our progress, we have one last change to make. While we’ve been hard at work on code quality and new functionality, one of our new teammates has been busy improving our stylesheets.


    Let’s edit index.html to make use of their work, by changing its <head> to the following:
<head>     <link rel="stylesheet" href="http://elm-in-action.com/styles.css"></head>

    Lovely! Now we can open our revised index.html to see the new button in all its glory.

    Figure 3.3 The new Surprise Me! button


    In addition to styling the Surprise Me! button, our helpful teammate went ahead and added styles for the Thumbnail Size Chooser too.


    How about we introduce some logic to go with those styles?


    3.2.2   Enumerating possibilities with Union Types


    Our second feature lets users choose one of three thumbnail sizes: small, medium, or large.


    The first step toward implementing it will be storing the current chosenSize in our model. In JavaScript we might represent chosenSize as a String—perhaps setting it to either "SMALL", "MEDIUM", or "LARGE"—but in Elm we can do better with union types.


    Definition A union type is a custom type you create by specifying the values it can contain.


    One use for a union type is as an enumeration of values. Figure 3.4 illustrates how we can define a union type to represent the different thumbnail size choices we’ll support:

    Figure 3.4 Defining an enumeration of values with a union type called ThumbnailSize


    This is saying “define a new type called ThumbnailSize with three possible values: Small, Medium, and Large.” Let’s add this code to PhotoGroove.elm, right above type alias Msg.


    We can now define chosenSize in one of the following ways:
chosenSize : ThumbnailSize chosenSize = Small   chosenSize : ThumbnailSize chosenSize = Medium   chosenSize : ThumbnailSizechosenSize = Large

    Notice that in each of these examples, the type of chosenSize is ThumbnailSize. These new values we’ve created—Small, Medium, and Large—aren’t “actually integers under the hood” or “just strings behind the scenes.” We really have scratch-built a brand new type here! Trying to compare a ThumbnailSize to a number, string, or any other type (using == or any other comparison) will yield an error at build time.


    This is different from type alias, which gives a name to an existing type—much like how a variable gives a name to an existing value.


    These values are also unique. The expression Medium == Medium is True, but put absolutely any other value (besides Medium) on either side of that == and the expression will no longer be True. The same can be said of Small == Small and Large == Large.


    Note Boolean values in Elm are capitalized because Bool is a union type: type Bool = True | False


    Let’s add the chosenSize field to our model and initialize it to Medium in initialModel:
type alias Model =     { photos : List Photo     , selectedUrl : String     , chosenSize : ThumbnailSize     }   initialModel : Model initialModel =     { photos = ...     , selectedUrl = "1.jpeg"     , chosenSize = Medium    }

    Great! Next we’ll render some radio buttons to let users change it.


    Rendering a thumbnail size radio button


    We can use our newfound knowledge of union types and case-expressions to write a helper function that takes a ThumbnailSize and renders a radio button which chooses that size. Let’s add the following to PhotoGroove.elm, right below viewThumbnail:


    Listing 3.4 Rendering a thumbnail size radio button
viewSizeChooser : ThumbnailSize -> Html Msg viewSizeChooser size =     label []         [ input [ type_ "radio", name "size" ] []         , text (sizeToString size)         ]     sizeToString : ThumbnailSize -> String sizeToString size =     case size of         Small ->    ❶              "small" ❶            Medium ->   ❷              "med"   ❷            Large ->    ❸             "large" ❸ 

    ❶   evaluates to “small” if size == Small


    ❷   evaluates to “med” if size == Medium


    ❸   evaluates to “large” if size == Large


    Note The underscore in that type_ attribute is very important! As we’ve seen, type is a reserved keyword in Elm (used to define union types), which means it can’t be used for an Html attribute name. The Html.Attributes module names the attribute type_ to work around this.


    Notice anything missing from that case-expression? It has no default branch! We accounted for the cases where size is Small, Medium, and Large...but what if size has some other value, such as null, undefined, or "halibut"?


    Can’t happen! sizeToString’s type annotation tells us that size is guaranteed to be a ThumbnailSize, and a ThumbnailSize can only be Small, Medium, or Large. It can’t be null or undefined because those don’t exist in Elm, and it can’t be "halibut" because that’s a String, not a ThumbnailSize.


    Elm’s compiler knows we’ve covered every possibility here, so it doesn’t require a default branch. In fact, if we tried to add one, the compiler would politely inform us that we’d written unreachable code.


    rendering three Radio buttons


    We can now call this viewSizeChooser function three times, passing Small, Medium, and Large, to render the three radio buttons on the page. Let’s do that right above thumbnails:
, h3 [] [ text "Thumbnail Size:" ] , div [ id "choose-size" ]     [ viewSizeChooser Small, viewSizeChooser Medium, viewSizeChooser Large ] , div [ id "thumbnails" ]    (List.map (viewThumbnail model.selectedUrl) model.photos)

    Once again we see some code duplication: we’re calling the same function (viewSizeChooser) on every element in a list. Any time we’re calling the same function on every element in a list, there’s a good chance we can make our code cleaner with List.map!


    This is absolutely one of those times. Let’s refactor our choose-size element to this:
, div [ id "choose-size" ]     (List.map viewSizeChooser [ Small, Medium, Large ])

    Not only does this refactor reduce duplication, it also makes our code more concise.


    rendering different thumbnail sizes


    Rendering the interface for choosing a thumbnail size is great, but it’s only half the battle! We still need to actually render our thumbnails differently based on what the user has chosen.


    Thanks to our coworker having written a stylesheet for us, we can implement this by adding the class "small", "med", or "large" to our thumbnails container. Since those classes conveniently correspond to the results of our sizeToString function, we can go ahead and replace our current div [ id "thumbnails" ] with the following:
div [ id "thumbnails", class (sizeToString model.chosenSize) ]

    Tip It would be best practice to write a separate sizeToClass function and use it here in place of sizeToString, even if their implementations were identical for the time being. Having separate functions means if someone later changes the text on the radio buttons, our thumbnail classes won’t accidentally break.


    Let’s see how it looks! Since we haven’t implemented the radio button logic yet, try changing the chosenSize value in initialModel to Large and then recompile to see how that affects the way the page looks. Then try recompiling again with Small, and finally back to Medium.

    Figure 3.5 Recompiling with chosenSize = Small, chosenSize = Medium, and chosenSize = Large


    Cool! Now that we have our interface rendering, it’s time to start making the logic work.


    3.2.3   Holding Data in Union Types


    Making the Surprise Me! button change our model’s selectedUrl field to a random photo’s url means we’ll need to access a random photo from our photoArray.


    In JavaScript we might implement reading a photo from an array as follows:
var photos = [ …maybe there are photos here, but maybe not… ]; var selectedId = photos[2].url;  // Select the third photo

    What happens if photos doesn’t have at least three elements? In that case, photos[2] would evaluate to undefined, and photos[2].url would throw an angry runtime exception such as:
Uncaught TypeError: Cannot read property 'url' of undefined

    Elm avoids this runtime exception by handling this situation differently in three ways:


    1. In Elm you call Array.get to get an element in an Array; there is no [2] accessor.


    2. Array.get never returns undefined or null, because Elm has neither of these.


    3. Array.get always returns a container value called a Maybe.


    What’s a Maybe? Glad you asked!


    Definition A Maybe is a container like a List, except it can hold at most one element.


    Maybe is implemented as a union type, but a flavor of union type we haven’t seen yet: one that holds data. Its definition looks like this:
type Maybe value     = Just value    | Nothing

    maybe versus undefined


    The most common use for Maybe is to represent the potential absence of a value. It provides a container-based alternative to JavaScript’s drop-in replacements of null and undefined.


    Figure 3.6 compares Elm’s Array.get and Maybe to JavaScript’s [2] and undefined.

    Figure 3.6 Comparing Elm’s Maybe to JavaScript’s undefined


    Array.get 2 photos expresses the desire to access the third element of an array called photos. However, the photos array may have fewer than three elements in it! If the index 2 is outside the bounds of photos, then Array.get 2 photos will return Nothing.


    On the other hand, if 2 is inside its bounds, then Array.get will return Just c—where c is the element at index 2 in the photos array.


    Maybe’s type variable


    As we’ve seen before with List, Array, and Html, the Maybe type has a type variable!


    We chose to name it value in the definition we gave for Maybe’s union type earlier:
type Maybe value     = Just value    | Nothing

    Similarly to how you can have a List String or a List Photo, you can also have a Maybe String or Maybe Photo. The difference is that whereas List Photo means “a list of photos,” Maybe Photo means “either a Photo or nothing at all.” Put another way, Maybe is a container that can hold at most one element.


    Now that we know Just "1.jpeg" is a Maybe String, what is the type of Just itself?


    "Just" is a function


    Argh, that heading totally gave away the answer. Oh well.


    As the heading suggests, whereas Nothing is a Maybe value, Just is a function that returns a Maybe value. We can confirm this in elm-repl:
> Nothing Nothing : Maybe.Maybe a > Just <function> : a -> Maybe.Maybe a   > Just "dance"Just ("dance") : Maybe.Maybe String

    Note elm-repl calls the type Maybe.Maybe because the Maybe type lives in a module called Maybe. This is similar to Array.Array from earlier, except that unlike the Array module, the Maybe module is imported by default. (The Maybe.Maybe type is not exposed by default.)


    In Chapter 1 we noted that when you write a function by hand, its name must be lowercase...yet here we see that union type constructors like Just are functions with capitalized names. What makes them so special?


    A capital question! The answer is right around the corner.


    Destructuring Union Types


    In Chapter 1 we learned how we can destructure tuples to both extract and name their values.
multiply3d ( x, y, z ) = x * y * z

    As it turns out, we can also destructure union type constructors such as Just in the branches of case-expressions. Let’s use what we’ve learned about Array.get and Maybe to add this getPhotoUrl function right above our update function.


    Listing 3.5 Selecting a photo by index
getPhotoUrl : Int -> String getPhotoUrl index =     case Array.get index photoArray of         Just photo ->    ❶              photo.url           Nothing ->            ""           ❷ 

    ❶   Destructuring `Just` and naming its contained value `photo`


    ❷   Fall back on "" if there was no photo at that index


    Here, Just photo -> is saying two things:


    1. This branch matches a Maybe value that was created using the Just type constructor.


    2. We’re extracting the value that was passed to Just and naming it photo.


    Note This is where the distinction between capitalized and uncapitalized functions matters. By comparing their capitalizations, Elm’s compiler can tell that Just photo -> refers to a type constructor called Just which holds a value we’ve chosen to name photo. If we’d instead written Just True -> the compiler would know we meant “the Just type constructor holding exactly the value True.”


    Notice that this case-expression did not need a _ -> branch! This is because its branches already cover all possible values of Maybe. One branch covers Nothing, and the other covers any values created with Just. Since Nothing and Just are the only ways to obtain a Maybe, these branches have every possibility covered.


    Also notice how because Array.get returns a Maybe, you can’t help but remember to handle the case where index is outside the bounds of photoArray. This API design means that you’ll always be doing the minimal amount of coding necessary to handle error cases. You don’t need to defensively check whether something is null or undefined, because Elm’s compiler has your back in cases where the value you want might not be available.


    3.2.4   Representing Flexible Messages with Union Types


    We’ve made so much progress! We’ve updated our views to render the new interfaces, expanded our model to hold our new chosenSize field, and introduced helper functions for photoArray. Now it’s time to wire all that together with changes to update.


    Let’s start by updating that chosenSize field, which currently gets initialized to Medium and then never changes. We’ll want to update it whenever the user clicks a radio button.


    This might seem like the obvious change to make to viewSizeChooser:
viewSizeChooser : ThumbnailSize -> Html Msg viewSizeChooser size =     ...        onClick { operation = "SET_SIZE", data = size )

    There’s just one problem...this is a type mismatch! Remember our type alias for Msg?
type alias Msg =     { operation : String, data : String }

    This says the data field of a Msg needs to be a String, not a ThumbnailSize! To preserve backwards compatibility with existing code, we need a Msg type that can accommodate both ThumbnailSize and String. One way we can do this is by adding a new field called sizeData:
type alias Msg =     { operation : String, data : String, sizeData : ThumbnailSize }

    This solves one problem while creating another. Now our existing onClick handler will no longer compile, because it’s missing a field. Here’s how we wrote it back in Chapter 2:
onClick { operation = "SELECT_PHOTO", data = thumbnail.url }

    Without specifying a value for that new sizeData field, this record is no longer a valid Msg. We’d need to change it to something like this:
onClick { operation = "SELECT_PHOTO", data = thumbnail.url, sizeData = Small }

    Yuck. Having to set sizeData = Small for an operation that doesn’t care about sizeData? That doesn’t smell right. And are we really planning to add a field to every Msg in our program whenever we need it to support a new data type? This doesn’t seem like it will scale well.


    Is there a better way?


    Implementing Msg as a Union Type


    There totally is! Implementing Msg as a union type will work much better. Here’s the plan:


    1. Replace our type alias Msg declaration with a type Msg declaration.


    2. Revise update to use a case-expression which destructures our new union type.


    3. Change our onClick handler to pass a type constructor instead of a record.


    Listing 3.6 shows the first two changes: replacing type alias Msg with type Msg, and revising update accordingly. Let’s edit PhotoGroove.elm to incorporate these changes.


    Listing 3.6 Implementing Msg as a union type
type Msg                     ❶      = SelectByUrl String     ❷      | SurpriseMe             ❸      | SetSize ThumbnailSize  ❹      update : Msg -> Model -> Model update msg model =     case msg of         SelectByUrl url ->   ❺              { model | selectedUrl = url }           SurpriseMe ->        ❻             { model | selectedUrl = "2.jpeg" }

    ❶   Replaces our earlier declaration of type alias Msg


    ❷   Replaces our "SELECT_PHOTO" operation


    ❸   Replaces our "SURPRISE_ME" operation


    ❹   Our new operation: set thumbnail size


    ❺   Previous condition: msg.operation == "SELECT_PHOTO"


    ❻   Nothing to destructure for SurpriseMe!


    This change means our onClick handlers now expect type constructors instead of records. We’ll need to make this change in view:
Old: onClick { operation = "SURPRISE_ME", data = "" } New: onClick SurpriseMe

    Let’s also make this change in viewThumbnail:
Old: onClick { operation = "SELECT_PHOTO", data = thumbnail.url } New: onClick (SelectByUrl thumbnail.url)

    The missing patterns error


    If we recompile, we’ll see a type of error we haven’t seen before: a missing patterns error.
  -- MISSING PATTERNS -------------------------------- PhotoGroove.elm   This `case` does not have branches for all possibilities. ... You need to account for the following values:       PhotoGroove.SetSize _   Add a branch to cover this pattern! 

    Oops! The compiler noticed we only handled SelectByUrl and SurpriseMe in our update function’s case-expression; we never implemented the logic for SetSize. Fortunately, that deficiency will never reach our end users because the compiler didn’t let us forget.


    Tip Anytime you use the default case _ -> in a case-expression, you cannot get this error. That is not a good thing! The missing patterns error is your friend. When you see it, it’s often saving you from a bug you would have had to hunt down later. Try to use _ -> only as a last resort, so you can benefit from as many missing pattern safeguards as possible.


    Let’s add this to the end of our update function:
SurpriseMe index ->     { model | selectedUrl = "2.jpeg" }   SetSize size ->    { model | chosenSize = size }

    Finally, we need to add a new onClick handler to our radio button in viewSizeChooser:
Old: input [ type_ "radio", name "size" ] [] New: input [ type_ "radio", name "size", onClick (SetSize size) ] []

    If we recompile and click the radio buttons, we can now see the thumbnail sizes change!


    Tip There are at least two ways we could improve user experience here. One way would be making the “medium” option display as selected on page load. Another would be to use a broader event handler than onClick—one that detects whenever the radio state changes, even if it not from a click. Try implementing these improvements sometime for practice!


    Using union types for Msg has two major advantages over records.


    1. The compiler can save us from typos. Before, if we wrote "SELECT_PHOTA" instead of "SELECT_PHOTO", our code would silently fail and we’d have to hunt down the cause. Now, if we write SelectByUrk instead of SelectByUrl, Elm’s compiler will give us an error at build time—including the line number of the typo. No bug hunting necessary!


    2. Each flavor of Msg now holds only the minimum amount of data it needs. SelectByUrl holds only a String, and SetSize holds only a ThumbnailSize. Best of all, SurpriseMe doesn’t need to hold any data at all...and so it doesn’t!


    To sum up, implementing Msg as a union type has made our code more reliable, more concise, and easier to scale. These advantages make union types the typical choice for representing messages in production Elm applications, and they’re what we will use for the rest of the book.


    Now all that remains is to incorporate random number generation into SurpriseMe.


    3.3   Generating Random Numbers with Commands


    Currently our Surprise Me! button always selects the second photo, but we want it to select a random photo instead. Here’s how we’re going to do that:


    1. Generate a random integer between 0 and 2. This will be our index.


    2. Ask photoArray for the photo it’s storing at that index.


    3. Set our model’s selectedUrl to be that photo’s url.


    We’ll start by generating the random integer.


    3.3.1   Describing Random Values with Random.Generator


    In JavaScript, Math.random() is the typical starting point for generating random values. Calling Math.random() gives you a random Float between 0 and 1. To randomly generate an Int or a String, you manually convert that Float into the value you actually want.


    In Elm we start with a Random.Generator, which specifies the type of value we want to randomly generate. For example, the Random.int function takes a lower bound and an upper bound, and returns a Random.Generator that generates a random Int between those bounds.


    Here’s a generator that randomly generates integers between 0 and 2:
randomPhotoPicker : Random.Generator Int randomPhotoPicker =    Random.int 0 2

    Notice how the type annotation says Random.Generator Int. This means “we have a Random.Generator that produces Int values.” If we’d used Random.bool instead of Random.int, it would have returned a Random.Generator Bool instead.


    generating a random photo index


    Since randomPhotoPicker generates random integers bewteen 0 and 2, and photoArray has 3 photos in it, we could use this generator to pick an index within photoArray to select. That would work, but it would be brittle since it relies on photoArray having exactly 3 photos. We can improve it by replacing the hardcoded 2 with (Array.length photoArray - 1).


    Let’s add the following after our getPhotoUrl declaration.
randomPhotoPicker : Random.Generator Int randomPhotoPicker =    Random.int 0 (Array.length photoArray - 1)

    That’ll do nicely. We also need to add the Random module to our imports.
import Random

    Now that we’ve described the random value we want, it’s time to actually generate one!


    3.3.2   Introducing Commands to the Elm Architecture


    If you call JavaScript’s Math.random() five times, you’ll probably get back five different numbers. Elm functions are more consistent. If you call any Elm function five times with the same arguments, you can expect to get the same return value each time. This is no mere guideline, but a language-level guarantee! Knowing that all Elm functions have this useful property makes bugs easier to track down and reproduce.


    Since Elm forbids functions from returning different values when they receive the same arguments, it’s not possible to write an inconsistent function like Math.random as a plain Elm function. Instead, Elm implements random number generation using a command.


    Definition A command is a value that describes an operation for the Elm Runtime to perform. Unlike functions, running the same command multiple times can have different results.


    When the user clicks Surprise Me! we’ll use a command to translate our Random.Generator Int into a randomly-generated Int, which will represent our new selected photo index.


    returning commands from update


    Remember how we specified what onClick should do in terms of a message that got sent to our update function? We didn’t say “add this click event listener to the DOM right away,” we said “I want this Msg to get sent to my update function whenever the user clicks here.”


    Commands work similarly. We don’t say “generate a random number right this instant,” we say “I want a random number, so please generate one and send it to my update function gift-wrapped in a Msg.” As with onClick, we let update take it from there—and if we so desire, update can return new commands which trigger new calls to update when they complete.


    Figure 3.7 shows how commands flow through the Elm Architecture.

    Figure 3.7 How commands flow through the Elm Architecture


    Importantly, the addition of commands has not altered the fundamental structure of the Elm Architecture we learned in Chapter 2. All of the following still hold true.


    ·   Our Model value is still the single source of truth for the application state.


    ·   Our update function is still the only way to alter that model.


    ·   Sending a Msg to update remains the only way to tell update what we want done.


    All we’ve done is to give update some new powers: the ability to run logic that can have inconsistent results, such as generating a random number. As we will soon see, the key to unlocking these powers is graduating from Html.beginnerProgram to Html.program.


    running commands in response to user input


    We want to generate our random number in response to a click event, so we’ll need to alter our update function a bit. Specifically we’ll have update return a tuple containing not only the new Model we want, but also whatever commands we want the Elm Runtime to execute.


    Before we edit update’s implementation, let’s revise its type annotation to guide our work.
update : Msg -> Model -> ( Model, Cmd Msg )

    As you can see, tuple annotations look just like tuple values. ( Model, Cmd Msg ) means “a tuple where the first element is a Model and the second is a Cmd Msg.”


    Note We write Cmd Msg instead of just Cmd for the same reason that we write Html Msg instead of Html. Like Html, the Cmd type also has a type variable, and for a familiar reason: whenever a Cmd finishes, a Msg value is what it sends to update for further processing.


    Let’s revise our update implementation to reflect this new reality.
update : Msg -> Model -> ( Model, Cmd Msg ) update msg model =     case msg of         SelectByUrl url ->             ( { model | selectedUrl = url }, Cmd.none )           SurpriseMe ->             ( { model | selectedUrl = "2.jpeg" }, Cmd.none )           SetSize size ->            ( { model | chosenSize = size }, Cmd.none )

    A fine start! Next we’ll replace that Cmd.none in SurpriseMe’s branch with a command to randomly generate the index of the photo we want to select.


    3.3.3   Generating Random Values with Random.generate


    The Random.generate function returns a command that generates random values wrapped up in messages. It’s just what the doctor ordered here! Random.generate takes two arguments:


    1. A Random.Generator, which specifies the type of random value we want.


    2. A function which can wrap the resulting random value in one of our Msg values.


    The randomPhotoPicker we created earlier will do nicely for the Random.Generator, so all we’re missing is a way to wrap these randomly-generated integers in a Msg. We can introduce one by adding a new type constructor to Msg and a new branch to update:
type Msg     = SelectByUrl String     | SelectByIndex Int     ...   update : Msg -> Model -> ( Model, Cmd Msg ) update msg model =     case msg of         SelectByIndex index ->             ( { model | selectedUrl = getPhotoUrl index }, Cmd.none )         ...

    Nice! Now we just need Random.generate to produce one of these SelectByIndex messages.


    Calling Random.generate


    When a user clicks the Surprise Me! button, the SurpriseMe -> branch of our update function’s case-expression gets run. Currently that returns a Cmd.none, but now we want it to return a different command: one that generates a random Int and sends it back to update wrapped in a SelectByIndex message. We’ll get this command by calling Random.generate.


    Earlier we noted that Random.generate needs two ingredients:


    ·   A Random.Generator, which specifies what type of random value we want. (We’ll use randomPhotoPicker, our Random.Generator Int, for this.)


    ·   A function which can wrap the resulting random value in one of our Msg values. (We’ll use our SelectByIndex type constructor for this, as it is a function that returns a Msg.)


    Let’s revise our SurpriseMe branch to return an unchanged model, and to call Random.generate instead of returning Cmd.none.
SurpriseMe ->     ( model, Random.generate SelectByIndex randomPhotoPicker )

    Figure 3.8 illustrates how this would cause data to flow through our application, assuming the Elm Runtime randomly generated a 2 after the user clicked our Surprise Me! button.

    Figure 3.8 Data flowing through our application, assuming the Elm Runtime randomly generated a 2


    Calling Random.generate SelectByIndex randomPhotoPicker returns a command which:


    1. Randomly generates an Int between 0 and 2, because that’s the type of random value randomPhotoPicker specified it should generate.


    2. Takes that randomly-generated Int, passes it to SelectByIndex (an Int -> Msg function), and runs the resulting Msg through update.


    We could write a type annotation for Random.generate like so:
generate : (randomValue -> msg) -> Random.Generator randomValue -> Cmd msg

    Note Capitalization is important! There is a big difference between Msg and msg. A function that returns Msg returns an instance of the exact Msg union type we defined in PhotoGroove.elm. In contrast, msg is a type variable. A function that returns msg could return anything at all!


    Table 3.5 shows how the expressions involved in our call to Random.generate relate to the parameters and type variables in this annotation.


    Table 3.5 Types involved in calling Random.generate


    
      
        	
          
            Expression

          

        

        	
          
            Type

          

        

        	
          
            Type in Original Annotation

          

        
      


      
        	SelectByIndex

        	(Int -> Msg)

        	(randomValue -> msg)
      


      
        	randomPhotoPicker

        	Random.Generator Int

        	Random.Generator randomValue
      


      
        	Random.generate     SelectByIndex     randomPhotoPicker

        	Cmd Msg

        	Cmd msg
      

    


    Note Because it takes a function as an argument, Random.generate is another higher-order function like List.map or String.filter.


    upgrading to html.program


    We’re almost done, but if we compile our code right now, we’ll get an error. This is because Html.beginnerProgram wants update to return a Model, whereas our update function now returns a ( Model, Cmd Msg ) tuple. Fortunately beginnerProgram’s older sibling, Html.program, expects just such a tuple! Let’s replace our main declaration with this:
main =     Html.program         { init = ( initialModel, Cmd.none )         , view = view         , update = update         , subscriptions = (\model -> Sub.none)        }

    We’re passing two things differently from what we passed to Html.beginnerProgram.


    1. The model record has been replaced by an init tuple. The first element in the tuple is our initial Model, and the second is a command to run when the application loads. (init is the only place besides update that can specify commands to run.) Since we have nothing to run on startup, we write Cmd.none.


    2. We’ve added a subscriptions field. We’ll make use of that in Chapter 5, but for now let’s disregard it.


    Next we’ll give main a type annotation. As we’ve been going through adding annotations, you may have wondered why the compiler hasn’t been helping out with that. It clearly knows what the types are, since it infers them for us in elm-repl, so why can’t it tell us the annotations?


    Generating annotations with the --warn flag


    Actually, it totally can! Try running elm-make with the --warn flag:
elm-make PhotoGroove.elm --output elm.js --warn

    You should see this warning:
Top-level value `main` does not have a type annotation.   139| main =      ^^^^ I inferred the type annotation so you can copy it into your code:  main : Program Never Model Msg

    Thanks, compiler! Go ahead and copy this final annotation right into your code. (Chapter 5 will get into what Program Never Model Msg means.) Generated annotations won’t always match what you would have written yourself, but you can at least expect them to compile.


    Tip This is presented as a compiler warning because it’s best practice to document your top-level functions with type annotations. When the compiler can generate these for you, documentation takes much less effort!


    At this point you may be wondering something else: why did we go through this whole chapter writing annotations by hand, when we could have had the compiler generate immediately usable ones for us all along? To that I say...hey look, a distraction!


    The final build


    If you recompile with elm-make and open index.html, you should experience a gloriously functional application. Try clicking Surprise Me! to watch it randomly pick a different photo!


    [image: ]



    Figure 3.9 Randomly choosing the initially selected photo


    Ain’t it grand? Even after all the changes we made, once it compiled it just worked—no regressions! Not only that, but our code never came anywhere near causing a runtime exception. This is a normal experience with Elm, yet it never seems to get old.


    Compared to where we were at the end of Chapter 2, our revised application is not only more feature-rich, it’s also more reliable and better-documented. It does more, it does it better, and the code is cleaner. Huzzah!


    3.4   Summary


    We’ve now improved on our application from Chapter 2 in several ways.


    ·   We added documentation in the form of type annotations


    ·   Users can select from one of three thumbnail sizes


    ·   It has a Surprise Me! button which selects a thumbnail at random


    ·   From now on whenever we add a new Msg value, the compiler will give us a Missing Patterns error if we forget to handle it (like it did when we didn’t account for SetSize)


    Along the way we covered many new concepts, such as:


    ·   Type variables represent concrete types that have not been specified yet.


    ·   A type alias declaration assigns a name to a type, much like how a constant assigns a name to a value.


    ·   A type declaration defines a new union type, a custom type that did not exist before.


    ·   Union types can hold more flexible data than records or tuples can.


    ·   Union type constructors can be either values which are instances of that union type, or functions which return instances of that union type.


    ·   You can destructure union type constructors in case-expressions to extract their data.


    ·   If you don’t write a fallback _ -> branch in a case-expression, you’ll get a compiler error unless your code handles all possible cases.


    ·   Array.get prevents runtime crashes by returning a Maybe instead of a normal element.


    We also learned some differences between comments and type annotations.


    Table 3.6 Documenting Code with Comments and Type Annotations


    
      
        	
          
            Comment

          

        

        	
          
            Type Annotation

          

        
      


      
        	
          Arbitrary string, can describe anything

        

        	
          Can only describe a thing’s type

        
      


      
        	
          Can be inaccurate or get out of date

        

        	
          Compiler guarantees it’s accurate and up-to-date

        
      


      
        	
          Can be written just about anywhere

        

        	
          Always goes on the line above the thing it documents

        
      

    


    Finally, we learned about Maybe, the container value which Array.get returns to represent the potential absence of a value. Table 3.7 compares Maybe and List.


    Table 3.7 Comparing Maybe and List


    
      
        	
          
            Container Contents

          

        

        	
          
            List

          

        

        	
          
            Maybe

          

        
      


      
        	
          Empty

        

        	
          [] : List a

        

        	
          Nothing : Maybe a

        
      


      
        	
          One value

        

        	
          [ "foo" ] : List String

        

        	
          Just "foo" : Maybe String

        
      


      
        	
          Two values

        

        	
          [ "foo", "bar" ]: List String

        

        	
          not possible!

        
      

    


    In Chapter 4 we’ll take our application to the extreme, by talking to a server to obtain our list of photos to display. Stay tuned!


    Listing 3.7 The complete PhotoGroove.elm
module PhotoGroove exposing (..)   import Html exposing (..) import Html.Attributes exposing (..) import Html.Events exposing (onClick) import Array exposing (Array) import Random     urlPrefix : String urlPrefix =     "http://elm-in-action.com/"     type ThumbnailSize     = Small     | Medium     | Large     view : Model -> Html Msg view model =     div [ class "content" ]         [ h1 [] [ text "Photo Groove" ]         , button             [ onClick SurpriseMe ]             [ text "Surprise Me!" ]         , h3 [] [ text "Thumbnail Size:" ]         , div [ id "choose-size" ]             (List.map viewSizeChooser [ Small, Medium, Large ])         , div [ id "thumbnails", class (sizeToString model.chosenSize) ]             (List.map (viewThumbnail model.selectedUrl) model.photos)         , img             [ class "large"             , src (urlPrefix ++ "large/" ++ model.selectedUrl)             ]             []         ]     viewThumbnail : String -> Photo -> Html Msg viewThumbnail selectedUrl thumbnail =     img         [ src (urlPrefix ++ thumbnail.url)         , classList [ ( "selected", selectedUrl == thumbnail.url ) ]         , onClick (SelectByUrl thumbnail.url)         ]         []     viewSizeChooser : ThumbnailSize -> Html Msg viewSizeChooser size =     label []         [ input [ type_ "radio", name "size", onClick (SetSize size) ] []         , text (sizeToString size)         ]     sizeToString : ThumbnailSize -> String sizeToString size =     case size of         Small ->             "small"           Medium ->             "med"           Large ->             "large"     type alias Photo =     { url : String }     type alias Model =     { photos : List Photo     , selectedUrl : String     , chosenSize : ThumbnailSize     }     initialModel : Model initialModel =     { photos =         [ { url = "1.jpeg" }         , { url = "2.jpeg" }         , { url = "3.jpeg" }         ]     , selectedUrl = "1.jpeg"     , chosenSize = Medium     }     photoArray : Array Photo photoArray =     Array.fromList initialModel.photos     getPhotoUrl : Int -> String getPhotoUrl index =     case Array.get index photoArray of         Just photo ->             photo.url           Nothing ->             ""     randomPhotoPicker : Random.Generator Int randomPhotoPicker =     Random.int 0 (Array.length photoArray - 1)     type Msg     = SelectByUrl String     | SelectByIndex Int     | SurpriseMe     | SetSize ThumbnailSize     update : Msg -> Model -> ( Model, Cmd Msg ) update msg model =     case msg of         SelectByIndex index ->             ( { model | selectedUrl = getPhotoUrl index }, Cmd.none )           SelectByUrl url ->             ( { model | selectedUrl = url }, Cmd.none )           SurpriseMe ->             ( model, Random.generate SelectByIndex randomPhotoPicker )           SetSize size ->             ( { model | chosenSize = size }, Cmd.none )     main =     Html.program         { init = ( initialModel, Cmd.none )         , view = view         , update = update         , subscriptions = (\model -> Sub.none)        }
  


  
    4    Talking to Servers


    This chapter covers


    ·   Using Decoders to validate and translate JSON


    ·   Handling descriptive errors with Results


    ·   Communicating over HTTP with Requests


    We’ve made great progress! Users of our Photo Groove application can now do quite a bit:


    ·   Select a thumbnail to view a larger version


    ·   Choose Small, Medium, or Large thumbnail sizes


    ·   Click Surprise Me! to select a random thumbnail


    Our manager is impressed with our progress, but has even more in mind. “It’s time to take Photo Groove to the next level. The highest level. The cloud level. That’s right, we’re going to have Photo Groove start getting its photo information from our very own servers!”


    As with last time, our manager has one more minor feature request. “There’s also going to be some metadata associated with each photo—specifically download size and an optional caption. We can have it show those on top of the big photo, right?”


    Sure, since you asked so nicely.


    In this chapter we’ll teach our application how to talk to servers. We’ll validate and translate JSON data, communicate over HTTP using the Elm Architecture, and reliably handle errors in client-server communication.


    Let’s get to it!


    4.1   Preparing for Server-Loaded Data


    Since we’re going to be loading our initial list of photos from the server, our data modeling needs will change.


    4.1.1   Modeling Incremental Initialization


    Right now our initialModel looks like this:
initialModel : Model initialModel =     { photos = [ { url = "1.jpeg" }, { url = "2.jpeg" }, { url = "3.jpeg" } ]     , selectedUrl = "1.jpeg"     , chosenSize = Medium    }

    We’ll be loading our photos from the server, meaning initialModel.photos will be [] and we’ll update that list once we hear back from the server. This will have a few consequences:


    1. Because we have no initial photos, we can’t have an initial selectedUrl either.


    2. The initial request to load the data might fail, for example because the server is down. In that case, we should display an error.


    3. Our photoArray value references initialModel.photos, so it will be affected.


    4. Both getPhotoUrl and randomPhotoPicker reference photoArray, so they will be affected as well.


    Let’s translate these new considerations into data model changes!


    modeling absent values with Maybe


    The Maybe container we learned about in Chapter 3 seems like a nice way to implement the first consideration—that we won’t have an initial selectedUrl. Since Maybe represents the potential absence of a value—by holding either the value Nothing or some other value wrapped in a Just—it’s an intuitive way to represent the data that is absent until we hear back from the server.


    We can also use a Maybe to represent a potential loading error message. It will begin as Nothing, but might become something like Just "Server unavailable" if we get an error.


    Go ahead and change Model and initialModel to look like the following.
type alias Model =     { photos : List Photo     , selectedUrl : Maybe String     , loadingError : Maybe String     , chosenSize : ThumbnailSize     }     initialModel : Model initialModel =     { photos = []     , selectedUrl = Nothing     , loadingError = Nothing     , chosenSize = Medium    }

    Now that we’ve revised our data model, we can lean on the compiler to tell us what parts of our code base were affected by this revision. This will let us reliably propagate these changes throughout our code base!


    propagating changes


    If we recompile we’ll see two type mismatches: one in view and one in update.


    Since selectedUrl is now a Maybe String instead of a String, our SelectByUrl logic inside update now needs a Just. Let’s add it:
SelectByUrl url ->     ( { model | selectedUrl = Just url }, Cmd.none )

    Our SelectByIndex logic has a similar type mismatch, but there we can actually do better! One way to fix the mismatch is to wrap what getPhotoUrl returns in a Just...but another way is to propagate the Maybe itself! Let’s rewrite getPhotoUrl to return a Maybe:
getPhotoUrl : Int -> Maybe String getPhotoUrl index =     case Array.get index photoArray of         Just photo ->             Just photo.url           Nothing ->            Nothing

    Notice how this not only resolves the type mismatch, but also means we are no longer returning the clearly invalid URL of "" in the case where the index is out of bounds. One less potential bug to worry about!


    Fixing viewthumbnail


    Now that we’ve resolved our type mismatches in update, we can move on to resolving the ones in view. One of the problems in view is that the viewThumbnail function it calls has an inaccurate type annotation. It says it’s taking a String called selectedUrl, but selectedUrl is now a Maybe String. Let’s fix that annotation real quick:
viewThumbnail : Maybe String -> Photo -> Html Msg viewThumbnail selectedUrl thumbnail =

    There’s another problem in viewThumbnail: the code selectedUrl == thumbnail.url is now comparing a String to a Maybe String, which is a type mismatch! The easiest way to fix this is by calling Just on thumbnail.url, converting it to a Maybe String. Now we’re comparing a Maybe String to a Maybe String and balance has been restored to the galaxy.
classList [ ( "selected", selectedUrl == Just thumbnail.url ) ]

    Fixing the preview image


    With those “quick fixes” out of the way, we can recompile to see one final error: the code that renders the large preview image has a type mismatch.
src (urlPrefix ++ "large/" ++ model.selectedUrl)

    This case is not a “quick fix” like the others were. If model.selectedUrl is Nothing here, we really don’t want to render this img at all!


    Let’s pull the logic to view the large photo into its own function, and handle the Maybe in there by declining to render the img at all.
view model =     div [ class "content" ]         [ h1 [] [ text "Photo Groove" ]         ... (everything in between stays the same) ...         , div [ id "thumbnails", class (sizeToString model.chosenSize) ]             (List.map (viewThumbnail model.selectedUrl) model.photos)         , viewLarge model.selectedUrl         ]   viewLarge : Maybe String -> Html Msg viewLarge maybeUrl =     case maybeUrl of         Nothing ->             text ""           Just url ->            img [ class "large", src (urlPrefix ++ "large/" ++ url) ] []

    Notice that we’re returning text "" if maybeUrl is Nothing. This “render an empty text node” technique is a quick way to obtain some Html that has no visual impact on the page.


    Let’s recompile everything with elm-make --output=elm.js PhotoGroove.elm and open index.html to see our progress.

    Figure 4.1 The updated application, with no photos displaying


    Excellent! We’ve achieved that peculiar flavor of success where we smile at our desk...while hoping nobody walks by and sees how the previously-working application looks right now.


    4.1.2   Resolving Data Dependencies


    Now that we’ve resolved our type mismatches, let’s turn to our business logic.


    photoArray is still based on initialModel.photos. This is perfectly reasonable if initialModel.photos has a bunch of photos in it...but now that it’s hardcoded to []? Not so much. That means photoArray will always be empty, which means even once we’ve loaded some photos to display, our Surprise Me! button won’t be able to randomly select one.


    Fixing selectbyindex


    To fix this, we need to make our SelectByIndex logic reference the current photos, rather than the hardcoded ones we had at startup. We can do this by deleting photoArray and getPhotoUrl, and changing the SelectByIndex branch of our update function like so:
SelectByIndex index ->     let         newSelectedPhoto : Maybe Photo         newSelectedPhoto =             Array.get index (Array.fromList model.photos)           newSelectedUrl : Maybe String         newSelectedUrl =             case newSelectedPhoto of                 Just photo ->                     Just photo.url                   Nothing ->                     Nothing     in        ( { model | selectedUrl = newSelectedUrl }, Cmd.none )

    Our newSelectedUrl implementation fits a pattern that comes up fairly often:


    1. Run a case-expression on a Maybe


    2. In the Nothing branch, evaluate to Nothing


    3. In the Just branch, evaluate to another Just


    In other words, if we have a Nothing, we leave it alone, but if we have a Just, we transform it into a different Just.


    Refactoring to use Maybe.map


    The Maybe.map function provides a way to perform this transformation more concisely. Let’s refactor our getPhotoUrl implementation to use it.
newSelectedUrl : Maybe String   newSelectedUrl =    Maybe.map (\photo -> photo.url) newSelectedPhoto

    You can read this code as “Look at the result of Array.get index photoArray. If it’s Nothing, then return Nothing, but if it’s Just photo, return Just photo.url instead.”


    Maybe.map does this by accepting a function which transforms the old Just value into the new one. (If the Maybe is Nothing, this Just-transforming function does not get called.)


    Table 4.1 illustrates the similarities between Maybe.map and our old friend List.map.


    Table 4.1 Comparing List.map and Maybe.map


    
      
        	
          
            List.map

          

        

        	
          
            Maybe.map

          

        
      


      
        	List.map negate [ 39 ] == [ -39 ]

        	Maybe.map negate (Just 39) == Just -39
      


      
        	List.map negate [ 5 ]  == [ -5 ]

        	Maybe.map negate (Just 5)  == Just -5
      


      
        	List.map negate []     == []

        	Maybe.map negate Nothing   == Nothing
      

    


    This code is now significantly more concise than when we used an entire case-expression to implement the same logic...but wait! There’s more!


    Refactoring to use .url


    Anytime you see a function that takes a record and immediately return one of its fields—as we’re doing with (\photo -> photo.url) above—you can take advantage of a special syntax Elm offers for these occasions. If we replace (\photo -> photo.url) with .url instead, it will do exactly the same thing.


    That means we can delete newSelectedPhoto and refactor newSelectedUrl to this:
newSelectedUrl : Maybe String newSelectedUrl =    Maybe.map .url (Array.get index (Array.fromList model.photos))

    Tip You can think of .url as “a function which takes a record and returns its .url field.”


    We’ve now replaced a case-expression with one function call, but the result is not the nicest to read. Fortunately, Elm provides an excellent way to make expressions like this more readable!


    Using The pipeline operator


    We can express newSelectedUrl as a “pipeline” of sequential operations by using the |> operator like so:
newSelectedUrl : Maybe String newSelectedUrl =     model.photos         |> Array.fromList         |> Array.get index        |> Maybe.map .url

    This expression is exactly equivalent to writing Maybe.map .url (Array.get index (Array.fromList model.photos)) except it’s expressed in a different style.


    Figure 4.2 compares these two ways to write the same expression.

    Figure 4.2 The same expression, with and without pipelines


    The pipelined code is saying the following:


    1. Start with model.photos


    2. Pass it to Array.fromList


    3. Pass the previous step’s return value as the final argument to Array.get index


    4. Pass that previous step’s return value as the final argument to Maybe.map .url


    Pipelines make it easier to read and to modify sequences of transformations like this. We’ll be using them quite often throughout the rest of the book!


    Note Before Elm’s compiler generates JavaScript code, it quietly rewrites pipelines into normal function calls. So "foo" |> String.reverse compiles to the same thing as String.reverse "foo". Since they compile to the same thing, there’s no performance cost to choosing one over the other. Choose whichever makes for nicer code!


    Go ahead and revise our SelectByIndex logic to use this pipeline style, like so:
SelectByIndex index ->     let         newSelectedUrl : Maybe String         newSelectedUrl =             model.photos                 |> Array.fromList                 |> Array.get index                 |> Maybe.map .url     in        ( { model | selectedUrl = newSelectedUrl }, Cmd.none )

    inlining randomphotopicker


    Now we have one remaining problem: randomPhotoPicker is still referencing the photoArray value we deleted.


    Fortunately, it was only using photoArray to get its length inside the SurpriseMe branch of update’s case-expression. We can get that same length value using List.length model at that point in the code, so we can delete the top-level randomPhotoPicker in favor of this:
SurpriseMe ->     let         randomPhotoPicker =             Random.int 0 (List.length model.photos - 1)     in        ( model, Random.generate SelectByIndex randomPhotoPicker )

    Great! Now that we’ve installed the requisite package and prepared our data model for the transition, we’re ready to get some HTTP going!


    4.2   Fetching Data from a Server


    One of our helpful coworkers has set up a simple server endpoint which returns a comma-separated list of photo filenames. The list doesn’t yet contain the metadata that we’ll ultimately need, but it’ll give us a good starting point toward that goal.


    Let’s send a HTTP GET request to fetch that comma-separated string from a server!


    4.2.1   Describing HTTP Requests


    We’ll be using the Http module to send HTTP requests to our server. Since Http is not one of Elm’s core modules, we need to install the elm-lang/http package to gain access to it:
elm-package install elm-lang/http

    In Chapter 6 we’ll learn more about package management, but for now let’s keep moving!


    managed effects instead of side effects


    In Chapter 3 we saw how generating random numbers in Elm must be done with a Cmd rather than a plain function call.


    Whereas JavaScript’s Math.random() can return a different random number each time you call it, Elm functions must be more consistent. When you pass an Elm function the same arguments, it’s guaranteed return the same value.


    There’s another rule that applies to all Elm functions, and this one affects the HTTP requests we’re about to make. The rule is that Elm functions cannot have side effects.


    Definition An effect is an operation that modifies external state. A function which modifies external state when it executes has a side effect.


    HTTP requests can always modify external state—since even a GET request can result in a server changing values in a database—so performing an HTTP request is an effect. This means if we execute a function and it performs an HTTP request, that function has a side effect.


    Listing 4.1 shows a few JavaScript functions which consistently return the same value, but which have side effects because they modify external state along the way. None of these could be written as plain Elm functions.


    Listing 4.1 JavaScript functions which have side effects
function storeStuff() {         ❶      localStorage.stuff = "foo"; ❶      return 1;                   ❶  }                               ❶    function editField(object) {    ❷      object.foo = "bar";         ❷      return object;              ❷  }                               ❷    var something = 1;   function editVar() { ❸      something = 2;   ❸      return 0;        ❸ }                    ❸ 

    ❶   modifies the contents of localStorage


    ❷   modifies the object it receives


    ❸   modifies an external variable


    If you look through all the code we’ve written for Photo Groove, you won’t find a single function that has a side effect. All effects are performed by the Elm Runtime itself; our code only describes which effects to perform, by returning values from update.


    Note Calling update does not directly alter any state! All update does is return a tuple. If you wanted to, you could call update a hundred times in a row, and all it would do is give you a hundred tuples.


    This system of managed effects, where the Elm Runtime is in charge of performing all effects, means that Elm programs can be written entirely in terms of data transformations. In Chapter 6 we’ll see how nice this makes Elm code to test.


    This means that in order to perform a HTTP request, we’ll do what we did with random number generation in Chapter 3: use a Cmd to tell the Elm runtime to perform that effect.


    Describing requests with Http.getString


    The Http.send function returns a Cmd representing the HTTP request we want to make.


    To configure this request, we pass Http.send an Http.Request value describing what the request should do. This Request value serves a similar purpose to the Random.Generator value we passed to Random.generate in Chapter 3—except where Random.Generator described the random value we wanted, Request describes the HTTP request we want.


    One way we can obtain a Request value is with the Http.getString function, like so:
Http.getString "http://elm-in-action.com/photos/list"

    The getString function has this type:
getString : String -> Request String

    We pass it a URL string, and it returns a Request that asks the Elm Runtime to send a HTTP GET request to that URL. If the request succeeds, we’ll end up with a plain String containing the response body the server sent back.


    Translating Requests into commands WITH HTTP.SEND


    Now that we have our Request String, we can pass it to Http.send to get a Cmd Msg. Just like in Chapter 3 with random number generation, once this Cmd completes, it will send a Msg to update telling us what happened.

    Figure 4.3 The Elm Runtime executing a Cmd from Http.send


    Unlike random number generation, a lot can go wrong with a HTTP request! What if we send the server an invalid URL? Or an unrecognized one? What if the server has been unplugged by a mischievous weasel? If the request fails for any reason, we’ll want to know went wrong so we can inform the user.


    Http.send produces a value that accounts for this possibility: a Result.


    4.2.2   Handling HTTP Responses


    To get a feel for how Result works, let’s compare it to its cousin Maybe.


    In Chapter 3 we saw how Array.get returned a Maybe in order to account for the possibility of a missing value. The definition of Maybe looks like this:
type Maybe value     = Just value    | Nothing

    Elm has a similar union type for representing the result of an operation that can fail, such as performing a HTTP request. It’s called Result, and it looks like this:
type Result errValue okValue     = Err errValue    | Ok okValue

    As we can see, Result has two type variables:


    ·   okValue represents the type of value we’ll get if the operation succeeded. For example, if we are trying to obtain a String from the server, then okValue would be String.


    ·   errValue represents the type of error we’ll get if the operation failed. In our case, errValue will refer to a union type that enumerates various mishaps that can befall HTTP requests. That union type is called Http.Error.


    Similarly to how Maybe requires that we write logic to handle both the Just and Nothing cases, so too does Result require that we handle both the Ok and Err cases.


    Handling the Result


    Let’s start by importing the Http module we’re about to begin using.
import Http

    Next let’s add a Msg type constructor to handle the Result value we’ll get from Http.send:
type Msg     = SelectByUrl String     | SelectByIndex Int     | SurpriseMe     | SetSize ThumbnailSize    | LoadPhotos (Result Http.Error String)

    We’ll learn more about that Http.Error type in a bit. For now, let’s add a branch to our update function’s case-expression for this new Msg. We’ll start with something like this:
LoadPhotos result ->     case result of         Ok responseStr ->             ...translate responseStr into a list of Photos for our Model...           Err httpError ->            ...record an error message here...

    With that basic skeleton in place, we can implement these two cases one at a time. We’ll start with the Ok responseStr case, where the server gave us a valid response and all is well.


    READING Photo FILENAMES with string.split


    Remember earlier when we mentioned how we’ve got a server endpoint which returns a comma-separated list of photo filenames? If not, here it is again:


    One of our helpful coworkers has set up a simple server endpoint which returns a comma-separated list of photo filenames.


    
      —This book, earlier.

    


    Sure enough, if we visit this endpoint at http://elm-in-action.com/photos/list we can see the list of filenames. It’s a plain old string:
"1.jpeg,2.jpeg,3.jpeg,4.jpeg"

    When we wrote Ok responseStr a moment ago, that responseStr value will refer to exactly this string! Now we can split responseStr into a list of individual filenames using Elm’s String.split function. It has this type:
split : String -> String -> List String

    We give String.split a separator (in this case a comma) and a string to split, and it gives us back a list of all the strings it found between those separators.


    Table 4.2 shows what String.split "," will do to the string we get from the server.


    Table 4.2 Calling String.split "," on a string


    
      
        	
          stringFromServer

        

        	
          String.split "," stringFromServer

        
      


      
        	
          "1.jpeg,2.jpeg,3.jpeg,4.jpeg"

        

        	
          [ "1.jpeg", "2.jpeg", "3.jpeg", "4.jpeg" ]

        
      

    


    Once we have this list of filename strings, we can use it to set the photos : List Photo field in our Model like in Listing 4.2.


    Listing 4.2 Incorporating responseStr into the Model
Ok responseStr ->     let         urls =             String.split "," responseStr           ❶            photos =             List.map (\url -> { url = url }) urls  ❷       in         ( { model | photos = photos }, Cmd.none ) ❸ 

    ❶   Split the String into a List String


    ❷   Translate the List String into a List Photo


    ❸   Set the photos in the model


    Now when we successfully receive a String response from the server, we’ll translate it into a list of Photo records and store them in our Model.


    Using Type Aliases to create Records


    Declaring type alias Photo = { url : String } does more than giving us a Photo type we can use in type annotations. It also gives us a convenience function whose job is to build Photo record instances. This function is also called Photo. Here it is in action:
Photo "1.jpeg" == { url = "1.jpeg" }

    This also works with record type aliases involving multiple fields, like the one for Model:
type alias Model =     { photos : List Photo     , selectedUrl : String     , chosenSize : ThumbnailSize    }

    This declaration gives us a convenience function called Model which builds a record and returns it. Since this record has three fields where Photo had only one, the Model function accepts three arguments instead of one. The type of the Model function looks like this:
Model : List Photo -> String -> ThumbnailSize -> Model

    The order of arguments matches the order of the fields in the type alias declaration. So if you were to move the photos : List Photo declaration to the end of the type alias, then the Model function would look like this instead:
Model : String -> ThumbnailSize -> List Photo -> Model

    We can use this knowledge to perform a quick refactor of our Photo construction:
Old: List.map (\url -> { url = url }) urls New: List.map Photo urls

    Lovely!


    Handling Errors (OR NOT)


    What about when things don’t go as planned? We’ll add some nice error handling later, but first we want to get something up and running. Elm won’t let us forget to handle the error case—if we omitted the Err branch, we’d get a MISSING PATTERNS compile error!—but the compiler won’t stop us from choosing to do nothing when we receive an Err.
Err httpError ->     ( model, Cmd.none )

    Since we aren’t using our httpError value, let’s rename it to an underscore:
Err _ ->     ( model, Cmd.none )

    
      The underscore parameter

    


    The underscore is a special parameter name whose purpose is to signify that there is a parameter here, but we’re choosing not to use it. Unlike a normal parameter, attempting to reference _ in our logic would be a compile error.


    One handy feature of _ is that you can use it multiple times in the same parameter list. For example:


     
functionThatTakesThreeArguments _ _ _ =     "I ignore all three of my arguments and return this string!" You can use _ either in function parameter lists or in case-expression branches.

    
       

    


    


    Later on we’ll circle back to this branch of our case-expression and use it to show the user a nice error message when an HTTP error happens.



    Pattern Matching


    Before we move on, let’s pause for a tasty refactor. If you zoom out a bit, you’ll notice we have a case-expression nested directly inside another case-expression.
case msg of       ...       LoadPhotos result ->         case result of             Ok responseStr ->                 ...               Err _ ->                ( model, Cmd.none )

    In situations like this, we can use concise pattern matching to express the same logic:
case msg of       ...       LoadPhotos (Ok responseStr) ->         ...       LoadPhotos (Err _) ->        ( model, Cmd.none )

    Definition Pattern matching is a way of destructuring values based on how their containers look. In the example above, if we have a LoadPhotos containing an Ok containing a value, that value will go into a variable called responseStr.


    This refactored code is equivalent to what we had before. The difference is that now each of our branches is expressing two conditions at once:


    ·   The LoadPhotos (Ok responseStr) branch runs if msg is a LoadPhotos type constructor which contains an Ok value.


    ·   The LoadPhotos (Err _) branch runs if msg is a LoadPhotos type constructor and it contains an Err value.


    You can nest patterns like these as much as you want, even assembling elaborate creations like NestAllTheThings (Just (Ok (Listen (This (IsReally "great" _ _))))) - but try not to overdo it.


    We’ll encounter other pattern matching techniques as we progress through the book.


    4.2.3   Sending HTTP Requests


    At this point we’ve assembled all the ingredients necessary to launch our HTTP rocketship in the direction of our server:


    ·   The Request String returned by Http.getString "http://elm-in-action.com/photos/list"


    ·   The LoadPhotos message that will handle the response


    The Http.send function will wire these up for us! It has this type:
send : (Result Http.Error val -> msg) -> Request val -> Cmd msg

    Http.send has two type variables: msg and val. We can infer what they’ll become in our case:


    ·   Since we’re passing a Request String, the val type variable must become String


    ·   Since we need it to return a Cmd Msg, the msg type variable must become Msg


    Knowing this, we can tell that the first argument to send is a function that looks like this:
(Result Http.Error String -> Msg)

    This means we could declare a new value called initialCmd using this call to Http.send:
initialCmd : Cmd Msg initialCmd =     Http.send         (\result -> LoadPhotos result)        (Http.getString "http://elm-in-action.com/photos/list")

    We’ll use this initialCmd value to run our HTTP request when the program starts up.


    This will compile, but we can simplify it! Back in Section 2.2.1 of Chapter 2, we noted that an anonymous function like (\foo -> bar baz foo) can always be rewritten as (bar baz) by itself. This means we can replace (\result -> LoadPhotos result) with LoadPhotos like so:
initialCmd : Cmd Msg initialCmd =     Http.send LoadPhotos        (Http.getString "http://elm-in-action.com/photos/list")

    Using pipeline style for initialcmd


    By the way, remember how earlier we refactored newSelectedUrl to use a pipeline style instead of multiple constants? We can use the same technique to express initialCmd in terms of the steps needed to build it up:
initialCmd : Cmd Msg initialCmd =     "http://elm-in-action.com/photos/list"         |> Http.getString        |> Http.send LoadPhotos

    Nice! Let’s add this version of initialCmd to PhotoGroove.elm right above our main declaration.


    Running a Command on Init


    Now we have a Cmd to run, but we want to kick this one off a bit differently than last time. We ran our random number generation command from update in response to a user click, but we want to run this command right when the program starts up.


    We can do that by updating main to use initialCmd instead of Cmd.none on init, like so:
main : Program Never Model Msg main =     Html.program         { init = ( initialModel, initialCmd )         , view = view         , update = update         , subscriptions = \_ -> Sub.none  ❶         }

    ❶   We don’t use this argument, so name it _


    While we’re at it, we also updated subscriptions to use _ for its anonymous function’s argument, since that argument never gets used.


    TIP An anonymous function which ignores its argument (like the \_ -> above) looks like a hockey stick.


    Figure 4.4 illustrates how our revised init kicks off the sequence of events leading up to our update function receiving a LoadPhotos message.

    Figure 4.4 Running a LoadPhotos command on init


    Now let’s recompile and open index.html, so we can bask in the glory of our new server-loaded photos!

    Figure 4.5 The updated application, with an additional photo loaded


    Excellent! However, we’ve lost something along the way: we no longer select the first photo like we did before.


    Selecting an initial photo using List.head


    We’d like to select the first URL that comes back from the server, whatever that URL may be. The List.head function is perfect for this! Its type looks like this:
head : List elem -> Maybe elem

    List.head someList and Array.get 0 someArray serve essentially the same purpose: returning Nothing if run on an empty collection, or Just whateverElementWasFirst if the collection had anything in it. This means we can grab the first URL out of the list like so:
List.head urls

    Since we’re going to set selectedUrl to this value, and selectedUrl is already a Maybe String, we don’t need run a case-expression on this Maybe String to do any special handling. We can drop it in and let whoever works with selectedUrl handle the Nothing case.


    Let’s expand the record update in the LoadPhotos (Ok responseStr) branch to set not only the photos field but the selectedUrl field as well:
LoadPhotos (Ok responseStr) ->     let         urls =             String.split "," responseStr           photos =             List.map Photo urls     in         ( { model             | photos = photos             , selectedUrl = List.head urls           }         , Cmd.none        )

    Now we can try it out and see how this looks. Recompile and...

    Figure 4.6 Selecting the first loaded photo by default.


    Beautiful! Now that we’re loading photos from the server, we’re ready to take the final step: obtaining the metadata and complete URLs from the server rather than just the filenames.


    Now that we’ve gotten things working in the case where the response successfully comes back, let’s revisit the case where everything is not all sunshine and kittens. It’s time for some proper error handling!


    4.2.4   Gracefully Handling Errors


    Let’s make it so that if there’s an error, we render a helpful message for our users. To do this, we’ll need to make two changes:


    1. Have update set this error message when it receives a ReportError message.


    2. Have view render this error message if it’s present.


    Let’s start with the change to update:
LoadPhotos (Err _) ->     ( { model         | loadingError = Just "Error! (Try turning it off and on again?)"       }     , Cmd.none    )

    Note We need the Just here because we’re storing loadingError in our Model as a Maybe String, whereas ReportError holds a plain String.


    Next let’s introduce a brand-new function which renders the error message, and then modify Html.program to use it instead of using view directly:
viewOrError : Model -> Html Msg viewOrError model =     case model.loadingError of         Nothing ->             view model           Just errorMessage ->             div [ class "error-message" ]                 [ h1 [] [ text "Photo Groove" ]                 , p [] [ text errorMessage ]                 ]     main : Program Never Model Msg main =     Html.program         { init = ( initialModel, initialCmd )        , view = viewOrError

    Whoa! Can we do that? Can we pass a function other than view to Html.program as the view function to use?


    Well, sure! All Html.program needs there is a function of type  Model -> Html Msg. As long as we give it a function of that type, Html.program could care less what that function is called.


    We’ve done several refactors where we split smaller functions out of view, but this is the first time we’ve wrapped view in another function. And it was no big deal! One of the nice things about working with plain old Elm functions all the time is that they’re easily interchangeable as long as the types line up.


    verifying that Error Handling works


    Let’s start by making sure the error handling works properly. Change initialCmd to fetch data from an invalid URL:
initialCmd : Cmd Msg initialCmd =     "http://elm-in-action.com/breakfast-burritos/list"         |> Http.getString        |> Task.perform handleLoadFailure handleLoadSuccess

    Now when you recompile and open index.html, you’ll see our new error message:

    Figure 4.7 Displaying an error as a status message


    Now let’s change initialCmd back to "http://elm-in-action.com/photos/list" and press on toward receiving that photo metadata!


    4.3   Decoding JSON


    Browsers can talk to servers using many different formats. JSON, XML, RPC, BBQ...the list goes on. Regardless of the format, when the browser sends off a HTTP request to a server, the response comes back as a raw string. It’s up to the program running in the browser to decode that raw string into a useful representation.


    The server with the metadata will be sending us that information via JSON. This means we’ll need some way to decode a JSON string into something more useful!


    4.3.1   Decoding JSON Strings into Results


    Plenty of things can go wrong during the process of decoding JSON. Suppose the server messes up and sends us XML instead of JSON. Our JSON-decoding logic will certainly fail! Or suppose we get back a response that’s valid JSON, but a required field is missing. What then?


    The decodestring function


    The Json.Decode.decodeString function returns a Result. Its complete type is:
decodeString : Decoder val -> String -> Result String val

    Notice that the Result we get back from decodeString has one concrete type—namely that errValue is always a String—and one that matches the type of the given decoder’s type variable. Table 4.3 shows how the decoder passed to decodeString affects its return type.


    Table 4.3 How the Decoder passed to decodeString affects its return type


    
      
        	
          
            Decoder passed in

          

        

        	
          
            decodeString returns

          

        

        	
          
            Example success value

          

        

        	
          
            Example failure value

          

        
      


      
        	Decoder Bool

        	Result String Bool

        	Ok True

        	Err "nope"
      


      
        	Decoder String

        	Result String String

        	Ok "Win!"

        	Err "nah"
      


      
        	Decoder (List Int)

        	Result String (List Int)

        	Ok [ 1, 2, 3 ]

        	Err "argh!"
      

    


    decoding primitives


    The Json.Decode module has a Decoder Bool called bool, which translates a JSON boolean string (either "true" or "false") into an Elm Bool (True or False). Let’s try it in elm-repl!


    Listing 4.1  Using decodeString bool
> import Json.Decode exposing (..)   > decodeString bool "true" Ok True : Result String Bool   > decodeString bool "false" Ok False : Result String Bool   > decodeString bool "42" Err "Expecting a Bool but instead got: 42" : Result String Bool   > decodeString bool "@&!*/%?"Err "Given an invalid JSON: Unexpected token @" : Result String Bool

    Note elm-repl prints types in fully-qualified style to avoid ambiguity, meaning it always prints Result.Result instead of Result. These examples will use the unqualified Result for brevity, which means they won’t have the quite same output as what you’ll see in your own elm-repl.


    Besides bool, the Json.Decode module offers other primitive decoders like string, int, and float. They work similarly to bool, as we can see in elm-repl:


    Listing 4.2  Using decodeString with int, float, and string
> import Json.Decode exposing (..)   > decodeString float "3.33" Ok 3.33 : Result String Float > decodeString string "\"backslashes escape quotation marks\"" Ok "backslashes escape quotation marks" : Result String String   > decodeString int "76" Ok 76 : Result String Int   > decodeString int "3.33"Err "Expecting an Int but instead got: 3.33" : Result String Int

    The only primitive decoders are bool, int, float, string, and null, since JavaScript’s undefined is not allowed in valid JSON. As we will soon see, there are more popular ways of handling null than with this primitive decoder.


    4.3.2   Decoding JSON Collections


    We’ve now seen how to decode primitives like booleans, integers, and strings from JSON into their Elm counterparts. This is a good start, but JSON also supports arrays and objects, and we’ll need to decode both of these types in order to receive our list of photos with metadata.


    Decoding json Arrays into Lists


    Suppose we have the JSON string "[true, true, false]". To decode a list of booleans from this JSON array, we can write list bool. This will give us a Decoder (List Bool) value.


    Here’s how Json.Decode.bool and Json.Decode.list compare:
bool : Decoder Bool list : Decoder value -> Decoder (List value)

    Whereas bool is a decoder, list is a function that takes a decoder and returns a new one. We can use it in elm-repl to make decoders for lists of primitives, or even lists of lists!


    Listing 4.2 Using Json.Decode.list
> import Json.Decode exposing (list, bool, string, int)   > list <function> : Decoder a -> Decoder (List a)  ❶    > list bool Decoder (List Bool)   > list string Decoder (List String)   > list (list int)Decoder (List (List Int))

    ❶   elm-repl chooses “a” for the type variable we called “value”


    Decoding Objects


    The simplest way to decode an object is with the field function. Suppose we write this:
decoder : Decoder String decoder =    field "email" string

    When this decoder runs, it performs three checks:


    1. Are we decoding an Object?


    2. If so, does that Object have a field called email?


    3. If so, is the Object’s email field a String?


    If all three are true, then decoding succeeds with the value of the Object’s email field.


    Table 4.4 shows how this decoder would work on a variety of inputs.


    Table 4.4. Decoding objects with field decoders


    
      
        	
          
            Decoder

          

        

        	
          
            JSON

          

        

        	
          
            Result

          

        
      


      
        	field "email" string

        	5

        	Err "Expected object, got 5"
      


      
        	{"email": 5}

        	Err "Expected string for 'email', got 5"
      


      
        	{"email": "cate@nolf.com"}

        	Ok "cate@nolf.com"
      

    


    Decoding multiple fields


    Building a decoder for a single field is all well and good, but typically when decoding objects, we care about more than one of their fields. How do we do that?


    The simplest is with a function like map2, which we can see in Table 4.5.


    Table 4.5. Decoding objects with field decoders


    
      
        	
          
            Decoder

          

        

        	
          
            JSON

          

        

        	
          
            Result

          

        
      


      
        	map2    (\x y -> ( x, y ))    (field "x" int)    (field "y" int)

        	
 {"x": 5}

        	
 Err "Missing 'y' field"
      


      
        	{"x": 5, "y": null}

        	Err "Expected int for 'y', got null"
      


      
        	{"x": 5, "y": 12}

        	Ok ( 5, 12 )
      

    


    Decoding Many fields


    The photo information we’ll be getting back from our server will be in the form of JSON which looks like this:
{"url": "1.jpeg", "size": 36, "title": "Beachside"}

    This is an object with three fields: two strings and one int. Let’s update our Photo type alias to reflect this:
type alias Photo =     { url : String     , size : Int     , title : String    }

    We could decode this using thet technique we just learned, with one slight difference. We’d have to use map3 instead of map2, because we have three fields instead of two.
photoDecoder : Decoder Photo photoDecoder =     map3         (\url size title -> { url = url, size = size, title = title })         (field "url" string)         (field "size" int)        (field "title" string)

    How far can this approach take us? If we added a fourth field, we’d change map3 to map4. The Json.Decode module also includes map5, map6, map7, and map8, but map8 is as high as it goes.


    From there we can either combine the decoders we’ve already used...or we can introduce a library designed for larger-scale JSON decoding!


    Pipeline Decoding


    The Json.Decode.Pipeline is designed to make life easier when decoding large objects. It comes from a popular third-party package called NoRedInk/elm-decode-pipeline - so let’s install it real quick before we proceed:
elm-package install NoRedInk/elm-decode-pipeline

    Let’s also write a decoder for it. We’ll do that by deleting the following import:
import Html.Attributes exposing (..)

    ...and then adding all of this to the end of our imports list:
import Html.Attributes exposing (id, class, classList, src, name, type_, title) import Json.Decode exposing (string, int, list, Decoder)import Json.Decode.Pipeline exposing (decode, required, optional)

    Then the coast is clear to replace photoDecoder with this:


    Listing 4.2 photoDecoder
photoDecoder : Decoder Photo photoDecoder =     decode buildPhoto                           ❶          |> required "url" string                ❷          |> required "size" int                  ❸          |> optional "title" string "(untitled)" ❹      buildPhoto : String -> Int -> String -> Photo buildPhoto url size title =    { url = url, size = size, title = title }

    ❶   Pass decoded values to the buildPhoto function


    ❷   "url" is required, and must be a string


    ❸   "size" is required, and must be an integer


    ❹   "title" is optional, and defaults to "(untitled)"


    Note We changed our import Html.Attributes line because if it were still exposing (..), then we’d be exposing two functions called required: both Json.Decode.Pipeline.required and Html.Attributes.required. Referencing a value that’s exposed by two modules at once is an error!


    Let’s break down what’s happening here.


    1. decode buildPhoto begins the pipeline. It says that our decoder will decode the arguments to buildPhoto, one by one, and ultimately the whole decoder will succeed unless any of the steps in this pipeline fails. Since buildPhoto accepts three arguments, we’ll need three pipeline steps after this. otherwise, the compiler will give an error.


    2. required "url" string says that we need what we’re decoding to be a JSON object with the String field "url". We’re also saying that if decoding succeeds, we should use this first result as the first argument to buildPhoto. Decoding could fail here, because either the "url" field was missing, or because the field was present...but was not a String.


    3. required "size" int does the same thing required "url" string except that it decodes to an integer instead of a string.


    4. optional "title" string "(untitled)" - this is similar to the required steps, but with one important difference: in this example, if the "title" field were either missing or null, instead of failing decoding, this decoder will default on that final argument—that is, the title string would default to "(untitled)".


    Figure 4.8. shows how the buildPhoto arguments and photoDecoder arguments match up.

    Figure 4.8 The relationship between photoDecoder and buildPhoto


    By the way, notice anything familiar about what buildPhoto does?


    All it does is take one argument for each of the fields in Photo, and then assign them uncritically without altering them in any way. We already have a function that does this! It’s the Photo function, which we got it for free because we defined the Photo type alias.


    Let’s delete buildPhoto and replace photoDecoder = decode buildPhoto with this:
photoDecoder =     decode Photo

    Much easier!


    Warning Reordering any function’s arguments can lead to unpleasant surprises. Since reordering the fields in the Model type alias has the consqeuence of reordering the Model function’s arguments, you should be exactly as careful when reordering a type alias as you would be when reordering any function’s arguments!


    4.3.3   Decoding JSON HTTP Responses


    We’ve already seen how we can use Http.getString to obtain a String from a server, and how we can use decodeString and a Decoder to translate a String into a list of Photo records for our Model.


    Although we could use Http.getString and decodeString to populate our Model in this way, there is another function in the Http module which will take care of both for us.


    Http.get


    The Http.get function requests data from a server and then decodes it. Here is how the types of Http.getString and Http.get match up:
getString :                  String -> Request String get       : Decoder value -> String -> Request value

    Comparing types like this suggests how these functions are similar and how they differ.


    They both accept a URL string and return a Request. However, where getString takes no other arguments and produces a String on success, get additionally accepts a Decoder value, and on success produces a value instead of a String.


    As you might expect, if we give Http.get a decoder of (list int) and the response it gets back is the JSON payload "[1, 2, 3]", then Http.get will successfully decode that into Ok (List Int). If decoding fails, we will instead get Err UnexpectedPayload.


    What’s UnexpectedPayload? It’s a type constructor for that Http.Error type we said we’d get back to! (Sure enough, here we are getting back to it.) Http.Error is a union type which describes various ways a HTTP request can fail. It looks like this:
type Error     = BadUrl String     | Timeout     | NetworkError     | BadStatus (Response String)    | BadPayload String (Response String)

    Since it’s a union type, we can run a case-expression on any Http.Error value—with different branches for Timeout, NetworkError, and so on—to do custom error handling based on what went wrong.


    tip The lower-level Http.request function lets you customize requests in more depth, including performing error handling based on raw status code.


    Since we want to decode the JSON from our server into a List Photo, and we have a Decoder Photo, we can use Json.Decode.list to end up with the information we want. Let’s change initialCmd to do just that while it references our shiny new URL:
initialCmd : Cmd Msg initialCmd =     list photoDecoder         |> Http.get "http://elm-in-action.com/photos/list.json"        |> Http.send LoadPhotos

    This means we’ll be sending a List Photo to LoadPhotos instead of a String, so we’ll need to update its definition to match:
type Msg     = SelectByUrl String     | SelectByIndex Int     | SurpriseMe     | SetSize ThumbnailSize    | LoadPhotos (Result Http.Error (List Photo))

    This lets us simplify the LoadPhotos (Ok ...) branch of update’s case-expression quite a bit! Having direct access to photos means we no longer need to build that value up using a let-expression. We can also use our new friends Maybe.map and .url to determine selectedUrl right on the same line!
LoadPhotos (Ok photos) ->     ( { model         | photos = photos         , selectedUrl = Maybe.map .url (List.head photos)       }     , Cmd.none    )

    rendering the metadata


    Now all that remains is to have our view render the new caption and download size metadata!


    We can do this by adding one line to viewThumbnail:
viewThumbnail selectedUrl thumbnail =     img         [ src (urlPrefix ++ thumbnail.url)         , title (thumbnail.title ++ " [" ++ toString thumbnail.size ++ " KB]")         , classList [ ( "selected", selectedUrl == Just thumbnail.url ) ]         , onClick (SelectByUrl thumbnail.url)         ]        []

    Great! At this point everything should compile, and we can open index.html to see the result:

    Figure 4.9 The final application


    Now we’re reading our list of photos from the server rather than hardcoding them, and we’ve given ourselves a nice foundation on which to build an even richer application!


    4.4   Summary


    We learned quite a few things in the course of making Photo Groove talk to servers!


    ·   A Decoder can validate and translate JSON into an Elm value.


    ·   The Json.Decode module provides primitive decoders like float, int, and string.


    ·   The Json.Decode.list function turns a Decoder Bool into a Decoder (List Bool).


    ·   The Json.Decode.Pipeline module offers functions to decode objects in pipeline style.


    ·   Pattern matching lets us trade nested case-expressions for longer branch conditions.


    ·   A Result is either Ok okValue in case of success, or Err errValue in case of failure.


    ·   String.split splits a string around a given separator, resulting in a list of strings.


    ·   A Request describes a request we want to make. Http.send turns it into a Cmd.


    ·   Http.getString requests a plain String from the server at the given URL.


    ·   Http.get works like Http.getString, except it runs the given Decoder on the result.


    ·   The init field passed to Html.program lets us specify a Cmd to run on startup.


    We also saw how the pipeline operator (|>) lets us write expressions by starting with a value and then running a series of transformations on it. Table 4.6 shows an example of this.


    Table 4.6 The same expression with and without pipelines


    
      
        	
          
            Pipeline Style

          

        

        	
          
            Without Pipelines

          

        
      


      
        	model.photos     |> Array.fromList     |> Array.get index

        	Array.get index     (Array.fromList         model.photos)
      

    


    Now that we’ve gotten Photo Groove talking to a server, we’ll get it talking to JavaScript. This will let us tap into the enormous ecosystem of JavaScript libraries out there. Let’s see what that can get us!


    Listing 3.7 The complete PhotoGroove.elm
module PhotoGroove exposing (..)   import Html exposing (..) import Html.Events exposing (onClick) import Array exposing (Array) import Random import Http import Html.Attributes exposing (id, class, classList, src, name, type_, title) import Json.Decode exposing (string, int, list, Decoder) import Json.Decode.Pipeline exposing (decode, required, optional)     photoDecoder : Decoder Photo photoDecoder =     decode Photo         |> required "url" string         |> required "size" int         |> optional "title" string "(untitled)"     urlPrefix : String urlPrefix =     "http://elm-in-action.com/"     type ThumbnailSize     = Small     | Medium     | Large     view : Model -> Html Msg view model =     div [ class "content" ]         [ h1 [] [ text "Photo Groove" ]         , button             [ onClick SurpriseMe ]             [ text "Surprise Me!" ]         , h3 [] [ text "Thumbnail Size:" ]         , div [ id "choose-size" ]             (List.map viewSizeChooser [ Small, Medium, Large ])         , div [ id "thumbnails", class (sizeToString model.chosenSize) ]             (List.map (viewThumbnail model.selectedUrl) model.photos)         , viewLarge model.selectedUrl         ]     viewLarge : Maybe String -> Html Msg viewLarge maybeUrl =     case maybeUrl of         Nothing ->             text ""           Just url ->             img [ class "large", src (urlPrefix ++ "large/" ++ url) ] []     viewThumbnail : Maybe String -> Photo -> Html Msg viewThumbnail selectedUrl thumbnail =     img         [ src (urlPrefix ++ thumbnail.url)         , title (thumbnail.title ++ " [" ++ toString thumbnail.size ++ " KB]")         , classList [ ( "selected", selectedUrl == Just thumbnail.url ) ]         , onClick (SelectByUrl thumbnail.url)         ]         []     viewSizeChooser : ThumbnailSize -> Html Msg viewSizeChooser size =     label []         [ input [ type_ "radio", name "size", onClick (SetSize size) ] []         , text (sizeToString size)         ]     sizeToString : ThumbnailSize -> String sizeToString size =     case size of         Small ->             "small"           Medium ->             "med"           Large ->             "large"     type alias Photo =     { url : String     , size : Int     , title : String     }     type alias Model =     { photos : List Photo     , selectedUrl : Maybe String     , loadingError : Maybe String     , chosenSize : ThumbnailSize     }     initialModel : Model initialModel =     { photos = []     , selectedUrl = Nothing     , loadingError = Nothing     , chosenSize = Medium     }     photoArray : Array Photo photoArray =     Array.fromList initialModel.photos     getPhotoUrl : Int -> Maybe String getPhotoUrl index =     case Array.get index photoArray of         Just photo ->             Just photo.url           Nothing ->             Nothing     type Msg     = SelectByUrl String     | SelectByIndex Int     | SurpriseMe     | SetSize ThumbnailSize     | LoadPhotos (Result Http.Error (List Photo))     randomPhotoPicker : Random.Generator Int randomPhotoPicker =     Random.int 0 (Array.length photoArray - 1)     update : Msg -> Model -> ( Model, Cmd Msg ) update msg model =     case msg of         SelectByIndex index ->             let                 newSelectedUrl : Maybe String                 newSelectedUrl =                     model.photos                         |> Array.fromList                         |> Array.get index                         |> Maybe.map .url             in                 ( { model | selectedUrl = newSelectedUrl }, Cmd.none )           SelectByUrl url ->             ( { model | selectedUrl = Just url }, Cmd.none )           SurpriseMe ->             let                 randomPhotoPicker =                     Random.int 0 (List.length model.photos - 1)             in                 ( model, Random.generate SelectByIndex randomPhotoPicker )           SetSize size ->             ( { model | chosenSize = size }, Cmd.none )           LoadPhotos (Ok photos) ->             ( { model                 | photos = photos                 , selectedUrl = Maybe.map .url (List.head photos)               }             , Cmd.none             )           LoadPhotos (Err _) ->             ( { model                 | loadingError = Just "Error! (Try turning it off and on again?)"               }             , Cmd.none             )     initialCmd : Cmd Msg initialCmd =     list photoDecoder         |> Http.get "http://elm-in-action.com/photos/list.json"         |> Http.send LoadPhotos     viewOrError : Model -> Html Msg viewOrError model =     case model.loadingError of         Nothing ->             view model           Just errorMessage ->             div [ class "error-message" ]                 [ h1 [] [ text "Photo Groove" ]                 , p [] [ text errorMessage ]                 ]     main : Program Never Model Msg main =     Html.program         { init = ( initialModel, initialCmd )         , view = viewOrError         , update = update         , subscriptions = (\_ -> Sub.none)        }
  


  
    5    Talking to JavaScript


    This chapter covers


    ·   Rendering Custom Elements


    ·   Sending data to JavaScript


    ·   Receiving data from JavaScript


    ·   Initializing our Elm application using data from JavaScript


    Now we’ve gotten Photo Groove loading photos from a server. It’s looking better and better! However, our manager has a concern. “We have the photos, and they certainly look sharp. But where is the groove? This is Photo Groove, not Photo Browse! You know what we need? Filters. You know, the kind that make normal photos look all wacky and messed-up? Those!”


    Sounds groovy.


    We collaborate with our team’s visual designer to create a mockup of the filtering feature. The final design calls for a status bar to report on the filtering process, and sliders to control the filters’ settings. And not just any sliders! Our designer wants each to have the specific look and feel of a Paper Slider, a Web Component built on Google’s Material Design guidelines.


    As luck would have it, we won’t have to code either the filter effects or the sliders from scratch. They’re both available as open-source libraries, but there’s a catch: these libraries are written in JavaScript, not Elm.


    Fortunately, Elm applications are not limited to using Elm libraries alone! In this chapter we’ll expand Photo Groove to include a filtering feature. We’ll add sliders which let users apply filters in varying degrees to the large photo. We’ll create a status bar that displays information from the JavaScript libraries. Along the way we’ll learn two different ways to incorporate JavaScript code into an Elm application: custom elements and ports.


    Here we go!


    5.1   Using Custom Elements


    We’ll start with the sliders. We want to create three sliders in total: Hue, Ripple, and Noise. Adjusting each value will transform how the large photo appears to the end user.


    Figure 5.1 shows how the sliders fit into what we’ll be building toward in this chapter.

    Figure 5.1 Adding sliders which control filtering for the large photo


    Each slider controls an integer value which the filtering library will use for the magnitude of the filter effect. Sliding all three to 0 would disable filtering—leaving the photos unaltered, like how they look now—whereas setting all three to the maximum value of 11 would apply each filter at full power.


    Implementing the sliders using Custom Elements will involve three steps:


    1. Run some JavaScript code before our Elm code executes, to register custom behavior for all <paper-slider> elements on the page.


    2. Write Elm code to render these <paper-slider> elements using plain Html values.


    3. Write custom event handlers to update our model based on the sliders’ states.


    Along the way, we’ll learn a new use for the JSON Decoders we saw in Chapter 4, and compare two alternative ways of integrating the sliders into our data model. Let’s get to it!


    5.1.1   Importing Custom Elements


    The Paper Slider we’ll be using is implemented as a Custom Element, which is a part of the Web Components specification. Here’s how Custom Elements work:


    ·   Someone writes some JavaScript which defines a new element type. In this case the new element is a <paper-slider>.


    ·   We run that JavaScript code when our page loads, to register this custom element with the browser.


    ·   From now on, whenever any code—in JavaScript or in Elm—creates an element whose tag name is "paper-slider", that element will behave according to this custom logic.


    WARNING Custom Elements are implemented in JavaScript, so they may throw runtime exceptions!


    Registering Custom Elements


    To register the custom <paper-slider> element on our page, we’ll first need to import the code for it. Let’s add a <link rel="import"> to the <head> of our index.html file, right below the <link rel="stylesheet"> for our stylesheet:
<link rel="stylesheet" href="http://elm-in-action.com/styles.css"> <link rel="import" href="http://elm-in-action.com/wc/slider/paper-slider.html">

    This <link> tag loads the JavaScript code, CSS styles, and markup necessary to describe a custom <paper-slider> element, and then registers that <paper-slider> on the page. From now on, whenever an element with the tag name "paper-slider" appears on this page, it will behave according to the implementation we’ve imported here.


    You can find more Custom Elements at http://webcomponents.org/, or create and serve your own from any URL you like.


    
      Polyfilling Custom Elements for Older Browsers

    


    Since custom elements and <link rel="import"> tags are relatively new features, not all older browsers support them.


    We can fix this for browsers as old as Internet Explorer 11 by loading some JavaScript code known as a polyfill, which patches support for custom elements into these browsers. To polyfill support for Custom Elements, add this <script> between your two <link> tags:


     
<link rel="stylesheet" href="http://elm-in-action.com/styles.css"> <script src="http://elm-in-action.com/polyfill/webcomponents.js"></script> <link rel="import" href="http://elm-in-action.com/wc/slider/paper-slider.html">

     


    If your users only run modern browsers which have already implemented Custom Elements, you don’t need the polyfill and can omit this <script>. (That said, the polyfill has no effect on browsers which support Custom Elements, so it should be harmless aside from being more for users to download.)


    Custom elements from the Polymer Project use something called Shady DOM to improve performance in browsers that do not natively support certain parts of the Web Components specification. However, if you use Polymer elements inside other Polymer elements, they may not work properly in their stock configuration due to how Shady DOM interacts with virtual DOM systems such as Elm’s.


    A workaround is to disable Shady DOM by adding this before your first <link rel="import"> tag:


     
<script>window.Polymer = {dom: "shadow"};</script>

     


    For more information on configuring Shady DOM, see https://www.polymer-project.org/1.0/docs/devguide/settings


    In practice, polyfills are never as good as proper support for a browser feature. This polyfill often incurs significant performance penalties for some Custom Elements, and users have reported some polyfilled Custom Elements not working on Mobile Safari 10. Be sure to cross-browser test! See http://webcomponents.org/polyfills to learn more.


    
       

    


    Adding a <paper-slider> to the page


    Back at the beginning of Chapter 2 we saw the Html.node function. It takes three arguments:


    1. Tag name


    2. List of attributes


    3. List of children


    Here are two ways to create the same button, one using node and the other using button:
node "button" [ class "large" ] [ text "Send" ]       button  [ class "large" ] [ text "Send" ]

    Functions that create elements like button, label and input have tiny implementations—they do nothing more than call node. For example, label and input can be implemented like so:
label attributes children =     node "label" attributes children   input attributes children =    node "input" attributes children

    TIP Because of partial application, we could also have written these as label = node "label" and input = node "input" - either way works!


    Now that the <paper-slider> custom element has been registered on the page with our <link> tag, the only Elm code necessary to use it is to call node "paper-slider".


    Let’s add a function that does this, at the end of PhotoGroove.elm:
paperSlider =     node "paper-slider"

    This paperSlider function will work the same way as functions like button, div, and so on. It also has the same type as they do:
paperSlider : List (Attribute msg) -> List (Html msg) -> Html msg

    Go ahead and add this type annotation right above our paperSlider implementation.


    Viewing the sliders


    Next let’s invoke paperSlider to render the sliders. Add this beneath the view function:


    Listing 5.1 viewFilter
viewFilter : String -> Int -> Html Msg viewFilter name magnitude =     div [ class "filter-slider" ]         [ label [] [ text name ]                 ❶          , paperSlider [ max "11" ] []            ❷          , label [] [ text (toString magnitude) ] ❸         ]

    ❶   Display the filter’s name


    ❷   <paper-slider> that goes from 0 to 11


    ❸   Display the current magnitude


    We’ll also need to add max to the Html.Attributes we’re exposing in our imports:


    
      import Html.Attributes exposing (id, class, classList, src, name, max, type_, title)
    


    If we try to compile this, we’ll get a naming error: max is ambiguous! The compiler suggests:
Maybe you want one of the following?       Basics.max    Html.Attributes.max

    The cause of this error is that Elm has a built-in math function called max. It’s in the Basics module, and every Elm file has an implicit import Basics exposing (..) so that we can use common functions like negate, not, and max without having to qualify them as Basics.negate, Basics.not, or Basics.max.


    Resolving ambiguous names


    The compiler is reporting an ambiguity in our code: when we wrote max, we might have meant either Basics.max or Html.Attributes.max, since both max functions are exposed in this file.


    Let’s clarify which we meant by fully qualifying max:
paperSlider [ Html.Attributes.max "11" ] []

    This resolves the compiler error, but Html.Attributes.max is pretty verbose! We can shorten it by giving Html.Attributes an alias of Attr using the as keyword:
import Html.Attributes as Attr exposing (id, class, classList, src, name, type_, title)

    Now anywhere we would write Html.Attributes.foo we can write Attr.foo instead.


    Let’s refactor our input to look like this:
paperSlider [ Attr.max "11" ] []

    Splendid! This should make everything compile neatly.


    Rendering three Sliders


    Now we can call viewFilter in our view function to get a few sliders rendering. Let’s do that right below the code for the Surprise Me! button.
view model =     div [ class "content" ]         [ h1 [] [ text "Photo Groove" ]         , button             [ onClick SurpriseMe ]             [ text "Surprise Me!" ]         , div [ class "filters" ]             [ viewFilter "Hue" 0             , viewFilter "Ripple" 0             , viewFilter "Noise" 0            ]

    Figure 5.2 shows how this looks on the page.


    [image: ]



    Figure 5.2 Adding paperSliders for Hue, Ripple, and Noise


    Those sliders look slick, but they aren’t completely operational yet. We want the numbers next to the sliders to change as the user slides their values around. Let’s make that happen!


    5.1.2   Handling Custom Events


    If we take a quick glance at the Paper Slider’s official documentation - hosted at https://www.webcomponents.org/element/PolymerElements/paper-slider/paper-slider - we can see that the slider fires a custom event, immediate-value-changed, as the user slides.


    How might we respond to that event? Html.Attributes has a variety of built-in event handler functions such as onClick, but there’s no onImmediateValueChange. The immediate-value-changed event is a custom event built into paper-slider, and the Html.Attributes module has no idea it exists. So how can we specify a handler for it?


    Html.Attributes.on


    The Html.Attributes.on function lets us create a custom event handler, similarly to how the Html.node function lets us create a custom element. The  on function has this type:
on : String -> Decoder msg -> Attribute msg

    The String is the name of the event, which in this case is "immediate-value-changed". The Decoder argument is a Json.Decode.Decoder, the same type of decoder we built in Chapter 4 for our HTTP responses. Here we won’t be using the Decoder on a JSON string coming back from a server, but rather on a JavaScript event object!


    Decoding immediate-value-changed


    The event object JavaScript uses for "immediate-value-changed" looks something like this:
{target: {immediateValue: 7}}

    The target field refers to the slider object itself, which in turn has a field called value—an Int representing the slider’s current value. We can use the Json.Decode.field and Json.Decode.int functions to write a decoder for this like so:
field "target" (field "immediateValue" int)

    Table 5.1 compares this decoder to the example email decoder we wrote in Chapter 4.


    



    Table 5.1. Decoding objects with field decoders


    
      
        	
          
            Decoder

          

        

        	
          
            JSON

          

        

        	
          
            Result

          

        
      


      
        	
           


          field "email" string

        

        	
          5

        

        	
          Err "Expected object, got 5"

        
      


      
        	
          {"email": 5}

        

        	
          Err "Expected string for 'email', got 5"

        
      


      
        	
          {"email": "cate@nolf.com"}

        

        	
          Ok "cate@nolf.com"

        
      


      
        	
          
            Decoder

          

        

        	
          
            JavaScript

          

        

        	
          
            Result

          

        
      


      
        	
          field "target"


              (field


                  "immediateValue"


                  int)

        

        	
          9

        

        	
          Err “Expected object, got 9"

        
      


      
        	
          {"target": 9}

        

        	
          Err "Expected object for 'target', got 9”

        
      


      
        	
          {"target":


              {"immediateValue": 9}


          }

        

        	
          Ok 9

        
      

    



    Json.decode.at


    There’s a convenience function in Json.Decode for the case where we want to call field on another field like this: Json.Decode.at. It takes a list of field strings and traverses them in order. These two decoders do the same thing:
field "target" (field "immediateValue" int)  at [ "target", "immediateValue" ] int

    decoding a msg


    This Decoder Int will decode an integer from a JavaScript object such as {target: {value: 7 }}. But is that what we want? Let’s look at the type of on again:
on : String -> Decoder msg -> Attribute msg

    Notice that it wants a Decoder msg and then returns an Attribute msg. That tells us we want it to decode not an integer, but a message. Ah! We have a message type named Msg. So how do we convert between the Decoder Int that we have and the Decoder Msg that on expects?


    Using Json.decode.map


    The Json.Decode.map function is just what the doctor ordered! It converts between decoded values, as shown in Table 5.2.


    Table 5.2 Json.Decode.map


    
      
        	
          
            Expression

          

        

        	
          
            Description

          

        
      


      
        	
          Json.Decode.map negate float

        

        	
          Decode a float, then negate it.

        
      


      
        	
          Json.Decode.map (\num -> num * 2) int

        

        	
          Decode an integer, then double it.

        
      


      
        	
          Json.Decode.map (\_ -> "[[redacted]]") string

        

        	
          Decode a string, then replace it with "[[redacted]]" no matter what it was originally. Note that this will still fail if it attempts to decode a non-string value!

        
      

    


    Since Json.Decode.map takes a function which converts one decoded value to another, we can use it to convert our decoded Int into a Msg.


    Updating imports


    We’ll need to expose the on and at functions in our imports in order to reference them:
import Html.Events exposing (onClick, on) import Json.Decode exposing (string, int, list, Decoder, at)

    With those changes in place, we’re ready to add this onImmediateValueChange function to the end of PhotoGroove.elm:


    Listing 5.2 onImmediateValueChange
onImmediateValueChange : (Int -> msg) -> Attribute msg onImmediateValueChange toMsg =     let         targetImmediateValue : Decoder Int             ❶          targetImmediateValue =                         ❶              at [ "target", "immediateValue" ] int      ❶            msgDecoder : Decoder msg                       ❷          msgDecoder =                                   ❷              Json.Decode.map toMsg targetImmediateValue ❷      in        on "immediate-value-changed" msgDecoder        ❸ 

    ❶   Decode the integer located at event.target.immediateValue


    ❷   Convert that integer to a message using toMsg


    ❸   Create a custom event handler using that decoder


    Notice how onImmediateValueChange takes a toMsg function? This is because we need it to be flexible. We plan to have multiple sliders on the page, and we’ll want each of them to have a unique Msg constructor so that we can tell their messages apart. The toMsg argument lets us pass in the appropriate constructor on a case-by-case basis, which will come in handy later.


    Refactoring to use pipelines


    Notice how we assemble this value in three steps (targetImmediateValue, msgDecoder, and on), and each step’s final argument is the previous step’s return value? That means we can rewrite this to use the pipeline style we learned in Chapter 4! Let’s do that refactor:
onImmediateValueChange : (Int -> msg) -> Attribute msg onImmediateValueChange toMsg =     at [ "target", "immediateValue" ] int         |> Json.Decode.map toMsg        |> on "immediate-value-changed"

    Try walking through each step in the pipeline and finding the equivalent code in Listing 5.2. All the same logic is still there, just reorganized!


    Adding event handling to viewfilter


    Now that we have onImmediateValueChange, we can use it in our viewFilter function.


    Remember how we made onImmediateValueChange accept an Int -> msg function, so that we could pass it a different Msg constructor on a case-by-case basis? We’ll want viewFilter to have that same flexibility. The only difference will be that since viewFilter returns an Html Msg, we won’t be using a type variable like msg; instead, viewFilter will accept an Int -> Msg function.


    Listing 5.3 using onImmediateValueChange in viewFilter
viewFilter : String -> (Int -> Msg) -> Int -> Html Msg viewFilter name toMsg magnitude =     div [ class "filter-slider" ]         [ label [] [ text name ]         , paperSlider [ Attr.max "11", onImmediateValueChange toMsg ] [] ❶          , label [] [ text (toString magnitude) ]        ]

    ❶   Calling onImmediateValueChange just like we did onClick


    With this revised viewFilter implementation at the ready, our display logic is ready to be connected to our model and update. Once we’ve revised those to work with viewFilter, our shiny new Custom Elements will be fully integrated into our application!


    5.1.3   Responding to Slider Changes


    We have three filters, each of which has a name and a magnitude. How should we track their current values in our model? Let’s walk through two approaches, comparing the pros and cons of each, and then at the end decide which to use.


    The first approach would be to add three fields to the model: one for Hue, one for Ripple, and one for Noise. The alternative would prioritize flexibility, and store the filters as a list of { name : String, amount : Int } records. Table 5.3 shows these approaches side-by-side.


    Table 5.3 Storing filter data as three ints versus one list of records


    
      
        	
          
            Three Ints

          

        

        	
          
            One List of Records

          

        
      


      
        	  type alias Model =     { photos : List Photo     ...     , hue : Int     , ripple : Int     , noise : Int     }  

        	
            type alias Model =     { photos : List Photo     ...     , filters : List { name : String, amount : Int }     }

           

        
      


      
        	  initialModel : Model initialModel =     { photos = []     ...     , hue = 0     , ripple = 0     , noise = 0     }

        	  initialModel : Model initialModel =     { photos = []     ...     , filters =         [ { name = "Hue", amount = 0 }         , { name = "Ripple", amount = 0 }         , { name = "Noise", amount = 0 }         ]     }  
      

    


    Each model design has its own strengths and weaknesses, which become clearer when we write code that references these parts of the model.


    Let’s consider how our update implementations might compare. In either approach, we’d expand our Msg type’s constructors and then add at least one branch to our update function’s case-expression. Table 5.4 compares how these revisions would look for each approach.


     


    Table 5.4 Updating the model


    
      
        	
          
            Three Ints

          

        

        	
          
            One List of Records

          

        
      


      
        	  type Msg     = SelectByUrl String     | SetHue Int     | SetRipple Int     | SetNoise Int  

        	  type Msg     = SelectByUrl String     | SetFilter String Int
      


      
        	  case msg of     SetHue hue ->         ( { model | hue = hue }         , Cmd.none         )       SetRipple ripple ->         ( { model | ripple = ripple }         , Cmd.none         )       SetNoise noise ->         ( { model | noise = noise }         , Cmd.none         )

        	  case msg of     SetFilter name amount ->         let             transform filter =                 if filter.name == name then                     { name = name                     , amount = amount                     }                 else                     filter               filters =                 model.filters                     |> List.map transform         in             ( { model | filters = filters }             , Cmd.none             )  
      

    


    The “list of records” approach is more flexible. If we decided to add a fourth filter, such as “blur”, we could add it to initialModel and boom! It would appear on the page instantly. With the three integers approach, adding a fourth integer would require expanding not only initialModel, but also adding a field to our type alias for Model, a constructor for our Msg, and a branch for update’s case-expression. That sure sounds like more work!


    Finally let’s compare how rendering would look.


    Table 5.5 Viewing filter data from three fields versus one


    
      
        	
          
            Three Ints

          

        

        	
          
            List of Records (assuming viewFilter tweaked)

          

        
      


      
        	, div [ class "filters" ]     [ viewFilter SetHue "Hue" model.hue     , viewFilter SetRipple "Ripple" model.ripple     , viewFilter SetNoise "Noise" model.noise     ]

        	, div [ class "filters" ]     (List.map viewFilter model.filters)  
      

    


    Assuming we tweaked viewFilter to accept a { name : String, amount : Int } record, we could save ourselves a few lines of code with the “list of records” approach, using List.map as shown in Table 5.5. Once again “list of records” comes out ahead!


    Choosing an approach


    So far we’ve compared these approaches by conciseness, and also by how much effort it would take to add more filters later. However, our analysis is overlooking a crucial consideration, one which tends to make a much bigger difference in Elm than in JavaScript:


    Which approach rules out more bugs?


    As we’ve seen, Elm’s compiler gives us certain guarantees that can rule out entire categories of bugs. Some code can better leverage these guarantees than others! We can “help the compiler help us” by taking it into consideration when making decisions like these.


    What potential future headaches would each approach let the compiler rule out?


    Considering potential bugs


    Suppose we go with the “list of records” approach and one of our coworkers makes an innocuous typo—writing "Rippl" instead of "Ripple"—such that a SetFilter "Rippl" 5 message comes through our update function. That code will compile, but it won’t work properly because our filter.name == "Rippl" condition will never pass. We’ll have to hope our coworker catches this bug in testing!


    What if our coworker makes the same typo in the three-ints approach? Attempting to create a SetRippl 5 message will be a compiler error, because we named our Msg constructor SetRipple, not SetRippl.


    Table 5.6 Making a Mistake


    
      
        	
           

        

        	
          
            Three Ints

          

        

        	
          
            List of Records

          

        
      


      
        	
          Message

        

        	
          SetRippl 5

        

        	
          SetFilter “Rippl” 5

        
      


      
        	
          Outcome

        

        	
          Elm’s compiler tells us we have a naming error.

        

        	
          This compiles, but now we have a bug!

        
      

    


    Considering Future Changes


    What about making changes? Suppose in the future we need to rename “Ripple” to “Swirl”.


    In the “three ints” approach, we can rename our Msg constructor from SetRipple to SetSwirl, and our Model field from ripple to swirl—and then watch Elm’s compiler tell us exactly what other parts of our code base need changing as a result. If we miss anything, we’ll get a helpful type mismatch error telling us every spot we overlooked.


    With the “list of records” approach, we have to hope none of our coworkers ever wrote code using the hardcoded magic string "Ripple" instead of a constant. If they did, and something breaks, the only way we’ll find out before the bug reaches production is if someone also wrote a test that happens to fail when we change "Ripple" to "Swirl".


    Choosing Reliability


    By using individual fields instead of a list of records, we can rule out the entire category of bugs related to invalid filter names.


    Increasing conciceness and saving potential future effort are nice, but preventing bugs in a growing code base tends to be more valuable over time. Verbosity has a predictable impact on a project, whereas the impact of bugs can range from “quick fix” to “apocalyptic progress torpedo.” Let’s rule them out when we can!


    We’ll go with the approach that prevents more bugs. Take a moment to look back at Tables 5.3, 5.4, and 5.5, and implement the changes in the first column. Then let’s revise our calls to viewFilter to use our model’s new fields:
, div [ class "filters" ]     [ viewFilter "Hue" SetHue model.hue     , viewFilter "Ripple" SetRipple model.ripple     , viewFilter "Noise" SetNoise model.noise    ]

    When the dust settles, you should be able to recompile with elm-make --output=elm.js PhotoGroove.elm and see the number labels change when the user slides.


    [image: ]



    Figure 5.3 Sliding now changes the numbers next to the sliders


    Now that we have our sliders set up, we can move on to introducing the filters themselves!


    5.2   Sending Data to JavaScript


    Now our model has all the data it needs to calibrate the filters. We’ll use some JavaScript once again to apply the filters, although this time not Custom Elements. Instead we’ll write code that passes configuration options and a <canvas> element to a JavaScript function, which will then proceed to draw groovy pictures on it.


    First we’ll get a basic proof-of-concept working, to confirm that we’re successfully communicating across languages, and then we’ll smooth out the implementation details until we’re satisfied with how things are working.


    5.2.1   Creating a Command using a Port


    You may recall from Chapter 4 that an effect is an operation that modifies external state. You may also recall that if a function modifies external state when it runs, that function has a side effect. Elm functions are not permitted to have side effects, but JavaScript functions are.


    Talking to javaScript is like talking to servers


    Since calling any JavaScript function may result in a side effect, Elm functions cannot call JavaScript functions anytime they please; this would destroy the guarantee that Elm functions have no side effects!


    Instead, Elm talks to JavaScript the same way it talks to servers: by sending data out through a command, and receiving data in through a message. Figure 5.4 illustrates this.

    Figure 5.4 Using commands and messages to talk to both servers and JavaScript


    This means that talking to JavaScript will have some characteritics in common with what we saw when talking to servers in Chapter 4:


    ·   Data can only be sent using a command


    ·   Data can only be received by update, and that data must be wrapped in a message


    ·   We can translate and validate this incoming data using Decoders


    NOTE In JavaScript, some effects are performed synchronously, with program execution halting until the effect completes. In contrast, an Elm Cmd always represents an asynchronous effect. This means that when we send data to JavaScript, it’s always possible that other code might run before data gets sent back to Elm!


    Creating a Command


    Let’s create a Cmd to send some data to JavaScript. First we’ll define a type alias for the data we’re going to send, right above our type alias for Photo:
type alias FilterOptions =     { url : String     , filters : List { name : String, amount : Int }    }

    This represents all the information our JavaScript library will need in order to apply the filters: the URL of the photo in question, plus the list of filters and their amounts.


    Using a port to define a function


    In Chapter 4 we created a Cmd using the Http.send function, but here we’ll instead use a language feature designed specifically for talking to JavaScript: the port keyword.


    Add this right above the type alias for FilterOptions:
port setFilters : FilterOptions -> Cmd msg

    This declares a function called setFilters. Its type is (FilterOptions -> Cmd msg). We don’t write an implementation for this function, because the port keyword automatically writes one for us! port only needs to look at the type we requested to decide what the function should do.


    All port functions that send data to JavaScript are defined using a very specific pattern:


    ·   The port keyword must be followed by a function name and a type annotation


    ·   The type annotation must be for a function that takes one argument


    ·   The function must return Cmd msg, and nothing else—not even Cmd Msg!


    type variables in values


    Notice that our new setFilters function returns Cmd msg, not Cmd Msg. We know from Chapter 3 that the lowercase “m” means msg is a type variable, like the val type variable we saw in Array.fromList:
fromList : List val -> Array val

    This List String -> Cmd msg function is the first time we’ve seen a function annotation with a type variable in its return type, but no type variable in its parameters!


    What does that mean? And how is Cmd msg different from Cmd Msg?


    port commands Never Send messages


    Believe it or not, we’ve been using a Cmd msg for quite some time now! The official documentation for Cmd.none shows that it has this type:
none : Cmd msg

    We’ll dive deeper into values like this in future chapters, but for now it’s enough to know that a Cmd msg is a command which produces no message after it completes.


    Both Cmd.none and setFilters produce no message after completing. The difference is that Cmd.none has no effect, whereas setFilters will perform the effect of sending data to JavaScript. (Specifically, it will send the FilterOptions value we pass it.) You can think of setFilters as a “fire and forget” command.


    Table 5.7 Comparing Cmd Msg and Cmd msg.


    
      
        	
          
            Expression

          

        

        	
          
            Type

          

        

        	
          
            Effect

          

        

        	
          
            Message sent back to update

          

        
      


      
        	
          list photoDecoder


              |> Http.get "http://..."


              |> Http.send LoadPhotos

        

        	
          Cmd Msg

        

        	
          Send HTTP request

        

        	
          LoadPhotos

        
      


      
        	
          Cmd.none

        

        	
          Cmd msg

        

        	
          (none)

        

        	
          (no message sent back)

        
      


      
        	
          setFilters filterOptions

        

        	
          Cmd msg

        

        	
          Send filterOptions to JavaScript

        

        	
          (no message sent back)

        
      

    


    NOTE While HTTP requests can fail, sending data to JavaScript cannot. We don’t miss out on any error handling opportunities just because setFilters sends no message back to update.


    Figure 5.5 shows the similarities and differences between a Cmd originating from an Http.send and a Cmd originating from a port.

    Figure 5.5 Comparing a Cmd from Http.send to a Cmd from a port


    Port Modules


    Any module that uses the port keyword must be declared using the port module keyword, which means we’ll need to change the first line of PhotoGroove.elm to this:
port module PhotoGroove exposing (..)

    TIP If you’re ever wondering whether a given module talks directly to arbitrary JavaScript code, checking to see if it’s a port module is usually the quickest way.


    calling setfilters when users select photos


    Now that we have our port set up, we need to call setFilters in order to send some data to JavaScript. We’ll want to apply the filters every time the user selects a photo, so that’s when we’ll want update to return the command we get from setFilter.


    Here’s one way we could modify the SelectByUrl branch of our update function to do this.


    Listing 5.4 SelectByUrl
SelectByUrl selectedUrl ->     let         filters =             [ { name = "Hue", amount = model.hue }             , { name = "Ripple", amount = model.ripple }             , { name = "Noise", amount = model.noise }             ]           url =             urlPrefix ++ "large/" ++ selectedUrl           cmd =             setFilters { url = url, filters = filters }     in        ( model, cmd )

    However, the SelectByUrl branch is not the only way a user can select a photo! Users can also click the Surprise Me! button to select a photo at random, and we’ll want to apply filters in that situation as well. This means we’ll want to reuse the above code in two places: both the SelectByUrl branch of our case-expression as well as the SurpriseMe branch.


    Sharing Code between Update Branches


    Usually the simplest way to share code is to extract common logic into a helper function and call it from both places. This is just as true for update as it is for any function, so let’s do that!


    Note The structure of update permits clever alternatives to this venerable code sharing technique—like having update call itself, passing a different Msg. This saves us from writing another function, but it’s more error-prone and less peformant. If we do this, and later a teammate innocently changes how that other Msg responds to user input, our code breaks! Having update call itself also runs a bit slower because we create an unnecessary Msg and run an unnecessary case-expression on it. A helper function not only runs faster, it’s less error-prone to maintain because it explicitly signals to future maintainers that code is being reused.


    We’ll name the helper function applyFilters, and add it right below update:


    Listing 5.5 applyFilters
applyFilters: Model -> ( Model, Cmd Msg ) applyFilters model =     case model.selectedUrl of         Just selectedUrl ->             let                 filters =                     [ { name = "Hue", amount = model.hue }                     , { name = "Ripple", amount = model.ripple }                     , { name = "Noise", amount = model.noise }                     ]                   url =                     urlPrefix ++ "large/" ++ selectedUrl             in                 ( model, setFilters { url = url, filters = filters } )           Nothing ->            ( model, Cmd.none )

    Now we can have both SelectByIndex and SelectByUrl call applyFilters directly:
SelectByIndex index ->     let         newSelectedUrl : Maybe String         newSelectedUrl =             ...     in         applyFilters { model | selectedUrl = newSelectedUrl }   SelectByUrl selectedUrl ->    applyFilters { model | selectedUrl = Just selectedUrl }

    Lovely! Now whenever a user clicks either a photo or the Surprise Me! button, setFilters will return a Cmd that sends the appropriate FilterOptions value over to JavaScript.


    Next we’ll wire up the logic on the JavaScript side, which will receive that FilterOptions value and use it to apply some filters!


    5.2.2   Receiving Data from Elm


    Now we’re going to write some JavaScript code. Whenever we access a JS library from Elm, it’s best to write as little JS as possible. This is because if something crashes at runtime, it’s a safe bet that the culprit is somewhere in our JS code—so the less of it we have, the less code we’ll have to sift through to isolate the problem.


    Adding Pasta


    The JavaScript code we need comes from an open-source image filtering library intuitively named Pasta.js. We can import Pasta by adding this <script> tag to index.html, right before the <script> that imports our compiled elm.js file:
<div id="elm-area"></div>   <script src="http://elm-in-action.com/pasta.js"></script><script src="elm.js"></script>

    This <script> adds a global JavaScript function called Pasta.apply to the page. It takes two arguments:


    1. A <canvas> element, which is where apply will draw the filtered photos.


    2. An options object, which—in a remarkable coincidence—has the same structure as the FilterOptions record Elm will be sending to JavaScript via our setFilter port.


    Let’s introduce that <canvas>. We can do this with a quick change to PhotoGroove.elm: having viewLarge render a canvas instead of an img.


    Let’s replace the Just branch of the case-expression in viewLarge with this:
Just url ->     img [ class "large", src (urlPrefix ++ "large/" ++ url) ] []    canvas [ id "main-canvas", class "large" ] []

    Pasta.apply will take care of drawing the filtered photo onto this canvas. (The photo URL to draw will be sent to Pasta.apply via the FilterOptions we’re sending through our port.)


    Receiving data from the setFilters port


    We can set up a callback function that receives data from our setFilters function like so:
<script src="http://elm-in-action.com/pasta.js"></script> <script src="elm.js"></script> <script>   var app = Elm.PhotoGroove.embed(document.getElementById("elm-area"));     app.ports.setFilters.subscribe(function(options) {     Pasta.apply(document.getElementById("main-canvas"), options);   });</script>

    We’ve never needed the return value of Elm.PhotoGroove.embed before, but now we do!


    This object, typically called app, lets us subscribe to data Elm sends to JavaScript via ports like setFilters. When the Elm runtime executes the Cmd returned by setFilters, the callback function we’ve passed to app.ports.setFilters.subscribe will run. The options argument it accepts is the FilterOptions record Elm sent over, but converted from an Elm record to a JavaScript object.


    WARNING Like Custom Elements, ports invoke JavaScript code—which may throw runtime exceptions. If we make a mistake in any of our JavaScript code here, Elm’s compiler has no way to help us catch it.


    Table 5.8 shows how a port translates immutable Elm values (like the FilterOptions record) into brand-new (potentially mutable) JavaScript values.


    Table 5.8 Translating Elm values into JavaScript values via ports


    
      
        	
          
            Elm Value

          

        

        	
          
            Elm Type

          

        

        	
          
            JavaScript Value

          

        

        	
          
            JavaScript Type

          

        
      


      
        	
          "foo"

        

        	
          String

        

        	
          "foo"

        

        	
          string

        
      


      
        	
          4.2

        

        	
          Float

        

        	
          4.2

        

        	
          number

        
      


      
        	
          True

        

        	
          Bool

        

        	
          true

        

        	
          boolean

        
      


      
        	
          ( "foo", True, 4.2 )

        

        	
          ( String, Bool, Float )

        

        	
          ["foo", true, 4.2]

        

        	
          Array

        
      


      
        	
          [ "drive", "line" ]

        

        	
          List String

        

        	
          ["drive", "line"]

        

        	
          Array

        
      


      
        	
          { name = "Shadow" }

        

        	
          { password : String }

        

        	
          {"name": "Shadow"}

        

        	
          object

        
      


      
        	
          Nothing

        

        	
          Maybe val

        

        	
          null

        

        	
          object

        
      

    


    Tip Since all Elm values must be immutable, mutable values (like JavaScript objects and arrays) can’t be sent through the port. Instead they automatically get copied into immutable data structures. This process has some overhead. Passing in values like strings and numbers, which are already immutable, has no such overhead.


    Since we gave our <canvas> an id of "main-canvas", we can easily pass it to Pasta.apply by calling document.getElementById("main-canvas") to find it on the page by id.


    Trying it out


    Let’s try it out! If we open the page, we see...well, an empty rectangle.


    [image: ]



    Figure 5.6 An empty rectangle. Yep.


    That’s not what we wanted! We’ll figure that bug out later, but first let’s see what these sliders do! Crank up that Ripple value, and then...hm, still nothing happens. What’s the deal?


    One last idea: maybe try selecting a new photo?


    Oh hey, look at that!

    Figure 5.7 Increasing Ripple, then selecting another photo.


    Sure enough, things changed...quite a lot! We are definitely talking to that JavaScript library.


    However, there are some bugs to iron out. Playing around with it some more, we observe the following problems.


    1.   All of the Ripple and Noise values between 1 and 11 have the same effect.


    2.  Changing Hue doesn’t seem to do anything.


    3.  When we initially load the page, we see an empty rectangle instead of the first photo.


    Clearly we have some fixing to do!


    Converting the amount value to a percentage


    The reason Hue isn’t working is the same reason that the Ripple and Noise values between 1 and 11 do the same thing: we’re not quite using the Pasta.js API correctly.


    We’re sending an Int between 1 and 11, but actually Pasta.js is expecting a percentage—a Float between 0 and 1. Since JavaScript does not draw a distinction between Int and Float, this mistake does not result in a type mismatch error—not even at runtime!—and instead, our code simply does not work as expected.


    We can fix this by dividing our Model’s hue, ripple, and noise fields by 11 before sending them to JavaScript. Let’s revise our logic in applyFilters:
filters =     [ { name = "Hue", amount = toFloat model.hue / 11 }     , { name = "Ripple", amount = toFloat model.ripple / 11 }     , { name = "Noise", amount = toFloat model.noise / 11 }    ]

    We need that toFloat because Elm’s division operator (/) only works if you give it two Float values. The Basics.toFloat function converts an Int to a Float, so toFloat model.hue converts model.hue from an Int to a Float—at which point we can divide it by 11 as normal.


    Note This is another example of Elm’s design emphasizing being explicit. If JavaScript required similar explicitness, we’d have caught this bug earlier on, when we tried to pass an Int to an API expecting a Float.


    We’ll also need to update FilterOptions to expect a Float for the amount field.
type alias FilterOptions =     { url : String     , filters : List { name : String, amount : Float }    }

    Now if we recompile and try our sliders again, we see a much more interesting range of Hue, Ripple, and Noise values. Lovely! There are still some bugs left, but we’re making progress.


    Changing as we slide


    This will be way more fun if the photos update every time we slide, right? We know that applyFilters calls our setFilters port, so to apply the filters every time we slide, all we need to do is to run our applyFilters command every time we slide:
update : Msg -> Model -> ( Model, Cmd Msg ) update msg model =     case msg of         SetHue hue ->              applyFilters { model | hue = hue }           SetRipple ripple ->             applyFilters { model | ripple = ripple }           SetNoise noise ->            applyFilters { model | noise = noise }

    Recompile, and...presto! We’re still greeted with an empty rectangle—we’ll fix that next—but now whenever we slide, the photos update in realtime to reflect the new filter settings. Whee!


    Let’s fix that pesky bug where we see an empty rectangle instead of a photo on page load.


    5.2.3   Timing DOM Updates


    Showing the initial photo when the page loads takes two steps—a straightforward step and a tricky step. Let’s start with the straightforward one: what we do after loading the photos.


    Applying Filters after Loading Photos


    The problem here is that we currently load the photos and then re-render the view, but re-rendering the view is no longer enough to make a photo show up! We now need to call applyFilters so that Pasta can render something to the canvas.


    To fix this, we need the LoadPhotos branch of our update function’s case-expression to call applyFilters after updating model:
LoadPhotos (Ok photos) ->     applyFilters         { model             | photos = photos             , selectedUrl = Maybe.map .url (List.head photos)        }

    However, if you recompile and reload, you’ll see the same behavior. Why wasn’t that enough?


    timing ports and rendering


    The remaining problem is a matter of timing. Let’s break down what is happening:


    1.   We initialize model.photos to [] and model.selectedUrl to Nothing.


    2.  We request a list of photos from the server.


    3.  view runs, passing Nothing to viewLarge because model.selectedUrl is Nothing.


    4.  Because viewLarge received Nothing, it declines to render the <canvas>!


    5.  The server responds with our photos, meaning update gets a LoadPhotos message.


    6.  update runs its LoadPhotos branch and returns a new model (with a selectedUrl that is no longer Nothing), as well as a Cmd that will instruct our JavaScript code to have Pasta render to the <canvas>.


    See the problem? Step 6 tells Pasta to render to a <canvas>, but as we noted in Step 4, no such <canvas> has been created yet!


    This means our JavaScript call to Pasta.apply(document.getElementById("main-canvas"), options) will silently fail. Shortly after this happens, view will run again with the new model. This time, model.selectedUrl will not be Nothing—meaning viewLarge will happily ask for a fresh, blank <canvas>. Great.


    We were so close, though! If Elm would have rendered the view before running the Cmd that invoked Pasta, the <canvas> would have been on the DOM before the JavaScript executed, and everything would have been fine! So why doesn’t Elm do this?


    optimized DOM Updates


    One reason the Elm Runtime has good performance is that it skips unnecessary renders.


    See, browsers only repaint the DOM as pixels on users’ screens every so often. If the Elm Runtime changes part of the DOM, and then changes it again before the next repaint, the first change will have been wasted time; only the second change will be painted for users to see.


    You might think that if your update function gets called a million times in a single second, that your view would also be called a million times. Not so! Although those million updates will result in a million potential Model changes, Elm waits until the browser’s next repaint to call view even once—with whatever Model is at that moment. Invoking view more frequently than that would result in DOM updates that the browser wouldn’t bother to paint anyway.


    Synchronizing with the Elm Runtime


    The JavaScript function requestAnimationFrame allows code to run just before the browser’s next repaint. Since this is when the Elm runtime will schedule its next DOM update, we can use requestAnimationFrame to delay our call to document.getElementById("main-canvas") until after our next view has added the <canvas> we need to the DOM!


    Since requestAnimationFrame accepts a single callback function, this means we can finally fix this bug by wrapping our call to Pasta.apply in a requestAnimationFrame callback function, like so:
app.ports.applyFilters.subscribe(function(options) {   requestAnimationFrame(function() {     Pasta.apply(document.getElementById("main-canvas"), options);   });});

    Tip Anytime you need to trigger some JavaScript port code to run after the next time view results in a DOM update, you can synchronize things by wrapping your port code in requestAnimationFrame like this.


    Let’s recompile and bring up the page. Now the initial photo loads right after the page does!


    5.3   Receiving Data from JavaScript


    Now we’re ready for the final piece of the puzzle: the status bar, which displays information reported by Pasta. First we’ll start subscribing to Pasta’s realtime status updates, which Pasta broadcasts as they occur. Then we’ll set an initial status message which displays the version of Pasta we’re using. Each change involves a different Elm concept we haven’t used before: first subscriptions, and then flags. Let’s put them to work!


    5.3.1   Receiving realtime data from JavaScript via Ports


    Here we’ll be receiveing realtime data from JavaScript in the same way we receive user input: through a Msg. Let’s start by introducing a status string to our Model, initialModel, and Msg.


    
      
        	
          
            Model

          

        

        	
          
            initialModel

          

        

        	
          
            Msg

          

        
      


      
        	  type alias Model =     { photos : List Photo     , status : String

        	  initialModel : Model initialModel =     { photos = []     , status = ""  

        	  type Msg     = SelectByUrl String     | SelectByIndex Int     | SetStatus String  
      

    


    With these in place, our change to update is straightforward: set the status field on model when we receive a SetStatus message.
update msg model =     case msg of         SetStatus status ->            ( { model | status = status }, Cmd.none )

    We’ll also need view to display the status:
, button     [ onClick SurpriseMe ]     [ text "Surprise Me!" ] , div [ class "status" ] [ text model.status ], div [ class "filters" ]

    Nice! We’re getting faster at this.


    Now all that’s missing is a source of SetStatus messages. We’ll get those from JavaScript!


    Sending data from JavaScript to Elm


    Let’s modify index.html to add a bit more JavaScript code right below our setFilters code.
var app = Elm.PhotoGroove.embed(document.getElementById("elm-area"));   app.ports.setFilters.subscribe(function(options) {   requestAnimationFrame(function() {     Pasta.apply(document.getElementById("main-canvas"), options);   }); });   Pasta.addStatusListener(function(status) {                  ❶    console.log("Got a new status to send to Elm: ", status); ❷ });

    ❶   Pasta calls this function whenever its status changes


    ❷   Log the new status to the developer console


    This uses Pasta’s addStatusListener function to log status changes to the browser’s JavaScript console. If we reload the page and open the browser’s console, we’ll see various status messages flowing through as we play around with the sliders and select photos.


    Next we need to send those status strings to Elm instead of to the console!


    Subscriptions


    Earlier in the chapter we sent data from Elm to JavaScript using a command. Now we’ll do the reverse: send some data from JavaScript to Elm—using not a command, but a subscription.


    Definition A subscription represents a way to translate certain events outside our program into messages that get sent to our update function.


    One use for subscriptions is handling user inputs that aren’t tied to a particular DOM element. For example, we’ve used onClick to translate a click event on a specific element into a message that gets sent to update. What if we instead want to detect when the user resizes the entire browser window?


    We do this by adding a subscription to our program, which translates global window resize events into messages. Those messages get sent to update just like the ones for onClick do.

    Figure 5.8 A subscription that translates browser window resize events into messages


    Note If you’d like to try out Window.resizes you’ll need to run elm-package install elm-lang/window first. You can also find subscriptions for global mouse events in the elm-lang/mouse package, and subscriptions for global keyboard events in the elm-lang/keyboard package.


    Defining a subscription port


    There are also subscriptions which translate data from JavaScript into messages that are sent to update. We can get one of these subscriptions by using a slightly different port declaration.


    Let’s add another port right below the code that defines setFilters:
port setFilters : FilterOptions -> Cmd msg  port statusChanges : (String -> msg) -> Sub msg

    We can already see some similarities and differences between our command port and our subscription port. Both define a function that takes one argument, but whereas the first function returned a command (Cmd msg), the second returns a subscription (Sub msg).


    Cmd and Sub are both parameterized on the type of message they produce. We noted earlier how setFilters returns a Cmd msg (as opposed to Cmd Msg) because it is a command which produces no message after it completes. In contrast, statusChanges returns a Sub msg, but here msg refers to the type of message returned by the (String -> msg) function we pass to statusChanges.


    Table 5.9 shows how various calls to statusChange can yield different return values.


    Table 5.9 Calling the statusChanges function


    
      
        	
          
            Function to pass to statusChanges

          

        

        	
          
            Expression

          

        

        	
          
            Return Type

          

        
      


      
        	
          String.length : String -> Int

        

        	
          statusChanges String.length

        

        	
          Sub Int

        
      


      
        	
          String.reverse : String -> String

        

        	
          statusChanges String.reverse

        

        	
          Sub String

        
      


      
        	
          SetStatus : String -> Msg

        

        	
          statusChanges SetStatus

        

        	
          Sub Msg

        
      

    


    Note Whereas it’s normal for setFilters to return Cmd msg, it would be bizarre for statusChanges to return Sub msg. After all, a Cmd msg is a command that has an effect but never sends a message to update...but subscriptions do not run effects. Their whole purpose is to send messages to update! Subscribing to a Sub msg would be like listening to a disconnected phone line: not terribly practical.


    If we call statusChanges SetStatus, we’ll get back a Sub Msg subscription. That’s all well and good, but what do we do with a Sub Msg?


    Passing subscriptions to html.program


    We use it with that subscriptions field we set up with Html.program!
main : Program Never Model Msg main =     Html.program         { init = ( initialModel, initialCmd )         , view = viewOrError         , update = update         , subscriptions = \_ -> Sub.none         , subscriptions = \_ -> statusChanges SetStatus        }

    Ever since Chapter 3 we’ve been setting this subscriptions field to an anonymous function that always returned Sub.none. Now we’ve made that anonymous function return statusChanges SetStatus instead, which means that whenever JavaScript sends a string to the statusChanges port, it will result in a SetStatus message being sent to update.


    Note The argument this anonymous subscriptions function accepts is a Model. Whenever our model changes, the new model is passed to this function, giving us a chance to return a different Sub depending on what’s in the new model. This lets us dynamically control which subscriptions our program pays attention to.


    Figure 5.9 illustrates how subscriptions fit into Elm programs.

    Figure 5.9 How Subscriptions fit into Elm programs


    Now our statusChanges port is fully connected to the rest of our application, meaning we’re ready to have our JavaScript code start sending it data!


    Calling app.ports.statuschanges.send


    Much like how we used app.ports.setFilters.subscribe to receive data from Elm, we can use app.ports.statusChanges.send to send data to Elm. Let’s replace our console.log in index.html with a call to app.ports.statusChanges.send:
Pasta.addEventListener("statusChange", function(status) {   console.log("Got a new status to send to Elm: ", status);   app.ports.statusChanges.send(status); });

    Remember, subscriptions = \_ -> statusChanges SetStatus specifies that we’ll wrap whatever status string we receive from the statusChanges port in a SetStatus message.


    This means that calling app.ports.statusChanges.send("Reticulating splines") from JavaScript will ultimately result in a SetStatus "Reticulating splines" message being sent to update on the Elm side. At that point, our existing status-rendering logic should kick in, which we can confirm by recompiling and seeing how things look.

    Figure 5.10 Showing the status when filters change


    Fantastic! There’s just one more problem to address. Right when we refresh the page, at first there is no status to show, leading to a brief flicker of nothingness before the first status update arrives. We’ll fix this by showing an initial status, using data obtained from Pasta prior to the first status change it sends us.


    5.3.2   Receiving initialization arguments via Flags


    We’d like to display an initial status which reports the version of Pasta we’re using. It will look something like this:


    Initializing with Pasta v4.2...


    We can easily obtain the current version from Pasta in JavaScript; all we have to do is reference Pasta.version. But how do we get that value into Elm in time for the initial render, to avoid that flicker of missing content?


    Sending initialization data to programwithflags


    Calling app.ports.statusChanges.send(Pasta.version) won’t cut it here. Elm would still render initialModel (with model.status being "", resulting in the flicker of missing content) before receiving a message from statusChanges and re-rendering with the new status.


    What we want is a way to give Elm a value from JavaScript that’s available right away, early enough to use in our initial. This means we don’t want ports...we want flags.


    Definition Flags are values passed to an Elm program’s initialization function.


    To use flags we’ll need to switch from Html.program to Html.programWithFlags. The primary difference between the two is the type of init that each accepts. program wants init to be a ( Model, Cmd Msg ) tuple, whereas programWithFlags wants it to be a function which accepts initialization flags as an argument and returns a ( Model, Cmd Msg ) tuple.


    This change lets us receive a Float from JavaScript—namely Pasta.version—as an argument to init, giving us the chance to incorporate it into our initial model like so:
main =     Html.programWithFlags         { init = init         , view = viewOrError         , update = update         , subscriptions = (\_ -> Sub.none)         }   init : Float -> ( Model, Cmd Msg ) init flags =     let         status =             "Initializing Pasta v" ++ toString flags     in        ( { initialModel | status = status }, initialCmd )

    Switching to programWithFlags affects our type annotation for main a bit:
main : Program Never Model Msg main : Program Float Model Msg

    Program Never Model Msg means “a Program with Model as its model type, Msg as its message type, and which has no flags.” Program Float Model Msg means the same thing, except that its flags are a Float.


    
      The Never Type

    


    The type of a Html.program always begins with Program Never. What is this Never type, exactly?


    The definition of Never lives in the Basics module, and its implementation looks something like this:


     
type Never     = PassMeNever Never

     


    How might we obtain a value of type Never? That’s easy, just call PassMeNever! What does PassMeNever take as an argument? Oh, a Never. Hm...how can we obtain a Never to pass to PassMeNever? Ah yes—call PassMeNever! Now what type of argument does it take again?


    To make an endlessly long story short, it’s not possible to instantiate a Never. Any function which accepts a Never for an argument is a function that can never be called! The reason Html.program returns a Program Never Model Msg is that it can never be passed any flags, Float or otherwise. As you may recall, Html.program’s init is not  even a function, so we can never call it!


    
       

    


    Sending flags to Elm


    On the JavaScript side, we can send flags to Elm by passing an extra argument to the Elm.PhotoGroove.embed method we’ve been using to start the program running.
var app = Elm.PhotoGroove.embed(document.getElementById("elm-area"), Pasta.version);

    
      Decoding JavaScript Values

    


    We can use the same Decoders we used in Chapter 4 to translate values from JavaScript into Elm values while gracefully handling errors.


    In Chapter 4 we used the Json.Decode.decodeString function to decode a String of JSON into various Elm values. There is a similar function. Json.Decode.decodeValue, which takes a Json.Decode.Value instead of a String. We can write our statusChanges port to expect a Value from JavaScript, like so:
  import Json.Decode exposing (Value)   port statusChanges : (String -> msg) -> Sub msg port statusChanges : (Value -> msg) -> Sub msg  

    From there, writing a Decoder Msg and passing it to Json.Decode.decodeValue will give us a Result String Msg. We can translate that into a Msg by introducing an error-handling Msg which ideally displays a helpful message to end users, explaining that something went wrong.


    We can use the same technique for Flags:
  main : Program Value Model Msg

     


    Using this decoder style is generally a better choice for production applications. For one, it lets us give users a better experience by gracefully handling errors. For another, any incoming type other than Value is liable to result in a runtime exception if JavaScript passes in something invalid!


    Elm will check this immediately upon receiving the value, and will throw the exception early rather than letting it snake its way through our program, but it’s better still to do the error handling ourselves. Besides, since Elm performs the automatic check using decoders anyway, it’s not as if there’s a performance penalty for doing it explicitly!


    
       

    


    That’s it! Now when we start up Photo Groove, we’ll see the version number proudly displaying for the briefest of moments before the first photo loads. Figure 5.11 shows the initial status, captured using an incredible high-speed screenshotting camera.


    [image: ]



    Figure 5.11 Showing the status on page load


    Nicely done!


    5.4   Summary


    Photo Groove is now substantially groovier than before. Specifically:


    ·   We added some cool sliders using Custom Elements written in JavaScript


    ·   As users slide their values around, it changes the filters on our large photo


    ·   The filters come from a JavaScript library, which Photo Groove talks to like a server


    ·   When the JS filtering library sends status reports, Photo Groove displays them


    ·   The initial status report includes the JS filtering library’s version number


    In the process we learned some new concepts and techniques for building Elm applications:


    ·   We can resolve cyclical dependency errors by making functions more flexible


    ·   Asking “Which approach rules out more bugs?” is a good way to decide between different ways to model data


    ·   Once Custom Elements have been registered on the page, we can access them using  the Html.node function


    ·   The as keyword lets us alias imported modules, for example import Html.Attributes as Attr


    ·   The Html.Attributes.on function lets us create a custom event handler using a Decoder


    ·   requestAnimationFrame can delay our JavaScript code’s execution until the next time Elm calls view and renders the result to the DOM


    ·   A subscription represents a way to translate certain events outside our program into messages that get sent to our update function


    ·   A port module can use the port keyword to define functions that return commands and subscriptions which talk to JavaScript


    ·   Html.programWithFlags lets us send initialization flags from JavaScript to Elm


    In Chapter 6 we’ll crank our application’s maintainability into overdrive by building it a hearty suite of automated tests!


    Listing 5.6 index.html
<!doctype html> <html>   <head>       <link rel="stylesheet" href="http://elm-in-action.com/styles.css">       <link rel="import" href="http://elm-in-action.com/wc/slider/paper-slider.html">   </head>       <body>         <div id="elm-area"></div>           <script src="http://elm-in-action.com/pasta.js"></script>         <script src="elm.js"></script>         <script>           var app = Elm.PhotoGroove.embed(document.getElementById("elm-area"), Pasta.version);             app.ports.setFilters.subscribe(function(options) {             requestAnimationFrame(function() {               Pasta.apply(document.getElementById("main-canvas"), options);             });               Pasta.addStatusListener(function(status) {               app.ports.statusChanges.send(status);             });           });         </script>     </body> </html> 

    Listing 5.7 PhotoGroove.elm
port module PhotoGroove exposing (..)   import Html exposing (..) import Html.Events exposing (onClick) import Array exposing (Array) import Random import Http import Html.Attributes as Attr exposing (id, class, classList, src, name, max, type_, title) import Json.Decode exposing (string, int, list, Decoder) import Json.Decode.Pipeline exposing (decode, required, optional) import Slider exposing (paperSlider)     photoDecoder : Decoder Photo photoDecoder =     decode buildPhoto         |> required "url" string         |> required "size" int         |> optional "title" string "(untitled)"     buildPhoto : String -> Int -> String -> Photo buildPhoto url size title =     { url = url, size = size, title = title }     urlPrefix : String urlPrefix =     "http://elm-in-action.com/"     type ThumbnailSize     = Small     | Medium     | Large     view : Model -> Html Msg view model =     div [ class "content" ]         [ h1 [] [ text "Photo Groove" ]         , button             [ onClick SurpriseMe ]             [ text "Surprise Me!" ]         , div [ class "status" ] [ text model.status ]         , div [ class "filters" ]             [ viewFilter SetHue "Hue" model.hue             , viewFilter SetRipple "Ripple" model.ripple             , viewFilter SetNoise "Noise" model.noise             ]         , h3 [] [ text "Thumbnail Size:" ]         , div [ id "choose-size" ]             (List.map viewSizeChooser [ Small, Medium, Large ])         , div [ id "thumbnails", class (sizeToString model.chosenSize) ]             (List.map (viewThumbnail model.selectedUrl) model.photos)         , viewLarge model.selectedUrl         ]     viewFilter : (Int -> Msg) -> String -> Int -> Html Msg viewFilter toMsg name magnitude =     div [ class "filter-slider" ]         [ label [] [ text name ]         , paperSlider [ Attr.max "11", Slider.onImmediateValueChange toMsg ] []         , label [] [ text (toString magnitude) ]         ]     viewLarge : Maybe String -> Html Msg viewLarge maybeUrl =     case maybeUrl of         Nothing ->             text ""           Just url ->             canvas [ id "main-canvas", class "large" ] []     viewThumbnail : Maybe String -> Photo -> Html Msg viewThumbnail selectedUrl thumbnail =     img         [ src (urlPrefix ++ thumbnail.url)         , title (thumbnail.title ++ " [" ++ toString thumbnail.size ++ " KB]")         , classList [ ( "selected", selectedUrl == Just thumbnail.url ) ]         , onClick (SelectByUrl thumbnail.url)         ]         []     viewSizeChooser : ThumbnailSize -> Html Msg viewSizeChooser size =     label []         [ input [ type_ "radio", name "size", onClick (SetSize size) ] []         , text (sizeToString size)         ]     sizeToString : ThumbnailSize -> String sizeToString size =     case size of         Small ->             "small"           Medium ->             "med"           Large ->             "large"     port setFilters : FilterOptions -> Cmd msg     port statusChanges : (String -> msg) -> Sub msg     type alias FilterOptions =     { url : String     , filters : List { name : String, amount : Float }     }     type alias Photo =     { url : String     , size : Int     , title : String     }     type alias Model =     { photos : List Photo     , status : String     , selectedUrl : Maybe String     , loadingError : Maybe String     , chosenSize : ThumbnailSize     , hue : Int     , ripple : Int     , noise : Int     }     initialModel : Model initialModel =     { photos = []     , status = ""     , selectedUrl = Nothing     , loadingError = Nothing     , chosenSize = Medium     , hue = 0     , ripple = 0     , noise = 0     }     photoArray : Array Photo photoArray =     Array.fromList initialModel.photos     getPhotoUrl : Int -> Maybe String getPhotoUrl index =     case Array.get index photoArray of         Just photo ->             Just photo.url           Nothing ->             Nothing     type Msg     = SelectByUrl String     | SelectByIndex Int     | SetStatus String     | SurpriseMe     | SetSize ThumbnailSize     | SetHue Int     | SetRipple Int     | SetNoise Int     | LoadPhotos (Result Http.Error (List Photo))     randomPhotoPicker : Random.Generator Int randomPhotoPicker =     Random.int 0 (Array.length photoArray - 1)     update : Msg -> Model -> ( Model, Cmd Msg ) update msg model =     case msg of         SetStatus status ->             ( { model | status = status }, Cmd.none )           SetHue hue ->             applyFilters { model | hue = hue }           SetRipple ripple ->             applyFilters { model | ripple = ripple }           SetNoise noise ->             applyFilters { model | noise = noise }           SelectByIndex index ->             let                 newSelectedUrl : Maybe String                 newSelectedUrl =                     model.photos                         |> Array.fromList                         |> Array.get index                         |> Maybe.map .url             in                 applyFilters { model | selectedUrl = newSelectedUrl }           SelectByUrl selectedUrl ->             applyFilters { model | selectedUrl = Just selectedUrl }           SurpriseMe ->             let                 randomPhotoPicker =                     Random.int 0 (List.length model.photos - 1)             in                 ( model, Random.generate SelectByIndex randomPhotoPicker )           SetSize size ->             ( { model | chosenSize = size }, Cmd.none )           LoadPhotos (Ok photos) ->             applyFilters                 { model                     | photos = photos                     , selectedUrl = Maybe.map .url (List.head photos)                 }           LoadPhotos (Err _) ->             ( { model                 | loadingError = Just "Error! (Try turning it off and on again?)"               }             , Cmd.none             )     applyFilters : Model -> ( Model, Cmd Msg ) applyFilters model =     case model.selectedUrl of         Just selectedUrl ->             let                 filters =                     [ { name = "Hue", amount = (toFloat model.hue) / 11 }                     , { name = "Ripple", amount = (toFloat model.ripple) / 11 }                     , { name = "Noise", amount = (toFloat model.noise) / 11 }                     ]                   url =                     urlPrefix ++ "large/" ++ selectedUrl             in                 ( model, setFilters { url = url, filters = filters } )           Nothing ->             ( model, Cmd.none )     initialCmd : Cmd Msg initialCmd =     list photoDecoder         |> Http.get "http://elm-in-action.com/photos/list.json"         |> Http.send LoadPhotos     viewOrError : Model -> Html Msg viewOrError model =     case model.loadingError of         Nothing ->             view model           Just errorMessage ->             div [ class "error-message" ]                 [ h1 [] [ text "Photo Groove" ]                 , p [] [ text errorMessage ]                 ]     main : Program Float Model Msg main =     Html.programWithFlags         { init = init         , view = viewOrError         , update = update         , subscriptions = \_ -> statusChanges SetStatus         }     init : Float -> ( Model, Cmd Msg ) init flags =     let         status =             "Initializing Pasta v" ++ toString flags     in         ( { initialModel | status = status }, initialCmd )     paperSlider : List (Attribute msg) -> List (Html msg) -> Html msg paperSlider =     node "paper-slider"     onImmediateValueChange : (Int -> msg) -> Attribute msg onImmediateValueChange toMsg =     at [ "target", "immediateValue" ] int         |> Json.Decode.map toMsg         |> on "immediate-value-changed" 

     

  


  
    A  Getting Set Up


    This covers installing


    ·   Node.js (4.0 or higher) and NPM


    ·   The Elm Platform


    ·   The Elm in Action Repository


    ·   Recommended optional tools


    Installing Node.js and NPM


    In addition to having the Elm Platform installed, the examples in this book require having Node.js 4.0 or higher and NPM 2.0 or higher installed as well. If you haven’t already, visit https://nodejs.org and follow the instructions to download and install them.


    Note The Node.js installer on https://nodejs.org also installs NPM for you, so if you install Node that way, you will not need to install NPM separately.


    To confirm that you have Node.js 4.0 or higher and NPM 2.0 or higher installed, you should be able to run these commands in your terminal and see similar output:
$ node --version v4.0.0   $ npm --version2.0.0

    If you have Node installed but not NPM, then your installation of Node is probably not the one from http://nodejs.org. Please make sure you have NPM installed before continuing!


    Installing the Elm Platform


    Now that you have NPM installed, you can use it to get the Elm Platform:
npm install -g elm

    Tip If npm gives you a lengthy error involving the word EACCESS, visit https://docs.npmjs.com/getting-started/fixing-npm-permissions for how to resolve it. If you are unable to resolve it, you can fall back on installing the Elm Platform directly from http://elm-lang.org/install - but this will only get you through Chapters  1 through 5. Starting in Chapter 6, being able to run npm install -g will be required to run the examples!


    Let’s verify that the Elm platform was installed properly:
elm-make --version elm-make 0.18.0 (Elm Platform 0.18.0)

    If you see a version higher than 0.18.0, then as of this writing, you are living in the future! Hello from the past, where 0.18.0 is the latest release.


    Note Any version that starts with 0.18 should work fine with this book, but according to semantic versioning, a version number beginning with 0.19 or higher indicates breaking changes. In such a case, there’s a good chance your examples will not compile! Try npm install -g elm@0.18.0 to get a compatible version.


    Obtaining the Elm In Action Repository


    This book does not require using Git or any Git knowledge. If you are not a Git user, you can download and extract a zip archive of the repository at https://github.com/rtfeldman/elm-in-action/archive/master.zip and proceed directly to the next section!


    Git users can download the repository by running this:
git clone https://github.com/rtfeldman/elm-in-action.git

    The repository has been tagged with various checkpoints. Suppose you are about to read Chapter 2, Section 2.2.1. You can visit this URL to obtain all the code necessary to work through the examples in Section 2.2.1:
https://github.com/rtfeldman/elm-in-action/tree/2.2.1

    You can replace the 2.2.1 in that URL with a different section number to view the code at the start of that section.


    For example, if you want to peek ahead and see where things stand at the end of Section 2.2.1, you can bring up the 2.2.2 tag by visiting https://github.com/rtfeldman/elm-in-action/tree/2.2.2 in a browser. Alternatively, you can run git checkout 2.2.2 from a terminal if you ran git clone on the repository earlier.


    Installing Recommended Optional Tools


    To get syntax highlighting and other niceties, visit http://elm-lang.org/install#syntax-highlighting and select your editor of choice to find an appropriate Elm plugin. Make sure your editor is configured to convert tabs to spaces, as tab characters are syntax errors in Elm!


    I also strongly recommend installing elm-format: https://github.com/avh4/elm-format


    I absolutely love elm-format. It automatically formats Elm code according to a nice, consistent standard. I use it for personal projects, my whole team at work uses it, and I have my editor configured to run it whenever I save. (Several editors enable this by default, but if not, there are easy instructions in the above link for how to enable format-on-save.)


    All of the examples in this book are formatted with elm-format, so using it will make life easier as you work through them!

